1. INTERFASES.  

1.1 Introducción.

Podemos definir el concepto de interfase como el nexo de unión entre dos sistemas diferentes. Por tanto queda claro que este concepto no es exclusivo de los sistemas de información sino que  abarca un espectro más grande.

Si focalizamos nuestra atención en los sistemas de información , podemos darnos cuenta rápidamente , que dentro de una empresa  existen  diferentes entornos que necesitan comunicarse, necesitan interactuar.

Un claro ejemplo de necesidad de comunicación es la existente entre el  entono SAP y  el resto de aplicaciones existentes dentro de una empresa.

Podríamos hacer varias clasificaciones de interfases:

·  En cuanto a su naturaleza :

· Basadas en ficheros.

El intercambio de la información se realiza a través de ficheros.  Así el sistema origen elabora el fichero , con un determinado formato, y se lo envía al sistema destino que lo procesa.

· Basadas en funciones RFC (Remote Function Call).

El intercambio de información se realiza a través de aplicaciones que unen virtualmente los sistemas y ponen a disposición los datos del sistema origen en el sistema destino.


· En cuanto a su modo de ejecución:

· On-line.

Podemos decir que una interfase es on-line cuando el proceso que la controla es on-line . Normalmente las funciones de naturaleza RFC , serán on-line.

· Batch.

Una interfase es Batch cuando , el proceso que la controla es BATCH. Las interfases basadas en ficheros irán asociadas a procesos batch.

1.2 Tratamiento de ficheros.

1.2.1 Instrucciones básicas.
Para poder  trabajar con ficheros que se encuentren en el servidor de aplicación, tenemos las siguientes instrucciones básicas:

· OPEN DATASET nfich .

Abre un fichero para su tratamiento, si no se añade ninguna opción abre el fichero para lectura en modo binario.

Las principales opciones son:

· .. FOR OUTPUT
Abre el fichero para escritura. Si el fichero existe lo borra y lo crea nuevamente.

· .. FOR INPUT
Abre el fichero para lectura.

· .. FOR APPENDING.

Abre un fichero para escritura si el fichero no existe se crea. Si ya existe, se comenzará a escribir al final del fichero.


· ..IN BINARY MODE.

Abre el fichero en modo binario , es decir, la información se tratará por bloques. De tal modo que en cada lectura se leerán /escribirán  n caracteres.

· ..IN TEXT MODE.

Abre el fichero en modo texto, es decir, la información se tratara línea a línea. Cada lectura/escritura procesará una línea del fichero. 

· ..MESSAGE d_mes

Almacena en ‘d_mes’ el  mensaje devuelto por el sistema operativo.

· ..AT POSITION d_pos.

Sitúa el puntero de escritura en la posición d_pos (bytes) desde el inicio del fichero. (Esta opción , no tiene sentido emplearla para tratamientos en modo texto).

· .. TYPE

Permite definir atributos del fichero.

· .. FILTER

Permite ejecutar comandos propios del sistema operativo.

· READ DATASET  nfich INTO  wa.

Lee un bloque de información del fichero:

Si el fichero se abre en modo TEXTO:


Se lee una línea completa de tal forma que si el campo ‘wa’ es de menor tamaño que la línea , se perderá su contendido , en caso contrario se completará con blancos.  El puntero de lectura  se situará al comienzo de la línea siguiente.

Si el fichero se abre en modo BINARIO:


Se lee del fichero tantos bytes como tamaño tenga el campo ‘wa’.

· TRANSFER wa TO nfich.

Escribe el contenido del campo WA en el fichero abierto para escritura.

Si en modo texto:


Escribe una línea completa.

Si en modo binario:


Escribe tantos bytes como tenga el campo ‘wa’.

· CLOSE DATASET nfich.

Cierra un fichero abierto para lectura o escritura.

· DELETE DATASET nfich.

Borra el fichero nfich.

 

1.2.2 Creación de ficheros.
Vamos a ver un ejemplo de como leer y escribir registros en un fichero.

El ejemplo va a consistir en volcar a un fichero, nuestra tabla maestro de clientes con todos sus campos.  

La  estructura del fichero será  así:

	Campo 
	Tipo 
	Long.
	Descripción

	BUKRS
	CHAR
	4
	Sociedad

	NCLIE
	CHAR
	10
	Número de cliente

	NOMBR
	CHAR
	20
	Nombre cliente

	APEL1
	CHAR
	25
	Primer apellido

	APEL2
	CHAR
	25
	Segundo apellido

	FNACI
	DATS
	8
	Fecha de nacimiento


Crearemos un programa tipo ‘REPORT’ que llamaremos ‘ZINTERXX’.

Como parámetros de selección pondremos la sociedad obligatoria y como rango de selección el nº de cliente.

Añadiremos un nuevo parámetro obligatorio que representará el nombre del fichero.

El fichero que vamos a crear va a ser en modo texto.

El código del programa será algo así:

************************************************************************ 

* PROGRAMA: ZINTERXX                                                   * 

* DESCRIPCION:  Vuelca los registros seleccionado de la tabla          * 

*             maestro de clientes ZCLIENXX a un fichero de texto.      * 

* AUTOR : MAD00                     FECHA: 28/08/2001                  * 

*----------------------------------------------------------------------* 

* CONTROL DE MODIFICACIONES                                            * 

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         * 

************************************************************************ 

REPORT ZINTERXX.                                                         

************************************************************************ 

* Tablas del diccionario de datos                                      * 

************************************************************************ 

TABLES: ZCLIENXX.                   " Maestro de clientes.               

************************************************************************ 

* Tablas internas                                                      * 

************************************************************************

* Tabla interna de selección de clientes                                 

DATA: BEGIN OF I_ZCLIENXX OCCURS 0,                                      

       BUKRS LIKE ZCLIENXX-BUKRS,      "Sociedad                         

       NCLIE LIKE ZCLIENXX-NCLIE,      "Número de cliente                

       NOMBR LIKE ZCLIENXX-NOMBR,      "Nombre cliente                   

       APEL1 LIKE ZCLIENXX-APEL1,      "Primer apellido                  

       APEL2 LIKE ZCLIENXX-APEL2,      "Segundo apellido                 

       FNACI LIKE ZCLIENXX-FNACI,      "Fecha de nacimiento              

     END OF I_ZCLIENXX.                                                  

************************************************************************ 

* Definición de variables globales                                     * 

************************************************************************ 

 DATA :D_MERROR(100) TYPE C.         " Mensaje de error                  

************************************************************************ 

* Pantalla de selección                                                * 

************************************************************************

* Parámetros                                                            

SELECTION-SCREEN BEGIN OF BLOCK BLOQ1 WITH FRAME TITLE TEXT-001.        

PARAMETERS P_BUKRS LIKE ZCLIENXX-BUKRS OBLIGATORY                       

                                       DEFAULT '0001'. " Sociedad       

SELECT-OPTIONS:                                                         

             S_NCLIE FOR ZCLIENXX-NCLIE. "  Nº de cliente               

SELECTION-SCREEN END OF BLOCK BLOQ1.                                    

* Parámetro nombre de fichero destino.                                  

SELECTION-SCREEN BEGIN OF BLOCK BLOQ2 WITH FRAME TITLE TEXT-002.        

* Nombre del fichero (activamos minúsculas permitidas).               

PARAMETERS P_NFICH(50) TYPE C OBLIGATORY DEFAULT ‘/usr/sap/tmp/zclienxx.dat’ LOWER CASE

.                                      

SELECTION-SCREEN END OF BLOCK BLOQ2.                                    

************************************************************************

* Comienzo de selección                                                *

************************************************************************

START-OF-SELECTION.                                                   

*  Seleccionamos los datos de los clientes que cumplen criterios      

   SELECT *                                                           

   FROM ZCLIENXX INTO CORRESPONDING FIELDS OF TABLE I_ZCLIENXX        

   WHERE BUKRS = P_BUKRS                                              

   AND   NCLIE IN S_NCLIE.                                            

*  Comprobamos si hay datos seleccionados en cuyo caso ,              

*  crearemos el fichero                                               

   IF ( SY-SUBRC = 0 ).                                               

*    Abrimos el fichero para escritura en modo texto                  

OPEN DATASET P_NFICH FOR OUTPUT IN TEXT MODE MESSAGE D_MERROR.   

*    Comprobamos si no hubo error                                     

     IF ( SY-SUBRC = 0 ).                                             

*      Para cada uno de los registros seleccionados                   

       LOOP AT I_ZCLIENXX.                                            

*        Escribimos en el fichero                                     

         TRANSFER I_ZCLIENXX TO P_NFICH.

      ENDLOOP.                                                       

*      Cerramos el fichero                                            

       CLOSE DATASET P_NFICH.                                         

*      Mostramos mensaje 'Fichero creado con éxito'.                  

       MESSAGE I000(38) WITH TEXT-003.                                

*    Si hay error , mostramos el mensaje, con un mensaje de error      

     ELSE.                                                            

        MESSAGE E000(38) WITH D_MERROR.                               

     ENDIF.                                                           

*  Si no hay datos seleccionados.                                     

   ELSE.                                                              

*   Mostramos mensaje 'No se seleccionaron datos'.                    

    MESSAGE E000(38) WITH TEXT-004.                                   

   ENDIF.

1.2.3 Lectura de Ficheros.
Vamos a continuación a realizar un programa (ZINTER1XX) que muestre un listado con el contenido de un fichero. Dando la posibilidad de borrar el fichero una vez procesado.

************************************************************************ 

* PROGRAMA: ZINTE1XX                                                   * 

* DESCRIPCION:  Muestra un listado con el contenido de un fichero      * 

* AUTOR : MAD00                     FECHA: 28/08/2001                  * 

*----------------------------------------------------------------------* 

* CONTROL DE MODIFICACIONES                                            * 

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         * 

************************************************************************ 

REPORT ZINTE1XX NO STANDARD PAGE HEADING.                                

************************************************************************ 

* Definición de variables globales                                     * 

************************************************************************ 

 DATA :D_MERROR(100) TYPE C,         " Mensaje de error

D_LINEA(255)  TYPE C.         " Línea de fichero.                 

************************************************************************ 

* Pantalla de selección                                                * 

************************************************************************ 

* Parámetro nombre de fichero                                            

SELECTION-SCREEN BEGIN OF BLOCK BLOQ2 WITH FRAME TITLE TEXT-002.         

* Nombre del fichero (activamos minúsculas permitidas).                

PARAMETERS: P_NFICH(50) TYPE C OBLIGATORY DEFAULT        '/usr/sap/tmp/zclienxx.dat'     LOWER CASE,                                      

            P_DELETE    AS CHECKBOX.  " Flag de borrado fichero          

SELECTION-SCREEN END OF BLOCK BLOQ2.                                     

************************************************************************ 

* Comienzo de selección                                                * 

************************************************************************

START-OF-SELECTION.                                                      

*  Abrimos el fichero para lectura en modo texto.                        

   OPEN DATASET P_NFICH FOR INPUT IN TEXT MODE MESSAGE D_MERROR.         

*  Comprobamos si no hubo error                                          

   IF ( SY-SUBRC = 0 ).                                                  

*    Leemos el primer registro del fichero                               

     READ DATASET P_NFICH INTO D_LINEA.                                  

*    Mientras no haya error de lectura                                   

     WHILE ( SY-SUBRC = 0 ).                                             

*      Escribimos la línea leída                                         

       WRITE / D_LINEA.                                                  

*      Leemos el siguiente registro                                      

       READ DATASET P_NFICH INTO D_LINEA.                                

     ENDWHILE.                                                           

*    Una vez procesado el fichero, borramos el fichero si está           

*    marcado el flag de borrado                                          

     CHECK NOT ( P_DELETE IS INITIAL ).

DELETE DATASET P_NFICH.                                         

*    Comprobamos si se pudo borrar                                   

      IF ( SY-SUBRC = 0 ).                                           

*       Mostramos mensaje de 'Fichero borrado'.                      

        MESSAGE S000(38) WITH TEXT-005.                              

      ENDIF.                                                         

*  Si hay error , mostramos el mensaje, con un mensaje de error      

   ELSE.                                                             

      MESSAGE E000(38) WITH D_MERROR.                                

   ENDIF.

1.2.4 Explorar ficheros.
Para poder visualizar los ficheros existentes en el servidor de aplicación disponemos de la  transacción  estandar AL11  que nos permite explorar el contenido del directorio /usr/sap.

Haciendo doble-click sobre el directorio a explorar  ‘/usr/sap/tmp’ en nuestro caso podremos visualizar los ficheros existentes así como el contenido de los mismos.

Nota: Esta es una herramienta bastante limitada ya que únicamente nos dejará visualizar el contenido de esta ruta (/usr/sap) , normalmente dispondremos de otras herramientas para poder visualizar y tratar ficheros en el servidor de aplicación.

1.3 Batch Input .

1.3.1 Introducción.
En capítulos anteriores , hemos visto como se introducen datos, que actualizan las bases de datos,  a través de pantallas o dynpros que se llaman dentro de una transacción . Introducimos los datos , navegando por cada una de las pantallas completando los datos en los campos y pulsando las opciones oportunas (pantalla siguiente , grabar …). En todo este proceso , es necesario la intervención de un usuario.

El sistema Batch Input o BDC (Batch Data Communication) para introducir datos en la base de datos de SAP tiene las siguientes características:

· Puede procesar grandes volúmenes de datos.

· Se puede planificar y ejecutar como proceso de fondo o como proceso on-line.

· Se puede procesar sin la intervención de un usuario.

· Permite usar transacciones como si los datos se introdujeran de forma manual, pero realizando idénticas comprobaciones de integridad, autorizaciones…

En este sistema de introducción de datos , es necesario tener un fichero especial que llamaremos ‘Juego de datos’ en el que se tiene perfectamente estructurada la siguiente información:

· Por una parte la información especifica a introducir .

· Por otra parte los comandos , pantallas, nombres de campo … en las que introducir la información.

Estos juegos de datos tienen siempre la misma  forma definida en la estructura estándar  ‘BDCDATA’ :

Construir el ‘Juego de datos’ consiste en ir añadiendo registros secuencialmente , en los que se definen las secuencias de pantalla necesarias y datos para lograr su almacenamiento en las tablas SAP.

Tendremos dos tipos de registros (aunque con la misma estructura).

· Registros de Identificación de programa y dynpro.

En este tipo de registros , se detalla en nombre del programa y nº de dynpro que va a contener los campos y los valores que van a tomar los registros siguientes.

· Registros de campos y valores .

Este tipo de registros , contienen el nombre de los campos del dynpro así como los datos específicos que se deben ‘introducir’ en ellos.  Un  registro especial dentro de este tipo es el que contiene el valor del comando de dynpro. BDC_OKCODE.

La secuencia de registros BDC será algo así:

	PROGRAM
	DYNPRO
	DYNBEGIN
	FNAM
	FVAL

	Programa1
	1ddd
	X
	
	

	
	
	
	Campo1d
	Valor campo1d

	
	
	
	Campo2d
	Valor campo 2d

	
	
	
	BDC_OKCODE
	Comando

	Programa2
	2ccc
	X
	
	

	
	
	
	Campo 1c
	Valor campo 1c

	
	
	
	Campo 2c
	Valor campo 2c

	
	
	
	BDC_OKCODE
	Comando

	….
	
	
	
	

	Programa n
	Njjj
	X
	
	

	
	
	
	Campo 1c
	Valor campo 1c

	
	
	
	Campo 2c
	Valor campo 2c

	
	
	
	BDC_OKCODE
	Comando


Para almacenar estos registros en un fichero especial al que denominamos ‘Juego de datos ‘ disponemos de tres funciones estándares que nos facilitan esta tarea:

· BDC_OPEN_GROUP.

Nos permite abrir la sesión para la creación del juego de datos y que éste se almacene en la cola de juegos de datos pendientes.

Los parámetros principales son:

GROUP:

Nombre con el que se almacenará el juego de datos.

USER:

Usuario con el que se ejecutará el juego de datos.

HOLDDATE:
Fecha a partir de la cual el juego de datos puede ser ejecutado.

KEEP:
Flag de indicador de borrado del juego de datos tras su ejecución con éxito. (‘ ‘ permanece , ‘X’ no se borra).

 

· BDC_INSERT.

Permite incluir los registros al juego de datos a través de una tabla interna.

Hay que indicar los parámetros.

TCODE:

Código de transacción a ejecutar.

DYNPROTAB:
Tabla interna que contiene los registros.

· CLOSE_GROUP.

Cierra el Juego de datos.

Una vez creado el fichero BDC o ‘Juego de datos’, este quedará almacenado para su posterior tratamiento:  Ejecución , borrado … .

Para saber el nombre del programa ,dynpro , nombre de los campos …. con los que debemos completar la tabla  DBC tenemos dos posibilidades:

· Obtener el nombre de los campos mediante la ayuda on-line disponible en todas las pantallas.

Para obtener los datos de un campo , situados sobre el campo de pantalla pulsaremos ‘F1’, nos saldrá una ventana mostrando la descripción del campo , esta ventana tiene una opción [image: image3.png]Datos técnicos



 pulsándolo nos aperecera una ventana con este aspecto:

Donde se muestra información sobre el programa , nombre del campo …

· Utilizar la utilidad de ‘grabar’ una transacción. 

Podemos acceder a ella a través del  menú Sistema (Existirá en todos los menús) seguiremos la ruta:

Servicios(Batch Input ( Grabadora (SHDB).

Llegaremos a la siguiente pantalla :

En esta pantalla podremos  crear grabaciones  , visualizar existentes, .... .

 Para crear una nueva , pulsamremos el botón [image: image6.png][ Grabacién nueva



 deberemos indicar un nombre para la grabación ‘ALTAS’ e indicaremos el código de la transacción ‘ZBXX’ .  A continuación pulsaremos el botón [image: image7.png]Iniciar grabacin



  nos saldrá la transacción preparada par introducir los datos, introduciremos los datos correspondientes (del mismo modo que si estubiéramos en la transacción directamente) al finalizar nos saldrá una pantalla con el siguiente aspecto:

Donde podremos visualizar el nombre del programa , campos , comandos …

1.3.2 Creación del juego de datos. 

Suponemos el siguiente caso. Tenemos que dar de alta a todos los clientes en nuestra tabla maestro de clientes ZCLIENXX , para ello , disponemos:

· De la transacción ‘ZBXX’(alta de clientes que creamos en capítulos anteriores).    

· Tenemos la información de los clientes almacenada en un fichero ‘/usr/sap/tmp/zclienxx.dat’.  (Cuya estructura es la que vimos en el apartado anterior).

Vamos a crear un programa ‘ZINTE2XX’que  lea los datos del fichero y elabore un juego de datos para dar de alta clientes .

El nombre del juego de datos y el nombre del fichero , serán parámetros del programa.

El código del programa será algo así:

************************************************************************ 

* PROGRAMA: ZINTE2XX                                                   * 

* DESCRIPCION:  Genera un Juego de datos de Alta de clientes , con     * 

*       los datos de los clientes que lee de un fichero.               * 

* AUTOR : MAD00                     FECHA: 28/08/2001                  * 

*----------------------------------------------------------------------* 

* CONTROL DE MODIFICACIONES                                            * 

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         * 

************************************************************************ 

REPORT ZINTE2XX NO STANDARD PAGE HEADING.                                

************************************************************************ 

* Tablas del diccionario de datos                                      * 

************************************************************************ 

TABLES: ZCLIENXX.                   " Maestro de clientes.               

************************************************************************ 

* Definición de tablas internas                                        * 

************************************************************************ 

* Tabla interna para almacenar los registros del fichero

DATA: BEGIN OF I_ZCLIENXX OCCURS 0,                                         

       BUKRS LIKE ZCLIENXX-BUKRS,      "Sociedad                            

       NCLIE LIKE ZCLIENXX-NCLIE,      "Número de cliente                   

       NOMBR LIKE ZCLIENXX-NOMBR,      "Nombre cliente                      

       APEL1 LIKE ZCLIENXX-APEL1,      "Primer apellido                     

       APEL2 LIKE ZCLIENXX-APEL2,      "Segundo apellido                    

       FNACI LIKE ZCLIENXX-FNACI,      "Fecha de nacimiento                 

     END OF I_ZCLIENXX.                                                     

* Tabla interna para almacenar los registros del juego de datos             

 DATA: BEGIN OF I_BDCTAB OCCURS 0.                                          

        INCLUDE STRUCTURE BDCDATA.                                          

 DATA: END OF I_BDCTAB.                                                     

************************************************************************    

* Definición de variables globales                                     *    

************************************************************************    

 DATA :D_MERROR(100) TYPE C.         " Mensaje de error

************************************************************************ 

* Pantalla de selección                                                * 

************************************************************************ 

* Parámetro nombre de fichero                                            

SELECTION-SCREEN BEGIN OF BLOCK BLOQ2 WITH FRAME TITLE TEXT-002.         

* Parámetros                                                            

PARAMETERS:                                                              

* Nombre del fichero (activamos minúsculas permitidas).                

       P_NFICH(50) TYPE C OBLIGATORY DEFAULT '/usr/sap/tmp/zclienxx.dat'         

                        LOWER CASE,                                      

*  Nombre del juego datos                                                

       P_JDATOS(12) TYPE C OBLIGATORY DEFAULT 'ZCLIENXX'.                

SELECTION-SCREEN END OF BLOCK BLOQ2.

************************************************************************ 

* Comienzo de selección                                                * 

************************************************************************ 

START-OF-SELECTION.                                                      

*  Abrimos el fichero para lectura en modo texto.                        

   OPEN DATASET P_NFICH FOR INPUT IN TEXT MODE MESSAGE D_MERROR.         

*  Comprobamos si no hubo error                                          

   IF ( SY-SUBRC = 0 ).                                                  

*    Inicializamos la tabla interna                                     

     REFRESH I_ZCLIENXX.                                                 

     CLEAR I_ZCLIENXX.                                                   

*    Leemos el primer registro del fichero sobre la tabla interna        

     READ DATASET P_NFICH INTO I_ZCLIENXX.                               

*    Mientras no haya error de lectura                                   

     WHILE ( SY-SUBRC = 0 ).                                             

*      Añadimos el registro leído a la tabla interna                     

       APPEND I_ZCLIENXX.                                                

*      Inicializamos la cabecera

        CLEAR I_ZCLIENXX.                                                

*      Leemos el siguiente registro                                     

       READ DATASET P_NFICH INTO I_ZCLIENXX.                            

     ENDWHILE.                                                          

*    Creamos el juego de datos con el contenido de la tabla I_ZCLIENXX  

     PERFORM CREAR_JUEGO_DATOS.                                         

*    Mostramos mensaje de 'Juego de datos creado'.                      

     MESSAGE I000(38) WITH TEXT-003 P_JDATOS.                           

*  Si hay error , mostramos el mensaje, con un mensaje de error         

   ELSE.                                                                

      MESSAGE E000(38) WITH D_MERROR.                                   

   ENDIF.

***********************************************************************

 Rutinas adicionales.                                                 *

***********************************************************************

*&---------------------------------------------------------------------*

*&      Form  CREAR_JUEGO_DATOS                                         

*&---------------------------------------------------------------------*

*      Crea un juego de datos con el nombre indicado por el parámetro *

*  P_JDATOS, para la transacción ZBXX con el  contenido de la        *

* tabla interna I_ZCLIENXX.                                         *

*----------------------------------------------------------------------*

*FORM CREAR_JUEGO_DATOS.                                                 

************************************************************************

*Definición de variables locales                                      *

************************************************************************

DATA: L_FNACI(10) TYPE C.          " Formateo de fecha para pantalla.

************************************************************************ 

* Proceso                                                              * 

************************************************************************ 

* Comprobamos que la tabla interna I_ZCLIENXX tiene registros            

  CHECK NOT ( I_ZCLIENXX[] IS INITIAL ).                                 

* Abrimos el juego de datos.                                             

  CALL FUNCTION 'BDC_OPEN_GROUP'                                         

       EXPORTING                                                         

*           CLIENT              = SY-MANDT                               

*           DEST                = FILLER8                                

            GROUP               = P_JDATOS                               

*           HOLDDATE            = FILLER8                                

            KEEP                = ' '                                    

            USER                = SY-UNAME                               

*           RECORD              = FILLER1                                

*      IMPORTING                                           

*           QID                 =                          

       EXCEPTIONS                                          

            CLIENT_INVALID      = 1                        

            DESTINATION_INVALID = 2                        

            GROUP_INVALID       = 3                        

            GROUP_IS_LOCKED     = 4                        

            HOLDDATE_INVALID    = 5                        

            INTERNAL_ERROR      = 6                        

            QUEUE_ERROR         = 7                        

            RUNNING             = 8                        

            SYSTEM_LOCK_ERROR   = 9                        

            USER_INVALID        = 10                       

            OTHERS              = 11.                      

*  Continuamos si no hay eror                              

   CHECK ( SY-SUBRC = 0 ).                                 

*  Completamos la tabla BDC de la forma oportuna

LOOP AT I_ZCLIENXX.                                                 

*    Formateamos el campo fecha para salida (dd.mm.aaaa)              

*    en la pantalla no podemos indicar el formato interno (aaaammdd)   

     WRITE I_ZCLIENXX-FNACI TO L_FNACI.                                

*    Completamos la pantalla inicial con los datos del cliente actual  

     PERFORM BDC_INSERT USING:                                         

         'X' 'SAPMZBXX'         '9000',                                

         ' ' 'ZCLIENXX-BUKRS'   I_ZCLIENXX-BUKRS,                      

         ' ' 'ZCLIENXX-NCLIE'   I_ZCLIENXX-NCLIE,                      

         ' ' 'ZCLIENXX-NOMBR'   I_ZCLIENXX-NOMBR,                      

         ' ' 'ZCLIENXX-APEL1'   I_ZCLIENXX-APEL1,                      

         ' ' 'ZCLIENXX-APEL2'   I_ZCLIENXX-APEL2,                      

         ' ' 'ZCLIENXX-FNACI'   L_FNACI,                               

         ' ' 'D_PARTICULAR'     'X',                                   

         ' ' 'BDC_OKCODE'       'ENTE',                                

*    En la pantalla inicial (con datos introducidos) damos alta      

         'X' 'SAPMZBXX'         '9000',

' ' 'BDC_OKCODE'       'ALTA'.                              

   ENDLOOP.                                                          

*  Introducimos el código de fin de transacción después del          

*  último cliente                                                    

   PERFORM BDC_INSERT USING:                                         

*  En la pantalla inicial (abandonamos la transacción).            

       'X' 'SAPMZBXX'         '9000',                                

       ' ' 'BDC_OKCODE'       '/ECANC'.                                  

*  Una vez completado la transacción , la incluimos en el J.datos    

   CALL FUNCTION 'BDC_INSERT'                                        

          EXPORTING                                                  

               TCODE            = 'ZBXX'                             

*              POST_LOCAL       = NOVBLOCAL                          

*              PRINTING         = NOPRINT                            

          TABLES                                                     

               DYNPROTAB        = I_BDCTAB                           

          EXCEPTIONS

INTERNAL_ERROR   = 1                        

               NOT_OPEN         = 2                        

               QUEUE_ERROR      = 3                        

               TCODE_INVALID    = 4                        

               PRINTING_INVALID = 5                        

               POSTING_INVALID  = 6                        

               OTHERS           = 7.                       

*  Comprobamos si hubo error, mostramos mensaje de error   

   IF ( SY-SUBRC <> 0 ).                                   

     MESSAGE E000(38) WITH SY-SUBRC.                       

   ENDIF.                                                  

*  Una vez completado el Juego de datos , lo cerramos      

   CALL FUNCTION 'BDC_CLOSE_GROUP'                         

        EXCEPTIONS                                         

             NOT_OPEN    = 1                               

             QUEUE_ERROR = 2                               

             OTHERS      = 3.

ENDFORM.                    " CREAR_JUEGO_DATOS                         

*&---------------------------------------------------------------------*

*&      Form  BDC_INSERT                                                

*&---------------------------------------------------------------------*

*      Inserta en la tabla interna global I_BDCTAB los campos recibido *

* Distinguiendo si se trata de un registro de dynpro o de un registro  *

* de campo.                                                            *

*----------------------------------------------------------------------*

*      -->PE_DYNBEGIN    Flag de pantalla/ campo                       *

*      -->PE_FNAM        Nombre de programa / campo                    *

*      -->PE_FVAL        Valor pantalla / campo                        *

*----------------------------------------------------------------------*

FORM BDC_INSERT USING VALUE(PE_DYNBEGIN)                                

                      VALUE(PE_FNAM)                                    

                      VALUE(PE_FVAL).                                   

* Proceso                                                              *

************************************************************************

* Incializamos la cabecera de la tabla                                  

  CLEAR I_BDCTAB.                                                       

* Informamos el flag de pantalla                                        

  I_BDCTAB-DYNBEGIN = PE_DYNBEGIN.                                      

* En función si es o no pantalla                                        

  IF ( PE_DYNBEGIN = 'X' ).                                             

    I_BDCTAB-PROGRAM = PE_FNAM.                                         

    I_BDCTAB-DYNPRO = PE_FVAL.                                          

* Si se trata de un campo                                               

  ELSE.                                                                 

    I_BDCTAB-FNAM = PE_FNAM.                                            

    I_BDCTAB-FVAL = PE_FVAL.                                            

  ENDIF.                                                                

* Insertamos el registro en la tabla interna                            

  APPEND I_BDCTAB.                                                      

ENDFORM.                    " BDC_INSERT

1.3.3 Tratamiento de un juego de datos.
Ya hemos generado un juego de datos , que estará almacenado en algún lugar del sistema, vamos a ver ahora como visualizan, procesan, borran … 

En el menú Sistema (Existirá en todos los menús) seguiremos la ruta:

Servicios(Batch Input ( Carpetas (SM35).

Llegaremos a la siguiente pantalla:

Donde podremos seleccionar los juegos de datos , por nombre y  fecha de creación  así como por su Status de proceso....

Para ver nuestro juego de datos pondremos el nombre dado (ZCLIENXX).

Aparecerá nuestro juego de datos, en  esta pantalla , podremos realizar las siguientes funciones:

Dentro del menú ‘Juego de datos’.

· Ejecutar Juego de datos (F8).

Nos permite ejecutar el juego de datos ofreciendo varias posibilidades de ejecución a través de la ventana:

Ejecutar visible:
Se visualizan las pantallas por las que va pasando proceso deteniéndose en cada una de ellas hasta que se pulse enter.

Visualizar sólo errores
Únicamente se visualizará la secuencia de pantallas cuando se encuentre un error .

Ejecutar Invisible:
Tendrá el mismo efecto que la opción ‘Procesar en Fondo’

· Procesar en Fondo (F5).

Ejecuta el juego de datos en un proceso en fondo (Batch). Para ello genera un JOB con el mismo nombre del juego de datos.(Podremos indicar el nombre de la máquina en la que queremos que se ejecute).

· Borrar (Shift + F2). [image: image11.png]



Borra el juego de datos.

Dentro del menú ‘Pasar a’ (o botones):

· Análisis juego datos. [image: image12.png]



Permite visualizar el contenido, pudiendo visualizar el nombre de las pantallas así como su contenido.

· Estadísticas. [image: image13.png][ Estadisticas.




Una vez ejecutado el juego de datos , muestra un informe con las transacciones correctas , erróneas …

· Log. [image: image14.png]Blog




Muestra un informe con el log de la ejecución , se crearán tantos logs como ejecuciones del juego de datos se realicen.

Ejecutaremos el juego de datos. A continuación comprobamos que las entradas se han dado de alta en la tabla.

1.4 Call transaction.

La filosofía del Call Transaction  es idéntica a la del juego de datos , se trata de crear un conjunto de registros de pantallas y de datos para una transacción determinada  con la misma estructura que para el juego de datos que se ejecuta de forma inmediata.

Las principales diferencias son:

· No se genera un juego de datos que se almacena en forma de fichero especial.

· La ejecución es inmediata.

· No se pueden incluir datos para dos transacciones distintas.

Por tanto para introducir datos mediante  Call Transaction solamente necesitaremos completar la tabla de registros I_BDCTAB del mismo modo que en un Batch Input, una vez completada llamaremos a su ejecución mediante la instrucción:

CALL TRANSACTION tcod.

Tiene las siguientes opciones:

· .. AND SKIP FIRST SCREEN.

Permite saltar la primera pantalla de la transacción , esto será útil para determinadas transacciones que la pantalla inicial sea informativa o no se recojan datos.

· ..USING itab

Permite indicar la tabla interna que contendrá los registros con la estructura BDCDATA. 

·  ... MODE mode


Permite señalar el modo de llamada los valores posibles son:


- ‘A’
Visualizando las pantallas.


- ‘E’
Visualizar pantallas sólo si hay error.


-‘N’
No visualizar pantallas.

·  ... UPDATE upd            

Permite señalar la manera en que se actualizarán los datos en las tablas  los valores posibles son:

‘A’
Actualización asíncrona.

‘S’
Actualización síncrona.

·  ... MESSAGES INTO messtab 

Permite recoger en la tabla interna ‘messtab’ los mensajes que se procuzcan durante la ejecución de la transacción.

Vamos a modificar nuestro programa anterior ‘ZINTE2XX’ para  que de la posibilidad de introducir los datos leídos del fichero bien  generando un juego de datos o bien ejecutando  Call transaction.

Para ello realizaremos las siguientes modificaciones:

En primer lugar , definiremos dos casillas de selección que den la posibilidad de seleccionar un método u otro. (Si se selecciona juego de datos , el parámetro que indica el nombre del juego de datos a crear será obligatorio).

El código resultante será algo así:

************************************************************************

* PROGRAMA: ZINTE2XX                                                   *

* DESCRIPCION: Con los datos de los clientes leídos de un  fichero     *

*     da la posibilidad de crear un juego de datos o cargar            *

*     directamente los datos con Call transaction.                     *

* AUTOR : MAD00                     FECHA: 28/08/2001                  *

*----------------------------------------------------------------------*

* CONTROL DE MODIFICACIONES                                            *

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         *

************************************************************************

REPORT ZINTE2XX NO STANDARD PAGE HEADING.

************************************************************************

* Tablas del diccionario de datos                                      *

************************************************************************

TABLES: ZCLIENXX.                   " Maestro de clientes.

************************************************************************

* Definición de tablas internas                                        *

************************************************************************

* Tabla interna para almacenar los registros del fichero

DATA: BEGIN OF I_ZCLIENXX OCCURS 0,

       BUKRS LIKE ZCLIENXX-BUKRS,      "Sociedad

       NCLIE LIKE ZCLIENXX-NCLIE,      "Número de cliente

       NOMBR LIKE ZCLIENXX-NOMBR,      "Nombre cliente

       APEL1 LIKE ZCLIENXX-APEL1,      "Primer apellido

       APEL2 LIKE ZCLIENXX-APEL2,      "Segundo apellido

       FNACI LIKE ZCLIENXX-FNACI,      "Fecha de nacimiento

     END OF I_ZCLIENXX.

* Tabla interna para almacenar los registros del juego de datos

 DATA: BEGIN OF I_BDCTAB OCCURS 0.

        INCLUDE STRUCTURE BDCDATA.

 DATA: END OF I_BDCTAB.

************************************************************************

* Definición de variables globales                                     *

************************************************************************

 DATA :D_MERROR(100) TYPE C.         " Mensaje de error

************************************************************************

* Pantalla de selección                                                *

************************************************************************

* Parámetro nombre de fichero

SELECTION-SCREEN BEGIN OF BLOCK BLOQ1 WITH FRAME TITLE TEXT-001.

* Parámetros

PARAMETERS:

* Nombre del fichero (activamos minúsculas permitidas).

       P_NFICH(50) TYPE C OBLIGATORY DEFAULT '/usr/sap/tmp/zclienxx.dat'

                        LOWER CASE.

SELECTION-SCREEN END OF BLOCK BLOQ1.

SELECTION-SCREEN BEGIN OF BLOCK BLOQ2 WITH FRAME TITLE TEXT-002.

PARAMETERS:

*           Flag de crear juego de datos

            P_JDFL  RADIOBUTTON GROUP 001,

*           Nombre del juego de datos.

            P_JDATOS(12) TYPE C DEFAULT 'ZCLIENXX'.

SELECTION-SCREEN SKIP 1.

PARAMETERS

*           Flag de call transaction.

            P_CTFL RADIOBUTTON GROUP  001.

SELECTION-SCREEN END OF BLOCK BLOQ2.

************************************************************************

* Comienzo de selección                                                *

************************************************************************

START-OF-SELECTION.

*  Abrimos el fichero para lectura en modo texto.

   OPEN DATASET P_NFICH FOR INPUT IN TEXT MODE MESSAGE D_MERROR.

*  Comprobamos si no hubo error

   IF ( SY-SUBRC = 0 ).

*    Inicializamos la tabla interna

     REFRESH I_ZCLIENXX.

     CLEAR I_ZCLIENXX.

*    Leemos el primer registro del fichero sobre la tabla interna

     READ DATASET P_NFICH INTO I_ZCLIENXX.

*    Mientras no haya error de lectura

     WHILE ( SY-SUBRC = 0 ).

*      Añadimos el registro leído a la tabla interna

       APPEND I_ZCLIENXX.

*      Inicializamos la cabecera

       CLEAR I_ZCLIENXX.

*      Leemos el siguiente registro

       READ DATASET P_NFICH INTO I_ZCLIENXX.

     ENDWHILE.

*    Comprobamos que se han almacenado datos en la tabla interna

     CHECK NOT ( I_ZCLIENXX[] IS INITIAL ).

*    En función de la opción seleccionada (Juego datos o call).

     IF NOT ( P_JDFL IS INITIAL ).

*      Creamos el juego de datos con el contenido de la tabla I_ZCLIENXX

       PERFORM CREAR_JUEGO_DATOS.

*      Mostramos mensaje de 'Juego de datos creado'.

       MESSAGE I000(38) WITH TEXT-003 P_JDATOS.

*    En caso contrario se ha seleccionado Call

     ELSE.

*      Completamos la tabla interna I_BDCTAB

       PERFORM COMPLETAR_BDCTAB.

*      Llamamos a la transacción en modo sincrono y no visualización

       CALL TRANSACTION 'ZBXX' USING I_BDCTAB MODE 'N' UPDATE 'S'.

*      Comprobamos el valor de retorno

       IF ( SY-SUBRC = 0 ).

*        Mostramos mensaje de datos almacenados

         MESSAGE I000(38) WITH TEXT-004.

       ENDIF.

     ENDIF.

*  Si hay error , mostramos el mensaje, con un mensaje de error

   ELSE.

      MESSAGE E000(38) WITH D_MERROR.

   ENDIF.

************************************************************************

* Control de la pantalla de selección.                                 *

************************************************************************

* Control de la salida de la pantalla de selección

AT SELECTION-SCREEN OUTPUT.

* Si esta seleccionada la opción de juego de datos

* hacemos que la entrada para el parámetro P_JDATOS sea obligatoria.

  IF NOT ( P_JDFL IS INITIAL ).

*  Recorremos todos los elementos de la pantalla.

   LOOP AT SCREEN.

*   Al parámetro P_JDATOS le hacemos obligatorio

    CHECK SCREEN-NAME = 'P_JDATOS'.

    SCREEN-REQUIRED = '1'.

    MODIFY SCREEN.

   ENDLOOP.

  ENDIF.

************************************************************************

* Rutinas adicionales.                                                 *

************************************************************************

*&---------------------------------------------------------------------*

*&      Form  CREAR_JUEGO_DATOS

*&---------------------------------------------------------------------*

*       Crea un juego de datos con el nombre indicado por el parámetro *

*    P_JDATOS, para la transacción ZBXX con el  contenido de la        *

*    tabla interna I_ZCLIENXX.                                         *

*----------------------------------------------------------------------*

FORM CREAR_JUEGO_DATOS.

************************************************************************

* Proceso                                                              *

************************************************************************

* Abrimos el juego de datos.

  CALL FUNCTION 'BDC_OPEN_GROUP'

       EXPORTING

*           CLIENT              = SY-MANDT

*           DEST                = FILLER8

            GROUP               = P_JDATOS

*           HOLDDATE            = FILLER8

            KEEP                = ' '

            USER                = SY-UNAME

*           RECORD              = FILLER1

*      IMPORTING

*           QID                 =

       EXCEPTIONS

            CLIENT_INVALID      = 1

            DESTINATION_INVALID = 2

            GROUP_INVALID       = 3

            GROUP_IS_LOCKED     = 4

            HOLDDATE_INVALID    = 5

            INTERNAL_ERROR      = 6

            QUEUE_ERROR         = 7

            RUNNING             = 8

            SYSTEM_LOCK_ERROR   = 9

            USER_INVALID        = 10

            OTHERS              = 11.

*  Continuamos si no hay eror

   CHECK ( SY-SUBRC = 0 ).

*  Completamos la tabla BDC de la forma oportuna

   PERFORM COMPLETAR_BDCTAB.

*  Una vez completado la transacción , la incluimos en el J.datos

   CALL FUNCTION 'BDC_INSERT'

          EXPORTING

               TCODE            = 'ZBXX'

*              POST_LOCAL       = NOVBLOCAL

*              PRINTING         = NOPRINT

          TABLES

               DYNPROTAB        = I_BDCTAB

          EXCEPTIONS

               INTERNAL_ERROR   = 1

               NOT_OPEN         = 2

               QUEUE_ERROR      = 3

               TCODE_INVALID    = 4

               PRINTING_INVALID = 5

               POSTING_INVALID  = 6

               OTHERS           = 7.

*  Comprobamos si hubo error, mostramos mensaje de error

   IF ( SY-SUBRC <> 0 ).

     MESSAGE E000(38) WITH SY-SUBRC.

   ENDIF.

*  Una vez completado el Juego de datos , lo cerramos

   CALL FUNCTION 'BDC_CLOSE_GROUP'

        EXCEPTIONS

             NOT_OPEN    = 1

             QUEUE_ERROR = 2

             OTHERS      = 3.

ENDFORM.                    " CREAR_JUEGO_DATOS

*&---------------------------------------------------------------------*

*&      Form  COMPLETAR_BDCTAB

*&---------------------------------------------------------------------*

*      Completa la tabla interna I_BDCTAB con los datos necesarios     *

* para llamar a la transacción ZBXX existentes en la tabla interna     *

* I_ZCLIENXX.                                                          *

*----------------------------------------------------------------------*

FORM COMPLETAR_BDCTAB.

************************************************************************

* Definición de variables locales                                      *

************************************************************************

DATA: L_FNACI(10) TYPE C.          " Formateo de fecha para pantalla

************************************************************************

* Proceso                                                              *

************************************************************************

*  Inicializamos la tabla interna  BDC    

   REFRESH I_BDCTAB.

*  Completamos la tabla BDC de la forma oportuna

   LOOP AT I_ZCLIENXX.

*    Formateamos el campo fecha para salida (dd.mm.aaaa)

*    en la pantalla no podemos indicar el formato interno (aaaammdd)

     WRITE I_ZCLIENXX-FNACI TO L_FNACI.

*    Completamos la pantalla inicial con los datos del cliente actual

     PERFORM BDC_INSERT USING:

         'X' 'SAPMZBXX'         '9000',

         ' ' 'ZCLIENXX-BUKRS'   I_ZCLIENXX-BUKRS,

         ' ' 'ZCLIENXX-NCLIE'   I_ZCLIENXX-NCLIE,

         ' ' 'ZCLIENXX-NOMBR'   I_ZCLIENXX-NOMBR,

         ' ' 'ZCLIENXX-APEL1'   I_ZCLIENXX-APEL1,

         ' ' 'ZCLIENXX-APEL2'   I_ZCLIENXX-APEL2,

         ' ' 'ZCLIENXX-FNACI'   L_FNACI,

         ' ' 'D_PARTICULAR'     'X',

         ' ' 'BDC_OKCODE'       'ENTE',

*    En la pantalla inicial (con datos introducidos) damos alta

         'X' 'SAPMZBXX'         '9000',

         ' ' 'BDC_OKCODE'       'ALTA'.

   ENDLOOP.

*  Introducimos el código de fin de transacción después del

*  último cliente

   PERFORM BDC_INSERT USING:

*  En la pantalla inicial (abandonamos la transacción).

       'X' 'SAPMZBXX'         '9000',

       ' ' 'BDC_OKCODE'       '/BACK'.

ENDFORM.                    " COMPLETAR_BDCTAB

*&---------------------------------------------------------------------*

*&      Form  BDC_INSERT

*&---------------------------------------------------------------------*

*      Inserta en la tabla interna global I_BDCTAB los campos recibido *

* Distinguiendo si se trata de un registro de dynpro o de un registro  *

* de campo.                                                            *

*----------------------------------------------------------------------*

*      -->PE_DYNBEGIN    Flag de pantalla/ campo                       *

*      -->PE_FNAM        Nombre de programa / campo                    *

*      -->PE_FVAL        Valor pantalla / campo                        *

*----------------------------------------------------------------------*

FORM BDC_INSERT USING VALUE(PE_DYNBEGIN)

                      VALUE(PE_FNAM)

                      VALUE(PE_FVAL).

************************************************************************

* Proceso                                                              *

************************************************************************

* Incializamos la cabecera de la tabla

  CLEAR I_BDCTAB.

* Informamos el flag de pantalla

  I_BDCTAB-DYNBEGIN = PE_DYNBEGIN.

* En función si es o no pantalla

  IF ( PE_DYNBEGIN = 'X' ).

    I_BDCTAB-PROGRAM = PE_FNAM.

    I_BDCTAB-DYNPRO = PE_FVAL.

* Si se trata de un campo

  ELSE.

    I_BDCTAB-FNAM = PE_FNAM.

    I_BDCTAB-FVAL = PE_FVAL.

  ENDIF.

* Insertamos el registro en la tabla interna

  APPEND I_BDCTAB.

ENDFORM.                    " BDC_INSERT

1.5 Direct Input.

Del mismo modo que la carga de datos en el sistema a través de Batch input o Call transaction utilizan la interfaz de pantallas , el sistema Direct Input utiliza una interfaz especial propia. Esta interfaz especial consiste en un conjunto de programas que actualizan las bases de datos sin necesidad de que los datos se introduzcan por pantalla, estos programas toman los datos , normalmente,  de un fichero con la  estructura esperada por el programa.

La carga de datos mediante programas de carga ‘Direct Input’ es mucho más rápida que la carga de dato mediante Batch Input o Call Transaction. Esto es debido, entre otras razonas  , a que durante la carga de los datos , aunque se realizan los mismos chequeos de autorizaciones, integridad … , se evita la lógica de las pantallas .

Existen programas de carga Direct Input predefinidos para determinados módulos de SAP.  La transacción ‘SXDA_OLD’ permite ver estos programas.

1.6 Ficheros locales.
Hasta ahora , siempre hemos trabajado con ficheros que se encontraban en servidor de aplicación SAP. Vamos a ver como trabajar con ficheros locales , entendiendo como tales aquellos que se encuentran dentro del terminal. (Bien en una unidad de disco, disco de red …).

Existen dos funciones estándares que nos permiten importar y exportar datos a ficheros locales:

· WS_DOWNLOAD

Para volcar a fichero.

· WS_UPLOAD

Para leer de fichero.

Vamos a ver con un ejemplo como utilizar estas funciones.

El ejemplo va a consistir en un programa que vuelque el contenido de la tabla ZCLIENXX a un fichero local .

El código de programa será algo así:

************************************************************************

* PROGRAMA: ZINTE3XX                                                   *

* DESCRIPCION: Baja los datos de la tabla ZCLIENXX a un fichero local  *

*                                                                      *

* AUTOR : MAD00                     FECHA: 28/08/2001                  *

*----------------------------------------------------------------------*

* CONTROL DE MODIFICACIONES                                            *

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         *

************************************************************************

REPORT ZINTE3X NO STANDARD PAGE HEADING.

************************************************************************

* Tablas del diccionario de datos                                      *

************************************************************************

TABLES: ZCLIENXX.                   " Maestro de clientes.

************************************************************************

* Definición de tablas internas                                        *

************************************************************************

* Tabla interna para almacenar los registros del fichero

DATA: BEGIN OF I_ZCLIENXX OCCURS 0,

       BUKRS LIKE ZCLIENXX-BUKRS,      "Sociedad

       NCLIE LIKE ZCLIENXX-NCLIE,      "Número de cliente

       NOMBR LIKE ZCLIENXX-NOMBR,      "Nombre cliente

       APEL1 LIKE ZCLIENXX-APEL1,      "Primer apellido

       APEL2 LIKE ZCLIENXX-APEL2,      "Segundo apellido

       FNACI LIKE ZCLIENXX-FNACI,      "Fecha de nacimiento

     END OF I_ZCLIENXX.

************************************************************************

* Pantalla de selección                                                *

************************************************************************

* Parámetro nombre de fichero

SELECTION-SCREEN BEGIN OF BLOCK BLOQ1 WITH FRAME TITLE TEXT-001.

* Parámetros

PARAMETERS:

* Sociedad

       P_BUKRS LIKE ZCLIENXX-BUKRS OBLIGATORY DEFAULT '0001',

* Nombre del fichero local (activamos minúsculas permitidas).

       P_NFICH(128) TYPE C OBLIGATORY DEFAULT 'C:\zclienxx.dat'

                          LOWER CASE.

SELECTION-SCREEN END OF BLOCK BLOQ1.

************************************************************************

* Comienzo de selección                                                *

************************************************************************

START-OF-SELECTION.

*  Seleccionamos los datos de la tabla

   SELECT *

   FROM ZCLIENXX INTO CORRESPONDING FIELDS OF TABLE I_ZCLIENXX

   WHERE BUKRS = P_BUKRS.

*  Comprobamos que se seleccionarón datos

   CHECK ( SY-SUBRC = 0 ).

*  Volcamos el contenido de la tabla interna al fichero local

   CALL FUNCTION 'WS_DOWNLOAD'

        EXPORTING

*            BIN_FILESIZE        = ' '

*            CODEPAGE            = ' '

             FILENAME            = P_NFICH

             FILETYPE            = 'ASC'

*            MODE                = ' '

*            WK1_N_FORMAT        = ' '

*            WK1_N_SIZE          = ' '

*            WK1_T_FORMAT        = ' '

*            WK1_T_SIZE          = ' '

*            COL_SELECT          = ' '

*            COL_SELECTMASK      = ' '

*            NO_AUTH_CHECK       = ' '

*       IMPORTING

*            FILELENGTH          =

        TABLES

             DATA_TAB            = I_ZCLIENXX

*            FIELDNAMES          =

        EXCEPTIONS

             FILE_OPEN_ERROR     = 1

             FILE_WRITE_ERROR    = 2

             INVALID_FILESIZE    = 3

             INVALID_TABLE_WIDTH = 4

             INVALID_TYPE        = 5

             NO_BATCH            = 6

             UNKNOWN_ERROR       = 7

             OTHERS              = 8.

*   Mostramos mensaje de fichero creado

    MESSAGE I000(38) WITH TEXT-001.

