1. Programación de listados (REPORTS) 

1.1 Introducción.

En el proceso de tratamiento de la información es necesario poder consultar los datos almacenados en el sistema de una forma organizada, para realizar este tipo de operaciones se utilizan programas tipo REPORT, con estos programas podemos generar listados en los que mostrar dicha información. Estos listados pueden estar orientados a visualización en pantalla y/o a impresora.

Podríamos distinguir dos tipos de listados por su grado de interacción con el usuario: 

· Reports planos: Orientados a visualizar información de una manera plana en único nivel. En este tipo de listados las opciones de usuario sobre el tratamiento de la información es nula.
· Reports interactivos: Orientados a presentar información en varios niveles, es decir,  pudiendo acceder desde un nivel de información a otro nivel de más detalle y así sucesivamente. Las opciones de usuario para el  tratamiento de la información son amplias ya que se puede ir navegando a través de los distintos niveles de detalle.
Por otra parte se da la necesidad de actualizar las bases de datos en las que no es necesaria la intervención directa del usuario, son los procesos que conocemos como ‘procesos batch’. Estos programas también se realizan mediante programas tipo REPORT pero lógicamente sin la parte de presentación de datos por pantalla.

Por tanto con programas tipo REPORT podemos realizar listados para la visualización de información así como programas de actualización masiva de datos. Estos programas se realizan en un lenguaje de programación propio de SAP, es el lenguaje ABAP/4.

1.2 Editor ABAP/4

Para poder desarrollar los programas tenemos un editor de lenguaje ABAP/4. Este editor nos permitirá escribir, verificar y generar los programas. Hay varias maneras de llegar al editor, el camino más  habitual  para crear un nuevo programa del tipo REPORT es  (desde el menú principal de SAP)  Herramientas( Workbench ABAP(Desarrollo( Editor ABAP (SE38).
Llegaremos a la pantalla principal del editor de programas:

Desde esta pantalla podremos crear, visualizar, modificar y ejecutar (también podremos ejecutar en DEBUG mediante la opción [image: image2.png]Debugging



, ya veremos su significado más adelante) los programas.
Para crear un programa, escribieremos el nombre del programa, seleccionaremos la opción ‘Texto fuente’ y seleccionaremos la opción de crear [image: image3.png]


. Nos aparecerá la siguiente pantalla donde se definen los atributos del programa.

Título:
Descripción de la funcionalidad del programa. (En nuestro caso pondremos ‘Listado de maestro de clientes’).
Tipo:
Determina el tipo de programa, tenemos los siguientes:


Programa ON-LINE (1): Programa ejecutable tanto de forma  ON-LINE y en fondo (BATCH)

Programa Include (I): Programa que contiene código pero que no puede ser ejecutado directamente si no que formará parte de otros programas ejecutable.

Module pool (M): Contiene código para controlar pantallas. Es necesario asociarle un código de transacción para que pueda ser ejecutado.

Grupo de funciones (F): Contiene programas denominados funciones, que no se pueden ejecutar directamente sino que tienen que ser llamados de otros programas.

Pool de Subrutinas (S): Contiene código compuesto por procedimientos, tampoco puede ser ejecutado directamente sino que puede ser llamado desde otros programas.
(En nuestro caso pondremos tipo ON-LINE ‘1’).

Aplicación:
Módulo al que pertenece el programa (FI, HHRR…) (En nuestro caso pondremos un  Aplicación Desconocida ‘*’)
Una vez completado estos campos obligatorios pulsaremos el botón de grabar [image: image5.png]o Grabar



 (F11). Nos pedirá, como en los objetos de diccionario una clase de desarrollo, grabaremos como objeto local botón  [image: image6.png]Objeto local



 a continuación pasaremos a la pantalla del editor:

Vamos a ver las funciones básicas del editor para poder escribir, verificar, generar los programas.
Traer código patrón. Tenemos la posibilidad de insertar código con un patrón predefinido, para distintos elementos del lenguaje, sentencias como el WRITE, llamadas a procedimientos, llamadas a funciones … para ello situados sobre la línea de código donde deseemos incluir esta llamada pulsaremos el botón [image: image8.png]Patrén



 nos aparecerá la siguiente ventana en la que indicararemos el nombre del objeto a insertar:

Obtención de ayuda sobre instrucciones…

Para obtener ayuda sobre una instrucción, pulsamos la tecla de ayuda ‘F1’ 

Navegar a objeto.

Esta función, permite ‘navegar’ hasta el objeto seleccionado, estos objetos pueden ser desde  variables, procedimientos, símbolos de texto … definidos dentro del propio programa hasta objetos definidos fuera del programa como pueden ser llamadas a transacciones, funciones, includes …
Generalmente si el objeto al que pretendemos ‘navegar’ no existe, nos dará la posibilidad de crearlo en ese momento.
Verificación del programa.

Para comprobar que el programa no tiene errores de sintaxis utilizaremos el botón de verificar [image: image10.png]


 (Ctrl. + F2).
Para verificar que no contiene errores de manera ampliada utilizaremos la opción de menú ‘Programa(Verificar(Verif. programa ampliada’. Nos llevará a la siguiente pantalla, pulsaremos el botón [image: image11.png]


 para continiuar con la verificación.

Para generar el programa seleccionaremos la opción ‘Programa(Generar’ y lo activaremos con la opción ‘Programa( Activar’ o [image: image13.png]


 (Ctrl. + F3).
Ejecución del programa desde en editor.

Desde el editor, podemos ejecutar un programa. Para ello seleccionamos la opción de l menú ‘Programa(Verificar’ o [image: image14.png]


 (F8).
Opción ‘DEBUG’.

Para poder depurar los programas, podemos ejecutarlos en modo ‘DEBUG’, es decir, ejecutando el programa pero pasando por las líneas del código escrito y viendo el resultado de cada una de ellas de forma individual, de esta forma podemos comprobar el efecto de cada una de las instrucciones escritas. Para poder ejecutar un programa en modo ‘DEBUG’ desde el editor, será necesario fijar un ‘Breakpoint’, es decir un punto de ruptura. El punto de ruptura hace que la ejecución del programa se detenga cuando llega a la línea de código donde se encuentra  y a partir de ese punto se puede ejecutar en modo ‘DEBUG’. Para fijar un ‘Breakpoint’ nos situamos el la línea de código deseada y  pulsamos el botón [image: image15.png]


 .
El aspecto de las pantallas del DEBUG es el  siguiente:

El icono [image: image17.png]


 indica que en esa línea de código hay un breakpoint.
El icono [image: image18.png]


 indica la líne actual de programa (próxima instrucción a ejecutarse).
En la  parte inferior (marco variables) podemos  escribir el nombre de la variable de la que se quiere conocer el valor en ese momento, en la parte de la derecha aparecerá su valor (En la pantalla tenemos el nombre de la variable SY-SUBR y su valor actual ‘0’).
Una vez dentro del ‘DEBUG’ de un programa tenemos las siguientes opciones:
Doble-click(F2)Situados sobre una variable y pulsando este botón automáticamente se escribe en el marco variables para visualizar el valor actual de la variable.
[image: image19.png]



Hace que se ejecute la línea de programa actual. Si la línea actual se trata de la llamada a un procedimiento, función … navegará hasta la primera línea ejecutable de dicho procedimiento.
[image: image20.png]



Ejecuta la línea actual de programa, teniendo en cuenta que si se trata de la llamada a un procedimiento … no navegará hasta el código del procedimiento sino que se ejecutará el procedimiento como si se tratará de una única instrucción.
[image: image21.png]



Continua con la ejecución de instrucciones hasta que se encuentre con el siguiente breakpoint. (Si no encuentra breakpoint continuará hasta el final).
[image: image22.png]



Retorna al nivel superior de anidamiento del programa, es decir, si hemos ‘navegado’ con la opción paso a paso dentro de un procedimiento,  esta opción hará que se ejecute el resto del procedimiento y regrese a la  instrucción siguiente a la llamada al procedimiento. (Si dentro del procedimiento existe algún breakpoint se parará en él).

Permite visualizar el contenido de una tabla interna.
1.2.1 Documentación y estructuración de un listado plano.

 Es fundamental que los programas estén bien estructurados y documentados, de esta manera haremos que su elaboración y el posterior mantenimiento sean mucho más sencillos. La estructura básica  de un programa tipo REPORT es la siguiente (Esta estructuración puede variar dependiendo de los estándares particulares de cada cliente):
· Documentación básica del programa.
· Sentencia REPORT.
· Definición de tablas diccionario.
· Definición de constantes.
· Definición de variables.
· Definición  de estructuras.
· Definición de tablas internas.
· Definición de rangos.
· Definición de la pantalla de selección:
· Parámetros
· Select-options
· Definición de FIELD-SYMBOLS.
· Definición de FIELD-GROUPS.
· Definición de Eventos:
· INICIALIZATION.
· START-OF-SELECTION.
· END-OF-SELECTION.
· TOP-OF-PAGE.
· END-OF-PAGE.
· AT LINE-SELECTION.
· AT USER-COMMAND.
· AT PFN.
· AT SELECTION-SECREEN.
· Includes.
· Subrutinas.
Antes de la  primera sentencia  REPORT, se deberá documentar el programa indicando:
· Nombre del  programa.
· Descripción de la funcionalidad del programa.
· Fecha y autor del programa.
Deberá también reservarse un espacio para el control de modificaciones del programa.
(En nuestro ejemplo que va a consistir en programa que muestra un listado de los clientes existentes,  con el siguiente formato)

C. Cliente  Nombre  Primer Apellido  Segundo Apellido  F.Nacimiento
X (10)
  X (20)     X (25)
         X (25)

      XX (10)

El programa dará la posibilidad de seleccionar clientes por Código de cliente, por nombre y por fecha de nacimiento de tal forma que solamente aparecerán en el listado aquellos clientes que cumplan los criterios de selección indicados.

 Pondremos como documentación en la cabecera del programa:

************************************************************************

* PROGRAMA: ZREPORXX 

* DESCRIPCION: Muestra un listado de los clientes existentes en el 

*             que se detalla en nº de cliente junto con el nombre y

*             apellidos.

* AUTOR: Nombre y Apellidos          FECHA: 13/08/2001

*----------------------------------------------------------------------*

* CONTROL DE MODIFICACIONES 

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION. 

************************************************************************

antes de la instrucción REPORT)
1.2.2 Definición de atributos de salida.

Todos los programas de este tipo empiezan con la instrucción  ‘REPORT’ de ahí su nombre. Esta instrucción además de indicar el nombre e inicio del programa, permite señalar algunos de los atributos del programa como el alto y el ancho de listado, la clase de mensajes que utiliza etc…
Nota: Para obtener ayuda sobre una instrucción, situando el cursor sobre ella pulsamos F1 nos saldrá una pantalla de ayuda con la sintaxis y opciones de cada una de ellas. Para la sentencia REPORT se muestra la ayuda:

En nuestro ejemplo añadiremos a la instrucción REPORT  las siguientes opciones:

REPORT zreporxx NO STANDARD PAGE HEADING LINE-SIZE 120 LINE-COUNT 80.

1.2.3 Definición de tablas externas.

Para utilizar tablas del diccionario de datos, es necesario hacer referencia a que tablas del diccionario se va a acceder desde el programa. Para ello será necesario indicar la palabra clave TABLES seguida del nombre o nombres de las tablas empleadas.
En nuestro ejemplo pondremos:
TABLES: zclienxx.

1.2.4 Pantalla de selección.

La ‘pantalla de selección’ es una pantalla en la que se indican los parámetros del programa. Estos parámetros van a utilizarse para ‘filtrar’ la información y procesar solamente aquella que pase el dicho ‘filtro’, es decir, cumpla con los criterios de selección.
En la pantalla de selección, podemos tener los siguientes elementos:
· Parameters: Recogen único valor, es decir, como criterio solamente se podrá indicar un único valor. Para incluir parámetros de selección podremos la palabra clave PARAMETERS seguida del nombre de parámetro y su tipo.
En nuestro ejemplo, vamos a definir un parámetro para seleccionar por Sociedad para ello escribiremos:
PARAMETERS p_bukrs LIKE zclienxx-bukrs OBLIGATORY DEFAULT '0001'.

· Select-options: Recogen múltiples valores o un rango de valores, es decir, se pueden indicar más de un valor como criterio de selección. Se definen poniendo la palabra clave SELECT-OPTIONS seguida del nombre del rango de selección y el campo al que hace referencia.
Vamos a definir tres rangos  de selección uno para el nº de cliente, otro para el nombre   y otro para la fecha de nacimiento, para ello escribiremos:
SELECT-OPTIONS:

    s_nclie FOR zclienxx-nclie,        " Nº de cliente

    s_nombr FOR zclienxx-nombr,        " Nombre cliente

    s_fnaci FOR zclienxx-fnaci.        " Fecha de nacimiento

Si en este punto grabamos, verificamos, generamos  y ejecutamos nuestro programa, únicamente tendríamos la pantalla de selección.  Que en nuestro caso tendrá el siguiente aspecto: 

Podemos hacer que la pantalla de selección tenga un aspecto más elaborado para ello tenemos la posibilidad de incluir marcos, saltos de línea…Para conseguir estos efectos utilizaremos la palabra clave SELECTION SCREEN.
En nuestro ejemplo vamos a crear dos marcos uno que va a incluir al parámetro P_BUKRS y otro que va a contener a los rangos de selección S_NCLIEN, S_NOMBR y S_FNACI. Para ello incluiremos las siguientes instrucciones, de tal forma que tendremos:

SELECTION-SCREEN BEGIN OF BLOCK bloq1 WITH FRAME TITLE text-001.

  PARAMETERS p_bukrs LIKE zclienxx-bukrs OBLIGATORY DEFAULT '0001'.
SELECTION-SCREEN END OF BLOCK bloq1.
SELECTION-SCREEN BEGIN OF BLOCK bloq2 WITH FRAME TITLE text-002.
  SELECT-OPTIONS:

    s_nclie FOR zclienxx-nclie,
     " Nº de cliente
    s_nombr FOR zclienxx-nombr,        " Nombre cliente
    s_fnaci FOR zclienxx-fnaci.        " Fecha de nacimiento
SELECTION-SCREEN END OF BLOCK bloq2.
Conseguiremos que la pantalla tenga el siguiente aspecto:

Los marcos creados, no tienen texto esto es debido a que los símbolos de texto ‘001’ y ‘002’ referenciados con la palabra clave TEXT, no están definidos.
1.2.5 Elementos de texto

Pasar a (Elementos de texto(Símbolos de texto’ (también se puede acceder desde la pantalla de acceso al editor seleccionando esta opción). Llegaremos a la siguiente pantalla:

1.2.5.1 Símbolos de texto.
Son cadenas de texto que se pueden utilizar a lo largo del programa para diferentes funciones. Para definir un símbolo  de texto seleccionaremos la opción de menú la opción ‘símbolos de texto’ donde definimos los símbolos de texto. El nº de símbolo puede estar compuesto por letras o números
Los campos LonD y LonM representan la longitud de la cadena introducida y la longitud máxima de salida del texto respectivamente, una vez introducido el texto, igualaremos el campo LonM con el valor de LonD.
En nuestro caso definiremos :
001
Datos sociedad
002
Datos cliente
Una vez introducidos los símbolos de texto grabamos [image: image28.png]


 (F11).
1.2.5.2 Textos de selección.
Son los textos que aparecerán en la pantalla de selección en lugar del nombre del parámetro. Si seleccionamos la pestaña ‘Textos selección ‘, aparecerán los parámetros y rangos de selección definidos en el programa.
En nuestro caso aparecerá la pantalla en la que completaremos los siguientes textos. Sociedad, Fecha de nacimiento, Número cliente, Nombre cliente. Una vez escritos los textos de selección grabamos [image: image29.png]


 (F11).
1.2.5.3 Títulos y cabeceras.
Finalmente si seleccionamos la pestaña  ‘ Cabeceras lista ’ tendremos

Cabecera de lista: Define el texto que aparecerá al comienzo de cada página.
Pondremos ‘Clientes seleccionados’.

Cabecera de columna: Define las líneas de cabecera de columna que aparecerán en cada una de las páginas, podemos definir hasta cuatro cabeceras de columna.
En nuestro ejemplo, pondremos ‘C. Cliente  Nombre  Primer Apellido  Segundo Apellido  F.Nacimiento’. Reservando para cada campo la longitud de salida correspondiente. Una vez completados los títulos y cabeceras grabamos [image: image31.png]


 (F11) y volvemos al editor del programa.
Las cabeceras de lista y de columna, no aparecerán en el listado, si se añade la opción NO STANDARD PAGE-HAEDING en los atributos de salida del listado (Instrucción REPORT).
(Siguiendo con el ejemplo . Si ejecutamos a continuación el programa, veremos que la pantalla de selección a tomado el aspecto siguiente::)

1.2.6 Sentencias de salida de datos

Vamos a ver un resumen de las principales instrucciones para el formateo  de datos por pantalla.
· WRITE:

Nos permite escribir el contenido de una variable declarada con DATA, el contenido de un registro de una tabla, el contenido de un FIELD-SYMBOL, un TEXT-SYMBOL o un literal en alguna posición del listado Las opciones a añadir a esta instrucción son:

... AT pl    (posición y longitud)

... options  (formato de salida)

... ofmt     (salida formateada para el campo)

... AS CHECKBOX

... AS SYMBOL

... AS ICON

... AS LINE


... AT /pl
Con el símbolo (/) podemos hacer que ejecute un salto de línea. Con ‘p’ se especifica la columna de la línea actual en la que queremos que empiece a escribir y con ‘l ’la longitud máxima de lo que vamos a escribir. 

Por ejemplo:

WRITE AT /5(15) ‘Esto es una cadena’.

Resultado:

La posición y longitud se pueden expresar directamente con constantes numéricas (en este caso la palabra clave AT se puede omitir) y también con variables.

Si la longitud de salida es demasiado corta para el contenido de lo que se va a escribir la salida se rellenará de ‘*’ si se trata de tipos P, I y F; en el resto de tipos se trucará el contenido por la derecha.
... options
Las opciones disponibles son:
· 
... NO-ZERO
· 
... NO-SIGN
· 
... DD/MM/YY
· 
... MM/DD/YY
· 
... DD/MM/YYYY
· 
... MM/DD/YYYY
· 
... DDMMYY
· 
... MMDDYY
· 
... CURRENCY w
· 
... DECIMALS d
· 
... ROUND R
· 
... UNIT u
· 
... EXPONENT e
· 
... USING EDIT MASK mask
· 
... UNDER g
· 
... NO-GAP
· 
... LEFT-JUSTIFIED
· 
.. CENTERED
· 
.. RIGHT-JUSTIFIED
Por ejemplo para dar formato de salida a un importe con la moneda correspondiente:
WRITE D_IMPORTE CURRENCY D_MONEDA.

... ofmt
La salida se escribirá con los formatos de color, intensidad, etc. Las opciones disponibles en este punto se detallarán más adelante, dentro de la instrucción FORMAT.
Por ejemplo:

DATA F.

WRITE F INVERSE COLOR 3.

Los formatos indicados únicamente aplican a la escritura de la instrucción actual.

· ... AS CHECKBOX

Hace que el campo escrito tenga la forma de  es una caja de selección. La caja saldrá marcada o no marcada en función del valor del campo. (SPACE es no marcada y ‘X ‘es marcada). Para poder hacer la caja de entrada/salida o solamente de salida añadiremos la opción INPUT ON/OFF.

Por ejemplo:
DATA D_CAJA(1) TYPE C VALUE SPACE.
WRITE CAJA AS CHECKBOX. ó WRITE CAJA AS CHECKBOX INPUT OF.
(Será de entrada/salida, valor por defecto)
WRITE CAJA AS CHECKBOX INPUT OFF. (Será sólo de salida).

· ... AS SYMBOL


Permite escribir símbolos predefinidos. Para poder acceder a ellos  será necesario incluir en nuestro programa la biblioteca de símbolos. Para ello incluiremos la sentencia  INCLUDE <SYMBOL>. (para ver los símbolos disponibles buscaremos a través de la ayuda).
Por ejemplo, la sentencia:

WRITE: SYM_RIGHT_HAND AS SYMBOL, ‘...apunta a la derecha’.

Tendría como efecto:

· ... AS ICON

Similar a la opción anterior pero con la biblioteca de iconos. Par acceder a ella  incluiremos la sentencia INCLUDE <ICON>. (Podemos ver los iconos con la ayuda).

Por ejemplo:

WRITE: ICON_GREEN_LIGHT AS ICON, ‘...Semáforo verde’.

Tendría como efecto:

· ... AS LINE

La forma de hacer líneas horizontales o verticales en un report y que éstas formen cuadrículas es escribir los caracteres SY-VLINE (línea vertical) y SY-ULINE (línea horizontal). Son equivalentes los caracteres ‘(‘y ‘-’.  La forma exacta en la que aparecen estos segmentos depende de los caracteres adyacentes.   Cuando tenemos un carácter de línea en una posición vertical y otro horizontal en la posición adyacente automáticamente se produce la unión de ambas. Si en la posición adyacente no hay otro carácter de línea ese carácter permanece inalterado.

En la mayoría de los casos esta técnica es suficiente para hacer cuadrículas, pero hay veces en las que las uniones no se producen de la forma en la que queremos. En esos casos será necesario echar mano de caracteres especiales como: LINE_TOP_LEFT_CORNER, LINE_BOTTOM_MIDDLE_CORNER que realizan esta operación. Para poder usar estos caracteres es necesario incluir INCLUDE <LINE>.
· FORMAT.
Sirve para definir los  atributos de salida de lo que se va a escribir en la pantalla. A diferencia de la opción ‘ofmt’ del la instrucción WRITE el efecto de esta instrucción comprende todas las escrituras realizadas hasta la siguiente sentencia FORMAT. (Los cambios de evento, START-OF-SELECTION, TOP-OF-PAGE… provocan que se inicialicen todas las opciones ). Las posibles opciones de la sentencia FORMAT son:

	FORMAT INTENSIFIED ON/OFF 
	Esta opción modifica el color de fondo.

	FORMAT INVERSE ON/OFF 
	Invierte el color de fondo y el texto escrito.

	FORMAT INPUT ON/OFF
	Habilita o deshabilita para entrada en los campos que se escriban a continuación.

	FORMAT HOTSPOT ON/OFF
	 Habilita o deshabilita el formato del cursor cuando se sitúe sobre las líneas o palabras que se escriban a continuación tome aspecto de mano [image: image34.png]


 . 

	FORMAT RESET 
	Inicializa todas las opciones de formato. Esta instrucción equivale a: 

FORMAT COLOR OFF INTENSIFIED OFF INVERSE OFF HOTSPOT OFF INPUT OFF.




Ejemplo:
FORMAT COLOR COL_BACKGROUND ON.


WRITE F.


“ Se escribirá en color BACK..


WRITE D.


“ Se escribirá en color BACK..

FORMAT COLOR COL_BACKGROUND OFF.


WRITE F.


“ Se escribirá en color normal

· ULINE
Escribe una línea horizontal en la línea actual. Por defecto el tamaño de la línea comprende todo el ancho del listado. Podemos seleccionar el ámbito de la línea con añadiendo AT pl donde:
	
	p
	Se corresponde con la columna en la que comienza la línea.

	
	l
	Se corresponde con la longitud que tiene la línea.


Las siguientes instrucciones producen el mismo resultado:
-ULINE AT 3(10).
-WRITE AT 3(10) SY-ULINE.
· SKIP y POSITION
Estas dos instrucciones SKIP y POSITION.  Permiten situar el puntero de escritura en una posición determinada. 
POSITION n
Hace que el puntero se sitúe en la columna n.
SKIP n

Hace que el puntero salte n líneas.
SKIP TO LINE n
Hace que el puntero se mueva a la línea n del listado.
Las siguientes instrucciones producen un resultado equivalente a  la instrucción   (WRITE AT /5(15) ‘Esto es una cadena’.)
DATA CADENA TYPE C VALUE ‘Esto es una cadena’.

SKIP.

POSITION 5.

WRITE (15) CADENA.

· NEW-LINE
Provoca una nueva línea en el listado.
Mediante la opción  NO-SCROLLING podemos hacer que la línea permanezca fija durante el desplazamiento (Scroll) horizontal. (El valor por defecto es que se pueda desplazar por tanto la opción SCROLLING se suele omitir).
· NEW-PAGE
Provoca una nueva página en el listado. Tiene varias opciones mediante las cuales se puede determinar el tamaño (nº filas y nº de columnas) de la nueva página, si se imprime o no el título….
· RESERVE
Sirve para reservar un cierto nº de líneas antes de un salto de página. 
RESERVE n LINES.
Si en la página actual no hay n líneas disponibles  entonces se provocará un salto de página.
De esta forma se puede asegurar que las n líneas escritas a continuación estarán en la misma página.
· MESSAGE.

Sirve para mostrar un determinado mensaje de una clase (colección) de mensajes. El mensaje puede ser mostrado en la barra de status, o en forma de ventana, dependiendo del tipo de mensaje. Existen los siguientes tipos de mensajes:
· I 
Informativos.                
· W 
De precaución 
· E 
Error                    
· A 
Terminación                 
· X 
De salida 
· S
Éxito.
Con un mensaje de tipo ’I’, se mostrará una ventana en la que visualiza el contenido del mensaje y continúa con la ejecución del programa, con ‘S’ se muestra el mensaje en la barra de status y continua el programa, con un tipo ‘E’ se muestra un mensaje y puede terminar la ejecución del programa, depende del evento en el que este situado, un tipo ‘W’ en un listado tiene el mismo significado,  un mensaje tipo ‘A’ muestra el mensaje y termina el programa, finalmente un tipo ‘X’ muestra aborta el programa provocando un ‘DUMB’.
Para indicar un tipo u otro podemos usar la sintaxis:
MESSAGE TNNN(XX) with p1 p2 p3 p4.
Donde T representa el tipo de mensaje (I, S, E…).
NNN  representa el nº de mensaje.
XX
representa la clase de mensajes. (Podemos no indicar esta clase de mensajes si se añade la opción MESSAGE-ID de la instrucción REPORT).
p1..p4 representan los parámetros del mensaje.
La definición del mensaje dentro de la clase consiste en un texto indicativo del mensaje junto a unos parámetros si los tiene:
Por ejemplo, podríamos definir un mensaje con el número 001 de la clase ZZ
‘El cliente ‘& ‘No existe.
Llamaríamos al mensaje con MESSAGE I001(ZZ) WITH D_NCLIENTE.
Para hacerlo informativo y  MESSAGE E001(ZZ) WITH D_NCLIENTE. 
Para hacerlo de tipo error.          

1.2.7 Eventos

1.2.7.1 INITIALIZATION

Se ejecuta antes de mostrarse la pantalla de selección. Normalmente se utiliza  para inicializar las variables de programa y las opciones de  la pantalla de selección.
1.2.7.2 START-OF-SELECTION

Este evento se lanza cuando se procesa la instrucción REPORT... En ese momento se empieza a ejecutar el código que se encuentra entre REPORT y la palabra START-OF-SELECTION. Inmediatamente después se procesa el bloque contenido dentro de este evento.
En este evento se sitúan las instrucciones de acceso a bases de datos una vez terminada este código se ejecuta el evento END-OF-SELECTION (Si está definido).
(Siguiendo con nuestro ejemplo, vamos a realizar la selección de los clientes que cumple los criterios definidos por los parámetros de selección. Los datos de estos clientes los vamos a almacenar en una tabla interna.
Definimos la siguiente tabla interna que situaremos en la parte correspondiente del programa.
* Tabla interna de datos a listar   

DATA: BEGIN OF I_ZCLIENXX OCCURS 0.    

       INCLUDE STRUCTURE ZCLIENXX.     

DATA: END OF I_ZCLIENXX.

Añadimos el evento START-OF-SELECTION en la parte del programa que le corresponde con el siguiente código:
START-OF-SELECTION.                                                   

*  Inicializamos la tabla interna.                                    

   REFRESH I_ZCLIENXX.                                                

   CLEAR I_ZCLIENXX.                                                  

*  Seleccionamos los datos de los clientes que cumplen criterios      

   SELECT *                                                           

   FROM ZCLIENXX INTO TABLE I_ZCLIENXX                                

   WHERE BUKRS = P_BUKRS                                              

   AND   NCLIE IN S_NCLIE                                             

   AND   NOMBR IN S_NOMBR                                             

   AND   FNACI IN S_FNACI.                                            

*  Comprobamos si no hay datos seleccionados en cuyo caso,           

*  escribiremos un texto que lo indique.                              

   IF NOT ( SY-SUBRC = 0 ).                                           

     WRITE / TEXT-003.  " No hay datos para la selección indicada     

   ENDIF.

1.2.7.3 END-OF-SELECTION.

El código asociado a este evento se procesa cuando se termina la selección de datos de tablas o bases de datos lógicas, es decir, cuando termina el evento START-OF-SELECTION. 
(En nuestro ejemplo escribiremos los datos de los clientes seleccionados, para ello añadiremos el siguiente código:
END-OF-SELECTION.                                                      

* Vamos a escribir el listado con los clientes seleccionados en la     

* Tabla interna I_ZCLIENXX                                             

  LOOP AT I_ZCLIENXX.                                                  

    WRITE: /1(10) I_ZCLIENXX-NCLIE,                                  

             11(20) I_ZCLIENXX-NOMBR,                                  

             31(25) I_ZCLIENXX-APEL1,                                  

             56(25) I_ZCLIENXX-APEL2,                                  

             81(10) I_ZCLIENXX-FNACI.                                  

  ENDLOOP.

Si llegado este punto, ejecutamos el programa, si hay datos en la tabla maestro de clientes y cumplen los criterios de selección aparecerá un listado algo así:

1.2.7.4 TOP-OF-PAGE

Se ejecuta al inicio de cada nueva página.
Se suele utilizar para dar formato a la cabecera de página, cuando no se utilizan las definidas en ‘Títulos y cabeceras’. (Se añade la opción NO STANDARD PAGE HEADING).
Se le puede añadir la opción DURING LINE-SELECTION, este evento se producirá al inicio de una nueva página de un nivel de lista secundaria. (Reports interactivos).
 Se pueden tener por tanto dos eventos, 
TOP-OF-PAGE que actúa en el nivel inicial.
TOP-OF-PAGE DURING LINE-SELECTION que actúa en los de niveles superiores.
1.2.7.5 END-OF-PAGE

Se ejecuta cuando una página finaliza bien porque no tiene más líneas bien porque se ejecuta la instrucción RESERVE n LINES. Este evento no se ejecutará si el nº de líneas de la página no está definido en la sentencia REPORT, tampoco se ejecutará se fuerza una nueva página con  NEW-PAGE.
1.2.7.6 AT SELECTION-SCREEN

Se utiliza para controlar la pantalla de selección.
Sus opciones permiten mostrar valores de ayuda en los campos de selección, mostrar/ocultar campos de selección….
(En nuestro ejemplo, vamos a dar la posibilidad de  mostrar los valores posibles para el rango de selección nombre del cliente. De esta forma aparecerá, al situarnos sobre el campo, el botón de ayuda de valores posibles [image: image36.png]


 (F4), al pulsarlo mostrará una ventana con los distintos nombres existentes en nuestra tabla.
Para ello añadiremos el siguiente código:
* Valores posibles para el rango S_Nombre valor desde.     

AT SELECTION-SCREEN ON VALUE-REQUEST FOR S_NOMBR-LOW.      

* Llamada a la rutina que muestra los valores existentes   

  PERFORM VALORES_POSIBLES_NOMBRE USING S_NOMBR-LOW.       

* Valores posibles para el rango S_Nombre valor hasta      

AT SELECTION-SCREEN ON VALUE-REQUEST FOR S_NOMBR-HIGH.     

* Llamada a la rutina que muestra los valores existentes   

  PERFORM VALORES_POSIBLES_NOMBRE USING S_NOMBR-HIGH.      

Los rangos  de selección tiene implícitos dos valores el mínimo y el máximo, para que aparezca la ayuda en los dos es necesario añadir dos eventos, uno para cada valor. La funcionalidad en ambos casos es la misma por tanto es susceptible de se introducida en un único procedimiento.
(En nuestro caso lo llamamos VALORES_POSIBLES_NOMBRE, ver el código correspondiente en el listado completo del programa). 

Para implementar este  procedimiento, nos hemos ayudado  con una función predefinida en el STANDARD. 'HELP_VALUES_GET_WITH_TABLE’. Esta función permite mostrar  datos en una ventana y seleccionar uno de ellos. Permite mostrar información de más de un campo, para ello en la tabla interna  de definición de campos añadiremos tantas entradas como campos deseemos visualizar. Para indicar los valores lo haremos de forma secuencial es decir, el la entrada uno, el valor para el campo 1 en la entrada 2 el valor para el campo 2 …así hasta completar todos los campos, a continuación otra vez la entrada para el campo 1, campo 2 y así sucesivamente para cada secuencia .
1.2.8 Ejemplo de listado.

Vamos a dar otra apariencia a nuestro listado, para ello vamos a crear una cabecera propia y no vamos a utilizar la definida en ‘Título y Cabeceras’. En primer lugar para desactivar esta cabecera, añadiremos la opción NO STANDARD PAGE HEADING a la instrucción REPORT. Por otra parte vamos a hacer que el campo correspondiente al nº de cliente permanezca fijo durante el desplazamiento horizontal por el informe. En la pantalla de selección, vamos a utilizar el match code de búsqueda de clientes creado ‘ZCXX’.
El código resultante será algo así:
************************************************************************

* PROGRAMA: ZREPORXX.                                                  *

* DESCRIPCION:  Muestra un listado de los clientes existentes en el    *

*             que se detalla en nº de cliente junto con el nombre y    *

*             apellidos                                                *

* AUTOR : MAD00                     FECHA: 13/08/2001                  *

*----------------------------------------------------------------------*

* CONTROL DE MODIFICACIONES                                            *

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         *

************************************************************************

REPORT zreporxx NO STANDARD PAGE HEADING LINE-SIZE 120 LINE-COUNT 80.
************************************************************************

* Tablas del diccionario de datos                                      *

************************************************************************

TABLES: zclienxx.                    " Maestro de clientes.
************************************************************************

* Definición de constantes                                             *

************************************************************************

CONSTANTS:
    c_pos_nclie(3)   TYPE n VALUE   1,  " Posición nº cliente
    c_pos_nombr(3)   TYPE n VALUE  12,  " Posición nombre
    c_pos_apel1(3)   TYPE n VALUE  33,  " Posición apellido1
    c_pos_apel2(3)   TYPE n VALUE  59,  " Posición apellido2
    c_pos_fnaci(3)   TYPE n VALUE  85,  " Posición F. Nacimiento
    c_pos_final(3)   TYPE n VALUE  96,  " Posición final
    c_ancho_total(3) TYPE n VALUE  96,  " Ancho total del informe,
    c_pos_titul(3)   TYPE n VALUE   1,  " Posición título,
    c_pos_usuar(3)   TYPE n VALUE  37,  " Posición usuario
    c_pos_fecha(3)   TYPE n VALUE  60,  " Posición fecha
    c_pos_pagno(3)   TYPE n VALUE  85.  " Posición nº de página.
************************************************************************

* Tablas internas                                                      *

************************************************************************

* Tabla interna de datos a listar

DATA: BEGIN OF i_zclienxx OCCURS 0.
        INCLUDE STRUCTURE zclienxx.
DATA: END OF i_zclienxx.
************************************************************************

* Pantalla de selección                                                *

************************************************************************

* Parámetros

SELECTION-SCREEN BEGIN OF BLOCK bloq1 WITH FRAME TITLE text-001.
* Sociedad

PARAMETERS p_bukrs LIKE zclienxx-bukrs OBLIGATORY  DEFAULT '0001'.
SELECTION-SCREEN END OF BLOCK bloq1.
* Rangos de selección

SELECTION-SCREEN BEGIN OF BLOCK bloq2 WITH FRAME TITLE text-002.
SELECT-OPTIONS:
* Añadimos el match-code para el código de cliente

    s_nclie FOR zclienxx-nclie MATCHCODE OBJECT zcxx,  " Nº de cliente
    s_nombr FOR zclienxx-nombr,        " Nombre cliente
    s_fnaci FOR zclienxx-fnaci.        " Fecha de nacimiento
SELECTION-SCREEN END OF BLOCK bloq2.
************************************************************************

* Comienzo de selección                                                *

************************************************************************

START-OF-SELECTION.
*  Inicializamos la tabla interna.

  REFRESH i_zclienxx.
  CLEAR i_zclienxx.
*  Seleccionamos los datos de los clientes que cumplen criterios

  SELECT *
  FROM zclienxx INTO TABLE i_zclienxx
  WHERE bukrs = p_bukrs
  AND   nclie IN s_nclie
  AND   nombr IN s_nombr
  AND   fnaci IN s_fnaci.
*  Comprobamos si no hay datos seleccionados en cuyo caso,

*  escribiremos un texto que lo indique.

  IF NOT ( sy-subrc = 0 ).
    WRITE / text-003.  " No hay datos para la selección indicada
  ENDIF.
************************************************************************

* Final de la selección                                                *

************************************************************************

END-OF-SELECTION.
* Vamos a escribir el listado con los clientes seleccionados en la

* tabla interna I_ZCLIENXX

  LOOP AT i_zclienxx.
*   Escribimos el nº de cliente en color clave, el resto en otro color

    FORMAT COLOR COL_KEY.
    WRITE:  AT /c_pos_nclie '|' NO-GAP, i_zclienxx-nclie.
    FORMAT COLOR COL_NORMAL.
    WRITE:  AT  c_pos_nombr '|' NO-GAP, i_zclienxx-nombr,
            AT  c_pos_apel1 '|' NO-GAP, i_zclienxx-apel1,
            AT  c_pos_apel2 '|' NO-GAP, i_zclienxx-apel2,
            AT  c_pos_fnaci '|' NO-GAP, i_zclienxx-fnaci,
            AT  c_pos_final '|' NO-GAP.
  ENDLOOP.
*   Si se han seleccionado datos, escribimos la línea final.

  IF ( sy-subrc = 0 ).
    ULINE AT /c_pos_nclie(c_ancho_total).
  ENDIF.
************************************************************************

* Cabecera de página                                                   *

************************************************************************

TOP-OF-PAGE.
* Hacemos que la línea permanezca fija en el scroll horizontal

  NEW-LINE NO-SCROLLING.
* Activamos el color de cabecera

  FORMAT COLOR COL_HEADING INTENSIFIED OFF.
* Escribimos información de cabecera resaltando los valores

* Título y sociedad

  WRITE: AT c_pos_titul text-009.
  WRITE p_bukrs INTENSIFIED ON.
* Usuario de creación del informe

  WRITE AT c_pos_usuar text-010.
  WRITE sy-uname INTENSIFIED ON.
* Fecha de creación

  WRITE AT c_pos_fecha text-011.
  WRITE sy-datum INTENSIFIED ON USING EDIT MASK '__/__/____'.
* Nº de página.

  WRITE AT c_pos_pagno text-012.
  WRITE sy-pagno INTENSIFIED ON.
* Fecha de creación

  WRITE AT c_pos_fecha text-011.
  WRITE sy-datum INTENSIFIED ON.
* Nº de página.

  WRITE AT c_pos_pagno text-012.
  WRITE sy-pagno INTENSIFIED ON.
* Escribimos una línea horizontal

  ULINE AT /c_pos_nclie(c_ancho_total).
* Activamos el color de cabeceras intensificado

  FORMAT COLOR COL_HEADING INTENSIFIED ON.
* Escribimos la cabecera de la columnas separadas por el carácter '|'.

* Hacemos que la columna de nº de cliente sea fija en el scroll horiz.

  WRITE: AT /c_pos_nclie '|' NO-GAP, text-004.
* Hasta la columna hacemos fijo

  SET LEFT SCROLL-BOUNDARY.
* Escribimos el detalle de cabecera  de cliente

  WRITE: AT /c_pos_nclie '|' NO-GAP, text-004,
         AT  c_pos_nombr '|' NO-GAP, text-005,
         AT  c_pos_apel1 '|' NO-GAP, text-006,
         AT  c_pos_apel2 '|' NO-GAP, text-007,
         AT  c_pos_fnaci '|' NO-GAP, text-008,
         AT  c_pos_final '|' NO-GAP.
* Escribimos una línea horizontal

  ULINE AT /c_pos_nclie(c_ancho_total).
************************************************************************

* Control de la pantalla de selección.                                 *

************************************************************************

* Valores posibles para el rango S_Nombre valor desde.

AT SELECTION-SCREEN ON VALUE-REQUEST FOR s_nombr-low.
* Llamada a la rutina que muestra los valores existentes

  PERFORM valores_posibles_nombre USING s_nombr-low.
* Valores posibles para el rango S_Nombre valor hasta

AT SELECTION-SCREEN ON VALUE-REQUEST FOR s_nombr-high.
* Llamada a la rutina que muestra los valores existentes

  PERFORM valores_posibles_nombre USING s_nombr-high.
************************************************************************

* Rutinas adicionales.                                                 *

************************************************************************

*&---------------------------------------------------------------------*

*&      Form  VALORES_POSIBLES_NOMBRE

*&---------------------------------------------------------------------*

*       Muestra una ventana con los distintos nombres existentes en    *

* la tabla maestro de clientes ZCLIENXX. Dando la posibilidad de       *

* seleccionar uno de ellos. Devuelve en el parámetro PS_NOMBR  el      *

* nombre seleccionado o vacío si no se realiza selección.              *

*----------------------------------------------------------------------*

*      -->PS_NOMBR    Nombre seleccionado                              *

*----------------------------------------------------------------------*

FORM valores_posibles_nombre USING  ps_nombr  LIKE zclienxx-nombr.
************************************************************************

* Variables locales                                                    *

************************************************************************

* Tabla interna para almacenar los campos y atributos a visualizar.

  DATA : BEGIN OF i_fields OCCURS 1.
          INCLUDE STRUCTURE help_value.
  DATA:  END OF i_fields.
* Tabla interna para almacenar el valor de campos a mostrar.

  DATA : BEGIN OF i_values OCCURS 0,
           valor(20) TYPE c,
         END OF i_values.
************************************************************************

* Proceso                                                              *

************************************************************************

* Insertamos en la tabla I_FIELDS los atributos del campo a mostrar

  CLEAR i_fields.
  i_fields-tabname = 'ZCLIENXX'.      " Nombre de la tabla
  i_fields-fieldname = 'NOMBR'.       " Nombre del campo
  i_fields-selectflag = 'X'.          " Valor seleccionable
  APPEND i_fields.
* Seleccionamos los distintos nombres existentes en la tabla ZCLIENXX

* y los almacenamos en la tabla de valores

  SELECT DISTINCT nombr INTO zclienxx-nombr
  FROM zclienxx
  WHERE bukrs = p_bukrs.
    i_values-valor = zclienxx-nombr.
    APPEND i_values.
  ENDSELECT.
* Llamamos a la función predefinida que muestra los datos

  CALL FUNCTION 'HELP_VALUES_GET_WITH_TABLE'
*      EXPORTING

*           CUCOL                         = 0

*           CUROW                         = 0

*           DISPLAY                       = ' '

*           FIELDNAME                     = ' '

*           TABNAME                       = ' '

*           NO_MARKING_OF_CHECKVALUE      = ' '

*           TITLE_IN_VALUES_LIST          = ' '

*           TITEL                         = ' '

*           SHOW_ALL_VALUES_AT_FIRST_TIME = ' '

*           NO_CONVERSION                 = ' '

       IMPORTING
            select_value                  = ps_nombr
       TABLES
            fields                        = i_fields
            valuetab                      = i_values
       EXCEPTIONS
            field_not_in_ddic             = 1
            more_then_one_selectfield     = 2
            no_selectfield                = 3
            OTHERS                        = 4.
ENDFORM.                    " VALORES_POSIBLES_NOMBRE
Será necesario definir los símbolos de  texto:

El listado tendrá ahora el siguiente aspecto:

1.2.9 Eventos de rupturas de secuencia en tablas internas.

Para el tratamiento de los registros de una tabla interna con la instrucción LOOP-ENDLOPP, podemos utilizar las siguientes instrucciones de control. Para que el funcionamiento de los eventos AT NEW y  AT END   sea correcto, la tabla interna ha de estar ordenada por los campos controlados.
· AT FIRST … ENDAT.

Este evento se ejecuta para el primer registro de la tabla interna, por tanto las instrucciones escritas en este bloque únicamente se ejecutarán una vez. (Dentro del evento, no estará disponible el valor de ningún campo de la tabla interna).
· AT LAST … ENDAT

Este evento se ejecuta cuando se está procesando el último registro de la tabla interna, por tanto también sus instrucciones se ejecutarán una única vez. (Dentro del evento, no estará disponible el valor de ningún campo de la tabla interna).
· AT NEW campo  … ENDAT.

Este evento se ejecuta si el contenido de la cabecera de la tabla interna  desde el primer campo hasta el campo ‘campo’ incluido es distinta a la del paso anterior, es decir, si cambia la ‘subcabecera’  formada por estos campos. (Dentro del evento, estarán disponibles los valores de los campos de esta ‘subcabecera’, el resto de campos, no estarán disponibles). 
·  AT END OF campo …  ENDAT.

Tiene el mismo funcionamiento que AT  NEW salvo que se ejecuta cuando  se detecta que el siguiente registro no coincide con el registro actual. (Estarán disponibles los valores de los campos formados por la ‘subcabecera’).

Dentro de estos eventos podemos utilizar la instrucción SUM, que lo que hace es sumar todos los campo numéricos de la tabla interna, es decir, acumula en el campo actual de la tabla la suma de los registros tratados hasta la entrada en el evento.
· ON CHANGE OF campo … ENDON.

Este  evento se ejecuta cada vez que el  valor del campo no coincide con el valor anterior. (Este evento no exclusivo de las tablas internas, también se puede utilizar en acceso a base datos lógicas, en SELECT… ENDSELECT,….).
Vamos a ver un ejemplo de utilización de estos eventos. Vamos a mostrar las facturas de cada uno de los clientes, totalizando por moneda.
Para ello vamos a partir del programa anterior y haremos una copia, el nuevo programa se llamará ‘ZREPO1XX’. 
Para hacer una copia de un programa, en la pantalla de acceso al editor de programas: 

 (Ctrol+F5), aparecerá una ventana en la que escribiremos el programa fuente y el programa destino.  A continuación aparecerá otra ventana para indicar que elementos se desean copiar, dejamos las opciones por defecto y pulsamos la opción de copiar (como objeto local).

Sobre el programa copia ‘ZREPO1XX’ haremos  las siguientes modificaciones:

Añadiremos la tabla de facturas en la sentencia TABLES, crearemos una tabla interna para almacenar los datos de las facturas, modificaremos el evento END_OF_SELECTION para recuperar e imprimir las facturas y TOP-OF-PAGE para escribir el detalle de segunda línea de cabecera.
Es decir, el programa modificado será algo así:
************************************************************************

* PROGRAMA: ZREPO1 XX.                                                 *

* DESCRIPCION:  Muestra un listado de los clientes existentes en el    *

*             que se detalla en nº de cliente junto con el nombre y    *

*             apellidos, además, muestra el detalle de cada una  de  *

*             las facturas de los clientes totalizadas por moneda fac. *

* AUTOR : MAD00                     FECHA: 13/08/2001                  *

*----------------------------------------------------------------------*

* CONTROL DE MODIFICACIONES                                            *

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         *

************************************************************************

REPORT zrepo1xx NO STANDARD PAGE HEADING LINE-SIZE 120 LINE-COUNT 80.
************************************************************************

* Tablas del diccionario de datos                                      *

************************************************************************

TABLES: zclienxx,                   " Maestro de clientes.
        zfactuxx.                   " Facturas
************************************************************************

* Definición de constantes                                             *

************************************************************************

CONSTANTS:
    c_pos_nclie(3)   TYPE n VALUE   1,  " Posición nº cliente
    c_pos_nombr(3)   TYPE n VALUE  12,  " Posición nombre
    c_pos_apel1(3)   TYPE n VALUE  33,  " Posición apellido1
    c_pos_apel2(3)   TYPE n VALUE  59,  " Posición apellido2
    c_pos_fnaci(3)   TYPE n VALUE  85,  " Posición F. Nacimiento
    c_pos_final(3)   TYPE n VALUE  96,  " Posición final
    c_ancho_total(3) TYPE n VALUE  96,  " Ancho total del informe,
    c_pos_titul(3)   TYPE n VALUE   1,  " Posición título,
    c_pos_usuar(3)   TYPE n VALUE  37,  " Posición usuario
    c_pos_fecha(3)   TYPE n VALUE  60,  " Posición fecha
    c_pos_pagno(3)   TYPE n VALUE  85,  " Posición nº de página
    c_pos_nfact(3)   TYPE n VALUE  12,  " Posición nº factura
    c_pos_fefac(3)   TYPE n VALUE  23,  " Posición fecha fact
    c_pos_impnt(3)   TYPE n VALUE  34,  " Posición importe neto
    c_pos_moned(3)   TYPE n VALUE  51,  " Posición moneda
    c_pos_fifac(3)   TYPE n VALUE  57,  " Posición final factura.
    c_pos_total(3)   TYPE n VALUE  21.  " Posción total moneda
************************************************************************

* Tablas internas                                                      *

************************************************************************

* Tabla interna de datos a listar

DATA: BEGIN OF i_zclienxx OCCURS 0.
        INCLUDE STRUCTURE zclienxx.
DATA: END OF i_zclienxx.
* Tabla interna con las facturas por cliente

DATA: BEGIN OF i_factu OCCURS 5,
        moneda  LIKE zfactuxx-moned,       " Moneda de la factura
        fecha   LIKE zfactuxx-fecha,       " Fecha
        impnt   LIKE zfactuxx-impnt,       " Importe neto.
        nfact   LIKE zfactuxx-nfact,       " Nº de factura
      END OF i_factu.
************************************************************************

* Pantalla de selección                                                *

************************************************************************

* Parámetros

SELECTION-SCREEN BEGIN OF BLOCK bloq1 WITH FRAME TITLE text-001.
PARAMETERS p_bukrs LIKE zclienxx-bukrs OBLIGATORY
                                       DEFAULT '0001'. " Sociedad
SELECTION-SCREEN END OF BLOCK bloq1.
* Rangos de selección

SELECTION-SCREEN BEGIN OF BLOCK bloq2 WITH FRAME TITLE text-002.
SELECT-OPTIONS:
* Añadimos el match-code para el código de cliente

  s_nclie FOR zclienxx-nclie MATCHCODE OBJECT zcxx,  " Nº de cliente
  s_nombr FOR zclienxx-nombr,        " Nombre cliente
  s_fnaci FOR zclienxx-fnaci.        " Fecha de nacimiento
SELECTION-SCREEN END OF BLOCK bloq2.
************************************************************************

* Comienzo de selección                                                *

************************************************************************

START-OF-SELECTION.
*  Inicializamos la tabla interna.

  REFRESH i_zclienxx.
  CLEAR i_zclienxx.
*  Seleccionamos los datos de los clientes que cumplen criterios

  SELECT *
  FROM zclienxx INTO TABLE i_zclienxx
  WHERE bukrs = p_bukrs
  AND   nclie IN s_nclie
  AND   nombr IN s_nombr
  AND   fnaci IN s_fnaci.
*  Comprobamos si no hay datos seleccionados en cuyo caso,

*  escribiremos un texto que lo indique.

  IF NOT ( sy-subrc = 0 ).
    WRITE / text-003.  " No hay datos para la selección indicada
  ENDIF.
************************************************************************

* Final de la selección                                                *

************************************************************************

END-OF-SELECTION.
* Vamos a escribir el listado con los clientes seleccionados en la

* tabla interna I_ZCLIENXX y para cada uno de ellos sus facturas

  LOOP AT i_zclienxx.
*   Escribimos el nº de cliente en color clave, el resto en otro color

    FORMAT COLOR COL_KEY.
    WRITE:  AT /c_pos_nclie '|' NO-GAP, i_zclienxx-nclie.
    FORMAT COLOR COL_NORMAL.
    WRITE:  AT  c_pos_nombr '|' NO-GAP, i_zclienxx-nombr,
            AT  c_pos_apel1 '|' NO-GAP, i_zclienxx-apel1,
            AT  c_pos_apel2 '|' NO-GAP, i_zclienxx-apel2,
            AT  c_pos_fnaci '|' NO-GAP, i_zclienxx-fnaci,
            AT  c_pos_final '|' NO-GAP.
*   Inicializamos la tabla interna de facturas

    REFRESH i_factu.
    CLEAR i_factu.
*   Recuperamos las facturas correspondientes al cliente

    SELECT moned fecha impnt nfact INTO TABLE i_factu
    FROM zfactuxx
    WHERE bukrs = p_bukrs
    AND   nclie = i_zclienxx-nclie.
*   Comprobamos que se han seleccionado facturas para el cliente

*   si no es así pasamos al siguiente

    CHECK ( sy-subrc = 0 ).
*   Escribimos línea de separación

    ULINE AT /c_pos_nclie(c_ancho_total).
*   Ordenamos las facturas encontradas por moneda

    SORT i_factu BY moneda fecha.
*   Para cada una de las facturas encontradas

    LOOP AT i_factu.
*     Activamos color de detalle

      FORMAT COLOR COL_POSITIVE.
*     Escribimos el detalle de la facutura

      WRITE: AT /c_pos_nclie '|' NO-GAP,
             AT  c_pos_nfact '|' NO-GAP, i_factu-nfact,
             AT  c_pos_fefac '|' NO-GAP, i_factu-fecha,
             AT  c_pos_impnt '|' NO-GAP, i_factu-impnt CURRENCY
                                         i_factu-moneda,
             AT  c_pos_moned '|' NO-GAP, i_factu-moneda,
             AT  c_pos_fifac '|' NO-GAP,
             AT  c_pos_final '|' NO-GAP.
*     Si es la última entrada para la moneda actual

      AT END OF moneda.
*       Sumamos totales importes

        SUM.
*       Línea de separación

        ULINE AT /c_pos_nclie(c_ancho_total).
*       Activamos el color de totales

        FORMAT COLOR COL_TOTAL.
        WRITE: AT /c_pos_nclie '|' NO-GAP,
               AT  c_pos_total  text-013,
                         i_factu-impnt CURRENCY i_factu-moneda NO-GAP,
                         i_factu-moneda,
               AT  c_pos_final '|' NO-GAP.
*       Línea de separación

        ULINE AT /c_pos_nclie(c_ancho_total).
      ENDAT.
    ENDLOOP.
  ENDLOOP.
*   Si se han seleccionado datos, escribimos la línea final.

  IF ( sy-subrc = 0 ).
    ULINE AT /c_pos_nclie(c_ancho_total).
  ENDIF.
************************************************************************

* Cabecera de página                                                   *

************************************************************************

TOP-OF-PAGE.
* Hacemos que la línea permanezca fija en el scroll horizontal

  NEW-LINE NO-SCROLLING.
* Activamos el color de cabecera

  FORMAT COLOR COL_HEADING INTENSIFIED OFF.
* Escribimos información de cabecera resaltando los valores

* Título y sociedad

  WRITE: AT c_pos_titul text-009.
  WRITE p_bukrs INTENSIFIED ON.
* Usuario de creación del informe

  WRITE AT c_pos_usuar text-010.
  WRITE sy-uname INTENSIFIED ON.
* Fecha de creación

  WRITE AT c_pos_fecha text-011.
  WRITE sy-datum INTENSIFIED ON USING EDIT MASK '__/__/____'.
* Nº de página.

  WRITE AT c_pos_pagno text-012.
  WRITE sy-pagno INTENSIFIED ON.
* Escribimos una línea horizontal

  ULINE AT /c_pos_nclie(c_ancho_total).
* Activamos el color de cabeceras intensificado

  FORMAT COLOR COL_HEADING INTENSIFIED ON.
* Escribimos la cabecera de la columnas separadas por el caracter '|'.

* Hacemos que la columna de nº de cliente sea fija en el scroll horiz.

  WRITE: AT /c_pos_nclie '|' NO-GAP, text-004.
* Hasta la columna hacemos fijo

  SET LEFT SCROLL-BOUNDARY.
* Escribimos el detalle de cabecera de cliente

  WRITE: AT  c_pos_nombr '|' NO-GAP, text-005,
         AT  c_pos_apel1 '|' NO-GAP, text-006,
         AT  c_pos_apel2 '|' NO-GAP, text-007,
         AT  c_pos_fnaci '|' NO-GAP, text-008,
         AT  c_pos_final '|' NO-GAP.
* Escribimos una línea horizontal

  ULINE AT /c_pos_nclie(c_ancho_total).
* Escribimos el detalle de cabecera de facturas

  WRITE:  AT /c_pos_nclie '|' NO-GAP,
          AT  c_pos_nfact '|' NO-GAP, text-014,
          AT  c_pos_fefac '|' NO-GAP, text-015,
          AT  c_pos_impnt '|' NO-GAP, text-016,
          AT  c_pos_moned '|' NO-GAP, text-017,
          AT  c_pos_fifac '|' NO-GAP,
          AT  c_pos_final '|' NO-GAP.
* Escribimos una línea horizontal

  ULINE AT /c_pos_nclie(c_ancho_total).
************************************************************************

* Control de la pantalla de selección.                                 *

************************************************************************

* Valores posibles para el rango S_Nombre valor desde.

AT SELECTION-SCREEN ON VALUE-REQUEST FOR s_nombr-low.
* Llamada a la rutina que muestra los valores existentes

  PERFORM valores_posibles_nombre USING s_nombr-low.
* Valores posibles para el rango S_Nombre valor hasta

AT SELECTION-SCREEN ON VALUE-REQUEST FOR s_nombr-high.
* Llamada a la rutina que muestra los valores existentes

  PERFORM valores_posibles_nombre USING s_nombr-high.
************************************************************************

* Rutinas adicionales.                                                 *

************************************************************************

*&---------------------------------------------------------------------*

*&      Form  VALORES_POSIBLES_NOMBRE

*&---------------------------------------------------------------------*

*       Muestra una ventana con los distintos nombres existentes en    *

* la tabla maestro de clientes ZCLIENXX. Dando la posibilidad de       *

* seleccionar uno de ellos. Devuelve en el parámetro PS_NOMBR  el      *

* nombre seleccionado o vacio si no se realiza selección.              *

*----------------------------------------------------------------------*

*      -->PS_NOMBR    Nombre seleccionado                              *

*----------------------------------------------------------------------*

FORM valores_posibles_nombre USING  ps_nombr  LIKE zclienxx-nombr.
************************************************************************

* Variables locales                                                    *

************************************************************************

* Tabla interna para almacenar los campos y atributos a visualizar.

  DATA : BEGIN OF i_fields OCCURS 1.
          INCLUDE STRUCTURE help_value.
  DATA:  END OF i_fields.
* Tabla interna para almacenar el valor de campos a mostrar.

  DATA : BEGIN OF i_values OCCURS 0,
           valor(20) TYPE c,
         END OF i_values.
************************************************************************

* Proceso                                                              *

************************************************************************

* Insertamos en la tabla I_FIELDS los atributos del campo a mostrar

  CLEAR i_fields.
  i_fields-tabname = 'ZCLIENXX'.      " Nombre de la tabla
  i_fields-fieldname = 'NOMBR'.       " Nombre del campo
  i_fields-selectflag = 'X'.          " Valor seleccionable
  APPEND i_fields.
* Seleccionamos los distintos nombres existentes en la tabla ZCLIENXX

* y los almacenamos en la tabla de valores

  SELECT DISTINCT nombr INTO zclienxx-nombr
  FROM zclienxx
  WHERE bukrs = p_bukrs.
    i_values-valor = zclienxx-nombr.
    APPEND i_values.
  ENDSELECT.
* Llamamos a la función predefinida que muestra los datos

  CALL FUNCTION 'HELP_VALUES_GET_WITH_TABLE'
*      EXPORTING

*           CUCOL                         = 0

*           CUROW                         = 0

*           DISPLAY                       = ' '

*           FIELDNAME                     = ' '

*           TABNAME                       = ' '

*           NO_MARKING_OF_CHECKVALUE      = ' '

*           TITLE_IN_VALUES_LIST          = ' '

*           TITEL                         = ' '

*           SHOW_ALL_VALUES_AT_FIRST_TIME = ' '

*           NO_CONVERSION                 = ' '

       IMPORTING
            select_value                  = ps_nombr
       TABLES
            fields                        = i_fields
            valuetab                      = i_values
       EXCEPTIONS
            field_not_in_ddic             = 1
            more_then_one_selectfield     = 2
            no_selectfield                = 3
            OTHERS                        = 4.

ENDFORM.                    " VALORES_POSIBLES_NOMBRE
El Aspecto del listado es el siguiente:

1.2.10 Bases de datos lógicas.

Una base de datos lógica (LDB) proporciona una visión lógica de las tablas físicas, pudiendo relacionar tablas entre si. Las bases de datos lógicas (LDB) son herramientas en los que se definen los accesos a la información almacenada en las tablas del diccionario así como las relaciones entre ellas, proporcionan por tanto,  una visión lógica de las tablas. Están formadas por un conjunto de programas en los que se detallan las tablas utilizadas, las relaciones entre ellas, los parámetros de acceso….
Las LDB simplifican la programación de Reports ya que dentro de su lógica se pueden  encapsular los  accesos de lectura, verificación de autorizaciones, parámetros de selección…etc.  Por tanto se suelen utilizar si existen un conjunto elevado de programas que utilizan la misma información de base de datos.
El principal inconveniente de la bases de datos lógica es su no especialización, es decir, las selecciones a las bases de datos son generales y por tanto serán normalmente más lentas que las selecciones específicas que se pueden realizar dentro de los reports. 
Podemos crear nuestras propias bases se datos lógicas a través del menú ‘Herramientas ( Workbench  ABAP ( Desarrollo ( Entorno de programación ( Bases de datos lógicas’ (SE36). La creación de una base de datos lógica pasa por definir:
· Las tablas utilizadas y su relación.
· Los parámetros de selección sobre las tablas y sus textos. Con sus respectivos match-code si proceden.
· La códificación de las selecciones a las tablas dentro de los procedimientos PUT que serán invocados cuando se ejecute la instrucción de lectura de LDB  ‘GET’.
· La documentación de la LDB.
A su vez, existen s bases de datos lógicas predefinidas que podemos utilizar en nuestros programas, vamos a ver una de ellas. La estructura de la base de datos lógica ‘KDF’ es la siguiente:

· Evento GET.

GET permite obtener los datos de las tablas. El evento GET ‘tablan’ llamará al procedimiento PUT ‘tablan’ codificado en el programa de la base de datos lógica.

La estructura de acceso a los datos desde los reports que utilicen esta base de datos lógica será algo así:

GET LFA1.

Sentencias.

GET LFB1.

Sentencias.



GET BSIK.




…..

El funcionamiento sería el siguiente, para cada uno de los registros encontrados en la tabla LFA1 (que cumplan los criterios de selección), se procesarían los registros de la tabla LFB1 y con cada uno de estos los de la tabla BSIK y así sucesivamente. Tiene dos opciones:

LATE:

Hace que esta instrucción se ejecute una vez procesados todos los registros.

FIELDS:

Hace que solamente se recoja la información de los campos que se indican a continuación.

Si quisiéramos hacer un listado con las partidas abiertas (facturas pendientes de ser pagadas)  de cada acreedor bastaría realizar un REPORT en el que indicásemos que utiliza  la base de datos lógica KDF.   Para ello en la pantalla de atributos del programa indicaremos ‘KD’ en el campo ‘Base de datos lógica’ y ‘F’ en el campo ‘de la aplicación ‘. 

El Report podría ser algo así:

************************************************************************

* PROGRAMA: ZREPO1XX.                                                  *

* DESCRIPCION:  Muestra un listado de las facturas pendientes          *

*             de pago de los acreedores existentes.                    *

* AUTOR : MAD00                     FECHA: 20/08/2001                  *

*----------------------------------------------------------------------*

* CONTROL DE MODIFICACIONES                                            *

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         *

************************************************************************

REPORT zrepo1xx  LINE-SIZE 90.

************************************************************************

* Tablas de diccionario de datos

************************************************************************

TABLES: t001,               " Maestro de sociedades

        lfa1,               " Maestro acreedores general

        lfb1,               " Maestro acreedores  sociedad

        bsik.               " Partidas abiertas acreedor

************************************************************************

* Definición de variables globales                                     *

************************************************************************

DATA :d_dmbtr LIKE bsik-dmbtr.        " Total sociedad

************************************************************************

* Evento de comienzo de selección.

************************************************************************

START-OF-SELECTION.

* Obtenemos los datos de acreedores

GET lfa1.

  FORMAT COLOR COL_KEY.

*  Escribimos nº y nombre del acreedor

  WRITE: / lfa1-lifnr,                        " Nº de acreedor

           lfa1-name1.                        " Nombre

*  Obtenemos datos de sociedad

GET lfb1.

*   Obtenemos la moneda de la sociedad

  SELECT SINGLE waers INTO (t001-waers)

  FROM t001

  WHERE bukrs = lfb1-bukrs.

*   Obtenemos los datos de las partidas abiertas de los acreedores.

GET bsik.

  FORMAT COLOR COL_NORMAL.

*     Obtenemos el signo en función del campo debe/haber

  IF ( bsik-shkzg = 'S' ).

    bsik-wrbtr = bsik-wrbtr * -1.

    bsik-dmbtr = bsik-dmbtr * -1.

  ENDIF.

*     Escribimos los valores

  WRITE:  / bsik-bukrs,                      " Sociedad

            bsik-belnr,                      " Nº doc.

            bsik-zfbdt,                      " Fecha vencimiento

            bsik-wrbtr CURRENCY bsik-waers,   "Importe en mon. doc.

            bsik-waers,                      " Moneda doc.

            bsik-dmbtr CURRENCY t001-waers,   " Importe moneda loc.

            t001-waers.                      " Moneda local.

*     Acumulamos el importe en moneda local

  d_dmbtr = d_dmbtr + bsik-dmbtr.

*     Cuando finaliza el tratamiento de cada sociedad

GET lfb1 LATE.

  FORMAT COLOR COL_TOTAL.

*     Escribimos el importe en moneda de la sociedad del saldo

*     'Importe total para la sociedad : XXXXX   XXX'.

  WRITE: text-001, d_dmbtr CURRENCY t001-waers, t001-waers.

*     Fin de línea e inicialización de importe total sociedad

  ULINE /.

  CLEAR d_dmbtr.

Definiendo las cabeceras y textos oportunos, tendremos un listado con el siguiente aspecto:

1.2.11 FIELD-GROUPS.

Como su nombre indica, representa un conjunto de campos. Al definir un conjunto de campos, únicamente se define el nombre del grupo, para incluir campos en el conjunto será necesario utilizar la instrucción INSERT.

Son útiles cuando hay que mostrar varias líneas de diferente tipo dentro de un mismo informe, por ejemplo en caso anterior, mostramos líneas con datos de los proveedores junto con sus documentos.

Relacionados con los Field-groups tenemos las siguientes instrucciones:

· INSERT
Permite añadir un campo a un conjunto de campos.

· EXTRACT
Permite volcar la información a los campos que forman el conjunto. Formando un ‘registro’ con los campos de los que se compone el conjunto.  Cuando se define el conjunto especial HEADER, a cada uno de estos ‘registros’ se le añaden los campos contenidos en él, formando un ‘registro’ compuesto por los campos HEADER más los del conjunto.

· LOOP - ENDLOOP.


Con los ya conocidos eventos:

· AT NEW f

· AT END F

· AT FIST

· AT LAST

Y el nuevo:

· AT fg
Donde fg sería un conjunto de campos, de tal modo que este evento siempre se ejecutará cuando el tipo de registro actual del loop coincida con un tipo de registro definido por los campos que componen el conjunto.

Vamos a generar un Report, que genera el mismo listado que en el ejemplo anterior pero ayudándonos de FIELD-GROUPS.

Vamos a definir tres conjuntos de campos, 

HEADER:
Con los campos LFA1-LIFNR, BSIK-BUKRS y BSIK-BELNR.

DIRECCION:
Con los campos LFA1-NAME1, LFA1-STRAS, LFA1-PSTLZ y LFA1-ORT01.

PARTIDAS:
BSIK-DMBTR, BSIK-WRBTR, BSIK-SHKZG, BSIK-WAERS.

Definir el HEADER implicará que los tipos de registros de DIRECCION estén formados por los campos:

LFA1-LIFNR,  BSIK-BUKRS,  BSIK-BELNR, LFA1-NAME1, LFA1-STRAS, LFA1-PSTLZ, LFA1-ORT01.

Y los registros PARTIDAS estén compuestos por:

LFA1-LIFNR,  BSIK-BUKRS,  BSIK-BELNR, BSIK-DMBTR BSIK-WRBTR BSIK-SHKZG BSIK-WAERS.

De esta manera los campos del HEADER se convierten en campos claves para la ordenación.

 El código será algo así

************************************************************************

* PROGRAMA: ZREPO2XX.                                                  *

* DESCRIPCION:  Muestra un listado de las facturas pendientes          *

*             de pago de los acreedores existentes.                    *

* AUTOR : MAD00                     FECHA: 20/08/2001                  *

*----------------------------------------------------------------------*

* CONTROL DE MODIFICACIONES                                            *

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         *

************************************************************************

REPORT zrepo2xx  LINE-SIZE 75.

************************************************************************

* Tablas de diccionario de datos

************************************************************************

TABLES: t001,               " Maestro de sociedades

        lfa1,               " Maestro acreedores general

        lfb1,               " Maestro acreedores  sociedad

        bsik.               " Partidas abiertas acreedor

************************************************************************

* Definición de variables globales                                     *

************************************************************************

DATA :d_dmbtr LIKE bsik-dmbtr.        " Total sociedad

************************************************************************

* Definición de FIELD-GROUPS                                           *

************************************************************************

* Definimos los conjuntos de campos

FIELD-GROUPS : header, direccion, partidas.

* Incluimos los campos en cada uno de los conjuntos

INSERT lfa1-lifnr bsik-bukrs bsik-belnr INTO header.

INSERT lfa1-name1 lfa1-stras lfa1-pstlz lfa1-ort01 INTO direccion.

INSERT bsik-dmbtr bsik-wrbtr bsik-shkzg bsik-waers INTO partidas.

************************************************************************

* Evento de comienzo de selección.

************************************************************************

START-OF-SELECTION.

GET lfa1.

* Obtenemos los datos de los campos del conjunto dirección

  EXTRACT direccion.

GET bsik.

* Obtenemos los datos de los campos del conjunto partidas

  EXTRACT partidas.

************************************************************************

* Evento de comienzo final de selección

************************************************************************

END-OF-SELECTION.

* Ordenamos, será por los campos de la cabecera, es decir,

* los campos del conjunto HEADER.

  SORT.

* Recooremos todos los registros seleccionados

  LOOP.

*   Para los registros de tipo dirección

    AT direccion.

*     Activamos color

      FORMAT COLOR COL_KEY.

*     Escribimos datos del acreedor

      WRITE: lfa1-lifnr,lfa1-name1, lfa1-stras, lfa1-pstlz, lfa1-ort01.

    ENDAT.

*   Para cada nueva sociedad

    AT NEW bsik-bukrs.

*     Seleccionamos la moneda de la sociedad para totalizar.

      SELECT SINGLE waers INTO (t001-waers)

      FROM t001

      WHERE bukrs = bsik-bukrs.

    ENDAT.

*   Para cada tipo de registro de documentos

    AT partidas.

      FORMAT COLOR COL_NORMAL.

*     Obtenemos el signo en función del campo debe/haber

      IF ( bsik-shkzg = 'S' ).

        bsik-wrbtr = bsik-wrbtr * -1.

        bsik-dmbtr = bsik-dmbtr * -1.

      ENDIF.

*     Escribimos los valores

      WRITE:  / bsik-bukrs,                      " Sociedad

                bsik-belnr,                      " Nº doc.

                bsik-zfbdt,                      " Fecha vencimiento

                bsik-wrbtr CURRENCY bsik-waers,   "Importe en mon. doc.

                bsik-waers,                      " Moneda doc.

                bsik-dmbtr CURRENCY t001-waers,  " Importe mon soc

                t001-waers.                      " Moneda local.

*     Acumulamos el importe en moneda local

      d_dmbtr = d_dmbtr + bsik-dmbtr.

    ENDAT.

*   En la finalización de cada sociedad

    AT END OF bsik-bukrs.

      FORMAT COLOR COL_TOTAL.

*     Escribimos total para la sociedad.

      WRITE: / text-001,

               bsik-bukrs,  d_dmbtr CURRENCY t001-waers,

               t001-waers.

      ULINE /.

*     Inicializamos el total

      CLEAR d_dmbtr.

    ENDAT.

  ENDLOOP.

(La salida de este programa es idéntica al anterior, salvo que se muestran más datos personales del acreedor, dirección…).

1.3 Programación de listados interactivos

1.3.1 Introducción.

Como hemos visto en el apartado anterior, mediante los listados planos podemos mostrar información sin que el usuario actúe sobre  el informe, únicamente se limita a indicar los parámetros de selección. Además de mostrar esta información podemos hacer que el usuario tenga opciones sobre el informe para introducir datos, desglosar la información, seleccionar un conjunto de registros….Todo esto se realiza dando al informe opciones de usuario.

1.3.2 Eventos

Para manejar las opciones de usuario, disponemos de varios eventos en los cuales podemos codificar cada una de las funciones del listado.

· AT LINE-SELECTION
Este evento se ejecuta cada vez que se selecciona una línea de la pantalla, para seleccionar una línea de un listado,  existen varias formas, las más habituales son   hacer Doble-click sobre la línea,  hacer Click sobre la línea cuando nos aparece el cursor con la forma de mano [image: image43.png]


 o situados sobre la línea pulsar una tecla de función generalmente (F2).

· AT USER-COMMAND
Este evento se ejecuta cada vez que el usuario pulsa una opción, bien a través de un menú, bien a través de un botón…

Cada opción de usuario, lleva implícito un  código de comando, cuando se selecciona una opción,  en la variable del sistema SY-UCOMM  se almacena el código de comando asociado.  Dentro de este evento, realizaremos tantos tratamientos como opciones tengamos en el listado. Normalmente el esquema de programación de este evento es algo así:

CASE  SY-UCOMM.

   WHEN ‘OPC1’.




* Tratamiento de la opción 1

    WHEN ‘OPC2’.


* Tratamiento de la opción 2.

    ….

    WHEN ‘OPCn’.


* Tratamiento de la opción n.

Hay opciones que ya están predefinidas en un informe, por ejemplo las opción de seleccionar una línea del evento AT LINE-SELCTION que tiene el valor PICK, las opciones de paginar (P+, P-, P++ y P--). …

· AT PFn
‘n’ representa un número entre 0 y 99, por tanto podemos tener hasta 100 eventos distintos cada uno ‘gobernará’ la tecla de función correspondiente.

Este evento se ejecuta cada vez que se ejecute la tecla de función n. Normalmente este evento no se utiliza ya que las teclas de función van asociadas a comandos que se pueden controlar en el evento AT USER-COMAND.

1.3.3 Sentencias de lectura y escritura de líneas

Para poder capturar la información de la línea del listado seleccionada, tenemos las instrucciones:

· HIDE:

Permite almacenar el valor de un campo en relación con la línea de escritura, debe de indicarse a continuación de la sentencia WRITE. De esta forma al seleccionar la línea, directamente el campo tomará el valor que tenga en la línea. Para almacenar el valor de un campo con HIDE, no es necesario escribir el campo previamente con la sentencia WRITE, es decir, se pueden almacenar campos que no se escriban en la línea del listado.



Por ejemplo:



WRITE D_NCLIENTE, D_NOMBRE.



HIDE D_NCLIENTE.

Al seleccionar la línea,  en la variable D_NCLIENTE tendremos el valor del Nº de cliente.

· READ LINE...
Lee el contenido de una línea del informe y lo almacena sobre los campos indicados.

Tiene las siguientes variantes:


READ LINE lin.

READ LINE lin OF CURRENT PAGE.

READ LINE lin OF PAGE pag.    

READ CURRENT LINE.

Para indicar los campos que desean ser leídos se añade

FIELD VALUE f1 into g1 f2 into g2 . (Si el campo destino coincide con el origen, se puede abreviar con FIELD VALUE fi).

1.3.4 Niveles de listados anidados.

Cada vez que se produce un evento de usuario y a continuación, se escribe información en la pantalla, se produce un nuevo nivel de lista, es decir, ‘se escribe sobre una nueva hoja que está sobre la anterior’, la hoja anterior permanece y podremos volver a ella.


El nivel de lista se controla con la variable SY-LSIND.

De tal forma que esta será incrementando después de cada opción de usurario, si tras un evento de usuario queremos permanecer en el mismo nivel deberemos indicarlo decrementando  en uno esta variable. (SY-LSIND = SY-LSIND – 1). Otra variable relacionada con el nivel de lista es la variable SY-LISTI, que indica el nivel de lista actual, normalmente vale SY-LSIND – 1.

1.3.5 Ejemplo de listado interactivo I.

Vamos a crear un listado interactivo, partiendo del listado de clientes, programa ‘ZREPORXX’, haremos una copia de este programa y lo llamaremos ‘ZREPO3XX’.

La  primera modificación consistirá en escribir el campo número de cliente con la opción HOTSPOT para que sea sensible a un Click. Además almacenaremos el valor con HIDE para que se recupere automáticamente el valor del campo del cliente seleccionado.

Con el cliente seleccionado, vamos a mostrar un mensaje.

Para ello:

Dentro del evento END-OF-SELECTION.

Modificaremos la línea donde escribimos el código de cliente para activar la selección con Click sobre este campo.

*   Escribimos en modo Click.

WRITE:  AT /C_POS_NCLIE '|' NO-GAP, I_ZCLIENXX-NCLIE HOTSPOT.

*   Almacenamos el valor del campo

HIDE I_ZCLIENXX-NCLIE.

Después de escribir todos los clientes, inicializamos la cabecera de la tabla I_ZCLIENXX.

CLEAR I_ZCLIENXX.

Añadimos el evento de selección de líneas con el siguiente código.

AT LINE-SELECTION.                                                       

*  Comprobamos si es una línea válida, es decir el campo está informado  

   IF ( I_ZCLIENXX-NCLIE IS INITIAL ).                                   

     MESSAGE E000(38) WITH TEXT-013.            

*  Si se ha seleccionado una línea válida.                         

 ELSE.                                                                 

   MESSAGE I000(38) WITH TEXT -014 I_ZCLIENXX-NCLIE.                    

 ENDIF.                                                                

*  Inicializamos el contenido del campo.                                  

 CLEAR I_ZCLIENXX-NCLIE.

Utilizamos la clase de mensajes 38 y el mensaje 000 que no tiene texto, solamente parámetros. Como parámetros pasamos los textos correspondientes que son:

TEXT-013  -- ‘Línea no válida’.

TEXT-014  --‘Cliente seleccionado: ‘.

Podemos comprobar como con un Click sobre el campo NCLIE se muestra el mensaje, pero se necesita un Doble-click para conseguir el mismo efecto en el resto de la línea.

Vamos a complicar nuestro listado de tal forma que, en lugar de mostrar un mensaje con el cliente seleccionado, muestre en un nivel superior otro listado con las facturas del cliente.

Para ello tendremos que realizar las siguientes modificaciones:

Dentro del evento AT LINE-SELECTION.                                                       

*  Comprobamos si es una línea válida, es decir el campo está informado  

   IF ( I_ZCLIENXX-NCLIE IS INITIAL ).                                   

     MESSAGE E000(38) WITH TEXT-013.    

*  Si se ha seleccionado una línea válida.                                 

   ELSE.                                                       

*    Llamamos al procedimiento para mostrar las facturas del cliente.  

     PERFORM MOSTRAR_FACTURAS_CLIENTE USING I_ZCLIENXX-NCLIE.                    

   ENDIF.                                                                

*  Inicializamos el contenido del campo.                                  

   CLEAR I_ZCLIENXX-NCLIE.

Por otra parte será necesario añadir un nuevo evento TOP-OF-PAGE DURING LINE-SELECTION para escribir la cabecera de página del detalle de las facturas.

TOP-OF-PAGE DURING LINE-SELECTION.                  

* Escribimos una línea horizontal                   

  ULINE AT /C_POS_NFACT(C_ANCHO_FAC).               

* Activamos color de cabecera                       

  FORMAT COLOR COL_HEADING.                         

* Escribimos el detalle de la cabecera de factura.  

  WRITE:  AT /C_POS_NFACT '|' NO-GAP, TEXT-017,     

          AT  C_POS_FEFAC '|' NO-GAP, TEXT-018,     

          AT  C_POS_MESFA '|' NO-GAP, TEXT-019,     

          AT  C_POS_IMPNT '|' NO-GAP, TEXT-020,     

          AT  C_POS_MONED '|' NO-GAP, TEXT-021,     

          AT  C_POS_FIFAC '|' NO-GAP.               

* Escribimos una línea horizontal                   

  ULINE AT /C_POS_NFACT(C_ANCHO_FAC).

El código correspondiente al procedimiento MOSTRAR_FACTURAS_CLIENTE será algo así:

*&---------------------------------------------------------------------* 

*&      Form  MOSTRAR_FACTURAS_CLIENTE                                   

*&---------------------------------------------------------------------* 

*      Selecciona las facturas correspondientes al cliente recibido    * 

*  como parámetro y las muestra en pantalla.                           * 

*  Si no se seleccionan facturas, se muestra el mensaje correspondiente 

*----------------------------------------------------------------------* 

*      -->PE_CLIENX   -- Nº de cliente.                                * 

*----------------------------------------------------------------------* 

FORM MOSTRAR_FACTURAS_CLIENTE USING VALUE(PE_NCLIE) LIKE ZCLIENXX-NCLIE. 

************************************************************************ 

* Variables locales                                                    * 

************************************************************************ 

* Tabla interna para almacenar las facturas del cliente                  

DATA: BEGIN OF I_ZFACTUXX OCCURS 0.                                      

        INCLUDE STRUCTURE ZFACTUXX.                                      

DATA: END OF I_ZFACTUXX.

************************************************************************ 

* Proceso                                                              * 

************************************************************************ 

* Seleccionamos las facturas del cliente para la sociedad indicada       

  SELECT *                                                               

  FROM ZFACTUXX INTO TABLE I_ZFACTUXX                                    

  WHERE BUKRS = P_BUKRS                                                  

  AND   NCLIE = PE_NCLIE.                                                

* Si se han seleccionado facturas                                        

  IF ( SY-SUBRC = 0 ).                                                   

*   Activamos color de detalle                                           

    FORMAT COLOR COL_NORMAL.                                             

*   Para cada una de las facturas seleccionadas                          

    LOOP AT I_ZFACTUXX.                                                  

*    Escribimos los datos de las facturas.

WRITE: AT /C_POS_NFACT '|' NO-GAP, I_ZFACTUXX-NFACT,               

            AT  C_POS_FEFAC '|' NO-GAP, I_ZFACTUXX-FECHA,               

            AT  C_POS_MESFA '|' NO-GAP, I_ZFACTUXX-MES,                 

            AT  C_POS_IMPNT '|' NO-GAP, I_ZFACTUXX-IMPNT CURRENCY       

                                       I_ZFACTUXX-MONEDA,               

            AT  C_POS_MONED '|' NO-GAP, I_ZFACTUXX-MONEDA,              

            AT  C_POS_FIFAC '|' NO-GAP.                                 

    ENDLOOP.                                                            

*   Escribimos línea final                                               

    ULINE AT /C_POS_NFACT(C_ANCHO_FAC).                                 

* Si no se han seleccionado                                             

  ELSE.                                                                 

*   Mostramos mensaje de información con                                

*   'No existen facturas para el cliente:'xxx 'para la sociedad:' xxx   

    MESSAGE I000(38) WITH TEXT-015 PE_NCLIE TEXT-016 P_BUKRS.           

  ENDIF.                                                                

ENDFORM.                    " MOSTRAR_FACTURAS_CLIENTE

El listado con las facturas del cliente tendrá el siguiente aspecto:

El siguiente paso a realizar va a consistir en mostrar el detalle de una factura, cuando sea seleccionada en el  listado, mostrando los datos de la factura, del cliente y calculando el IVA correspondiente (Suponemos valor fijo 16%).  El formato de impresión debe ser:

	FACTURA Nº xxxxxxxxxx

Cliente: xxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxxx

Fecha:   dd.mm.aaaa

Mes facturación: xxxxxxxxxx

Moneda: xxxxx

Importe Neto : xxxxxxxxxxxxxxxx

Importe IVA  : xxxxxxxxxxxxxxxx

Total Factura: xxxxxxxxxxxxxxxx




Para conseguir esto realizaremos las siguientes modificaciones en el programa:

Dentro del evento AT-LINE SELECTION será necesario distinguir el nivel de detalle en el que nos encontramos para actuar de un modo u otro, es decir, mostrando la lista de facturas, si estamos en el nivel 0 o mostrando el detalle de la factura si estamos en el nivel 1. Mediante la variable del sistema SY-LISTI podemos saber en todo momento en que nivel de lista nos encontramos.

Del mismo modo tendremos que controlar en el evento TOP-OF-PAGE DURING LINE-SELECTION para escribir o no la cabecera de listado, ya que en el detalle de factura, no vamos a escribir ninguna cabecera de página.

Otra modificación importante que debemos realizar es hacer global la tabla interna de facturas I_ZFACTUXX, hasta ahora local al procedimiento, MOSTRAR_FACTURAS_CLIENTE, de este  modo podremos almacenar su valor con HIDE inmediatamente después de su escritura y este podrá ser recuperado con posterioridad en el evento AT LINE-SELECTION.

Finalmente crearemos un nuevo procedimiento MOSTRAR_DETALLE_FACTURA para escribir el detalle de la factura  seleccionada, recuperando los datos de la factura del cliente  correspondiente y calculando los importes de IVA y Total factura.

El código del programa resultante se muestra a continuación:

************************************************************************

* PROGRAMA: ZREPO3XX.                                                  *

* DESCRIPCION: Muestra un listado de los clientes existentes en el    *

*             que se detalla en nº de cliente junto con el nombre y    *

*             apellidos. Así mismo se podrán visualizar las facturas  *

*             de un cliente y también el detalle de las mismas.        *

* AUTOR: MAD00                     FECHA: 13/08/2001                  *

*----------------------------------------------------------------------*

* CONTROL DE MODIFICACIONES                                            *

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         *

************************************************************************

REPORT ZREPO3XX NO STANDARD PAGE HEADING LINE-SIZE 120 LINE-COUNT 80.

************************************************************************

* Tablas del diccionario de datos                                      *

************************************************************************

TABLES: ZCLIENXX,                   " Maestro de clientes.

        ZFACTUXX.                   " Facturas clientes

************************************************************************

* Definición de constantes                                             *

************************************************************************

CONSTANTS:

*   Constantes posiciones clientes

    C_POS_NCLIE(3)   TYPE N VALUE   1,  " Posición nº cliente

    C_POS_NOMBR(3)   TYPE N VALUE  12,  " Posición nombre

    C_POS_APEL1(3)   TYPE N VALUE  33,  " Posición apellido1

    C_POS_APEL2(3)   TYPE N VALUE  59,  " Posición apellido2

    C_POS_FNACI(3)   TYPE N VALUE  85,  " Posición F. Nacimiento

    C_POS_FINAL(3)   TYPE N VALUE  96,  " Posición final

    C_ANCHO_TOTAL(3) TYPE N VALUE  96,  " Ancho total del informe,

    C_POS_TITUL(3)   TYPE N VALUE   1,  " Posición título,

    C_POS_USUAR(3)   TYPE N VALUE  37,  " Posición usuario

    C_POS_FECHA(3)   TYPE N VALUE  60,  " Posición fecha

    C_POS_PAGNO(3)   TYPE N VALUE  85,  " Posición nº de página

*   Constantes posiciones facturas.

    C_POS_NFACT(3)   TYPE N VALUE   1,  " Posición Nº de factura

    C_POS_FEFAC(3)   TYPE N VALUE  12,  " Posición Fecha

    C_POS_MESFA(3)   TYPE N VALUE  23,  " Posición Mes

    C_POS_IMPNT(3)   TYPE N VALUE  34,  " Posición Importe

    C_POS_MONED(3)   TYPE N VALUE  51,  " Poscición moneda

    C_POS_FIFAC(3)   TYPE N VALUE  57,  " Posición final factura.

    C_ANCHO_FAC(3)   TYPE N VALUE  57,  " Ancho detalle factura.

*   Constantes de posiciones de detalle de factura

    C_POS_D_INICI(3) TYPE N VALUE  1,  " Posición inicial

    C_POS_D_FINAL(3) TYPE N VALUE 80,  " Posición final

    C_POS_D_NFACT(3) TYPE N VALUE 20,  " Posicíon nfactura

    C_POS_D_IMPOR(3) TYPE N VALUE 20.  " Posición importes

************************************************************************

* Tablas internas                                                      *

************************************************************************

* Tabla interna de datos a listar

DATA: BEGIN OF I_ZCLIENXX OCCURS 0.

       INCLUDE STRUCTURE ZCLIENXX.

DATA: END OF I_ZCLIENXX.

* Tabla interna con datos de facturas

DATA: BEGIN OF I_ZFACTUXX OCCURS 0.

        INCLUDE STRUCTURE ZFACTUXX.

DATA: END OF I_ZFACTUXX.

************************************************************************

* Pantalla de selección                                                *

************************************************************************

* Parámetros

SELECTION-SCREEN BEGIN OF BLOCK BLOQ1 WITH FRAME TITLE TEXT-001.

PARAMETERS P_BUKRS LIKE ZCLIENXX-BUKRS OBLIGATORY

                                       DEFAULT '0001'. " Sociedad

SELECTION-SCREEN END OF BLOCK BLOQ1.

* Rangos de selección

SELECTION-SCREEN BEGIN OF BLOCK BLOQ2 WITH FRAME TITLE TEXT-002.

SELECT-OPTIONS:

* Añadimos el match-code para el código de cliente

  S_NCLIE FOR ZCLIENXX-NCLIE MATCHCODE OBJECT ZCXX,  " Nº de cliente

  S_NOMBR FOR ZCLIENXX-NOMBR,        " Nombre cliente

  S_FNACI FOR ZCLIENXX-FNACI.        " Fecha de nacimiento

SELECTION-SCREEN END OF BLOCK BLOQ2.

************************************************************************

* Comienzo de selección                                                *

************************************************************************

START-OF-SELECTION.

*  Inicializamos la tabla interna.

   REFRESH I_ZCLIENXX.

   CLEAR I_ZCLIENXX.

*  Seleccionamos los datos de los clientes que cumplen criterios

   SELECT *

   FROM ZCLIENXX INTO TABLE I_ZCLIENXX

   WHERE BUKRS = P_BUKRS

   AND   NCLIE IN S_NCLIE

   AND   NOMBR IN S_NOMBR

   AND   FNACI IN S_FNACI.

*  Comprobamos si no hay datos seleccionados en cuyo caso,

*  escribiremos un texto que lo indique.

   IF NOT ( SY-SUBRC = 0 ).

     WRITE / TEXT-003.  " No hay datos para la selección indicada

   ENDIF.

************************************************************************

* Final de la selección                                                *

************************************************************************

END-OF-SELECTION.

* Vamos a escribir el listado con los clientes seleccionados en la

* tabla interna I_ZCLIENXX

  LOOP AT I_ZCLIENXX.

*   Escribimos el nº de cliente en color clave, el resto en otro color

    FORMAT COLOR COL_KEY.

*   Escribimos en modo Click.

    WRITE:  AT /C_POS_NCLIE '|' NO-GAP, I_ZCLIENXX-NCLIE HOTSPOT.

*   Almacenamos el valor del campo

    HIDE I_ZCLIENXX-NCLIE.

    FORMAT COLOR COL_NORMAL.

    WRITE:  AT  C_POS_NOMBR '|' NO-GAP, I_ZCLIENXX-NOMBR,

            AT  C_POS_APEL1 '|' NO-GAP, I_ZCLIENXX-APEL1,

            AT  C_POS_APEL2 '|' NO-GAP, I_ZCLIENXX-APEL2,

            AT  C_POS_FNACI '|' NO-GAP, I_ZCLIENXX-FNACI,

            AT  C_POS_FINAL '|' NO-GAP.

  ENDLOOP.

* Si se han seleccionado datos, escribimos la línea final.

  IF ( SY-SUBRC = 0 ).

    ULINE AT /C_POS_NCLIE(C_ANCHO_TOTAL).

  ENDIF.

* Inicializamos el valor de la cabecera de la tabla

  CLEAR I_ZCLIENXX.

************************************************************************

* Cabecera de página                                                   *

************************************************************************

TOP-OF-PAGE.

* Hacemos que la línea permanezca fija en el scroll horizontal

  NEW-LINE NO-SCROLLING.

* Activamos el color de cabecera

  FORMAT COLOR COL_HEADING INTENSIFIED OFF.

* Escribimos información de cabecera resaltando los valores

* Título y sociedad

  WRITE: AT C_POS_TITUL TEXT-009.

  WRITE P_BUKRS INTENSIFIED ON.

* Usuario de creación del informe

  WRITE AT C_POS_USUAR TEXT-010.

  WRITE SY-UNAME INTENSIFIED ON.

* Fecha de creación

  WRITE AT C_POS_FECHA TEXT-011.

  WRITE SY-DATUM INTENSIFIED ON USING EDIT MASK '__/__/____'.

* Nº de página.

  WRITE AT C_POS_PAGNO TEXT-012.

  WRITE SY-PAGNO INTENSIFIED ON.

* Escribimos una línea horizontal

  ULINE AT /C_POS_NCLIE(C_ANCHO_TOTAL).

* Activamos el color de cabeceras intensificado

  FORMAT COLOR COL_HEADING INTENSIFIED ON.

* Escribimos la cabecera de la columnas separadas por el caracter '|'.

* Hacemos que la columna de nº de cliente sea fija en el scroll horiz.

  WRITE: AT /C_POS_NCLIE '|' NO-GAP, TEXT-004.

* Hasta la columna hacemos fijo

  SET LEFT SCROLL-BOUNDARY.

* Escribimos el detalle de cabecera  de cliente

  WRITE: AT  C_POS_NOMBR '|' NO-GAP, TEXT-005,

         AT  C_POS_APEL1 '|' NO-GAP, TEXT-006,

         AT  C_POS_APEL2 '|' NO-GAP, TEXT-007,

         AT  C_POS_FNACI '|' NO-GAP, TEXT-008,

         AT  C_POS_FINAL '|' NO-GAP.

* Escribimos una línea horizontal

  ULINE AT /C_POS_NCLIE(C_ANCHO_TOTAL).

************************************************************************

* Cabecera de página  durante la selección                             *

************************************************************************

TOP-OF-PAGE DURING LINE-SELECTION.

* Comprobamos el nivel de listado para determinar la cabecera

  CASE SY-LISTI.

*   Si nivel 0. Activar cabecera facturas

    WHEN 0.

*     Escribimos una línea horizontal

      ULINE AT /C_POS_NFACT(C_ANCHO_FAC).

*     Activamos color de cabecera

      FORMAT COLOR COL_HEADING.

*     Escribimos el detalle de la cabecera de factura.

      WRITE:  AT /C_POS_NFACT '|' NO-GAP, TEXT-017,

              AT  C_POS_FEFAC '|' NO-GAP, TEXT-018,

              AT  C_POS_MESFA '|' NO-GAP, TEXT-019,

              AT  C_POS_IMPNT '|' NO-GAP, TEXT-020,

              AT  C_POS_MONED '|' NO-GAP, TEXT-021,

              AT  C_POS_FIFAC '|' NO-GAP.

*     Escribimos una línea horizontal

      ULINE AT /C_POS_NFACT(C_ANCHO_FAC).

*   Si nivel 1. No escribimos cabecera.

    WHEN 1.

  ENDCASE.

************************************************************************

* Evento de selección de línea                                         *

************************************************************************

AT LINE-SELECTION.

*  En función del nivel de lista en el que nos encontremos

   CASE SY-LISTI.

*   Si nivel 0, mostrar  facturas cliente

    WHEN 0.

*     Comprobamos si es una línea válida,

*     es decir el campo de código de cliente está informado

      IF ( I_ZCLIENXX-NCLIE IS INITIAL ).

        MESSAGE E000(38) WITH TEXT-013.

*     Si se ha seleccionado un línea válida.

      ELSE.

*       Llamamos al procedimiento para mostrar las facturas del cliente.

        PERFORM MOSTRAR_FACTURAS_CLIENTE USING I_ZCLIENXX-NCLIE.

      ENDIF.

*     Inicializamos el contenido del campo.

      CLEAR I_ZCLIENXX-NCLIE.

*   Si nivel 1, mostrar detalle de factura

    WHEN 1.

*     Comprobamos si es una línea válida, nfact informado

      IF ( I_ZFACTUXX-NFACT IS INITIAL ).

        MESSAGE E000(38) WITH TEXT-013.

*     Si se ha seleccionado una línea válida

      ELSE.

*       Mostramos el detalle de factura.

        PERFORM MOSTRAR_DETALLE_FACTURA USING I_ZFACTUXX-NFACT.

*       Inicializamos el contenido del campo.

        CLEAR I_ZFACTUXX-NFACT.

      ENDIF.

   ENDCASE.

************************************************************************

* Control de la pantalla de selección.                                 *

************************************************************************

* Valores posibles para el rango S_Nombre valor desde.

AT SELECTION-SCREEN ON VALUE-REQUEST FOR S_NOMBR-LOW.

* Llamada a la rutina que muestra los valores existentes

  PERFORM VALORES_POSIBLES_NOMBRE USING S_NOMBR-LOW.

* Valores posibles para el rango S_Nombre valor hasta

AT SELECTION-SCREEN ON VALUE-REQUEST FOR S_NOMBR-HIGH.

* Llamada a la rutina que muestra los valores existentes

  PERFORM VALORES_POSIBLES_NOMBRE USING S_NOMBR-HIGH.

************************************************************************

* Rutinas adicionales.                                                 *

************************************************************************

*&---------------------------------------------------------------------*

*&      Form  VALORES_POSIBLES_NOMBRE

*&---------------------------------------------------------------------*

*       Muestra una ventana con los distintos nombres existentes en    *

* la tabla maestro de clientes ZCLIENXX. Dando la posibilidad de       *

* seleccionar uno de ellos. Devuelve en el parámetro PS_NOMBR  el      *

* nombre seleccionado o vacio si no se realiza selección.              *

*----------------------------------------------------------------------*

*      -->PS_NOMBR    Nombre seleccionado                              *

*----------------------------------------------------------------------*

FORM VALORES_POSIBLES_NOMBRE USING  PS_NOMBR  LIKE ZCLIENXX-NOMBR.

************************************************************************

* Variables locales                                                    *

************************************************************************

* Tabla interna para almacenar los campos y atributos a visualizar.

DATA : BEGIN OF I_FIELDS OCCURS 1.

         INCLUDE STRUCTURE HELP_VALUE.

DATA:  END OF I_FIELDS.

* Tabla interna para almacenar el valor de campos a mostrar.

DATA : BEGIN OF I_VALUES OCCURS 0,

         VALOR(20) TYPE C,

       END OF I_VALUES.

************************************************************************

* Proceso                                                              *

************************************************************************

* Insertamos en la tabla I_FIELDS los atributos del campo a mostrar

  CLEAR I_FIELDS.

  I_FIELDS-TABNAME = 'ZCLIENXX'.      " Nombre de la tabla

  I_FIELDS-FIELDNAME = 'NOMBR'.       " Nombre del campo

  I_FIELDS-SELECTFLAG = 'X'.          " Valor seleccionable

  APPEND I_FIELDS.

* Seleccionamos los distintos nombres existentes en la tabla ZCLIENXX

* y los almacenamos en la tabla de valores

  SELECT DISTINCT NOMBR INTO ZCLIENXX-NOMBR

  FROM ZCLIENXX

  WHERE BUKRS = P_BUKRS.

    I_VALUES-VALOR = ZCLIENXX-NOMBR.

    APPEND I_VALUES.

  ENDSELECT.

* Llamamos a la función predefinida que muestra los datos

  CALL FUNCTION 'HELP_VALUES_GET_WITH_TABLE'

*      EXPORTING

*           CUCOL                         = 0

*           CUROW                         = 0

*           DISPLAY                       = ' '

*           FIELDNAME                     = ' '

*           TABNAME                       = ' '

*           NO_MARKING_OF_CHECKVALUE      = ' '

*           TITLE_IN_VALUES_LIST          = ' '

*           TITEL                         = ' '

*           SHOW_ALL_VALUES_AT_FIRST_TIME = ' '

*           NO_CONVERSION                 = ' '

       IMPORTING

            SELECT_VALUE                  = PS_NOMBR

       TABLES

            FIELDS                        = I_FIELDS

            VALUETAB                      = I_VALUES

       EXCEPTIONS

            FIELD_NOT_IN_DDIC             = 1

            MORE_THEN_ONE_SELECTFIELD     = 2

            NO_SELECTFIELD                = 3

            OTHERS                        = 4.

ENDFORM.                    " VALORES_POSIBLES_NOMBRE

*&---------------------------------------------------------------------*

*&      Form  MOSTRAR_FACTURAS_CLIENTE

*&---------------------------------------------------------------------*

*      Selecciona las facturas correspondientes al cliente recibido    *

*  como parámetro y las muestra en pantalla.                           *

*  Si no se seleccionan facturas, se muestra el mensaje correspondiente

*----------------------------------------------------------------------*

*      -->PE_CLIENX   -- Nº de cliente.                                *

*----------------------------------------------------------------------*

FORM MOSTRAR_FACTURAS_CLIENTE USING VALUE(PE_NCLIE) LIKE ZCLIENXX-NCLIE.

************************************************************************

* Proceso                                                              *

************************************************************************

* Seleccionamos las facturas del cliente para la sociedad indicada

  SELECT *

  FROM ZFACTUXX INTO TABLE I_ZFACTUXX

  WHERE BUKRS = P_BUKRS

  AND   NCLIE = PE_NCLIE.

* Si se han seleccionado facturas

  IF ( SY-SUBRC = 0 ).

*   Activamos color de detalle

    FORMAT COLOR COL_NORMAL.

*   Para cada una de las facturas seleccionadas

    LOOP AT I_ZFACTUXX.

*    Escribimos los datos de las facturas.

     WRITE: AT /C_POS_NFACT '|' NO-GAP, I_ZFACTUXX-NFACT.

*    Almacenamos el valor del nº de factura.

     HIDE I_ZFACTUXX-NFACT.

     WRITE: AT  C_POS_FEFAC '|' NO-GAP, I_ZFACTUXX-FECHA,

            AT  C_POS_MESFA '|' NO-GAP, I_ZFACTUXX-MESFA,

            AT  C_POS_IMPNT '|' NO-GAP, I_ZFACTUXX-IMPNT CURRENCY

                                       I_ZFACTUXX-MONED,

            AT  C_POS_MONED '|' NO-GAP, I_ZFACTUXX-MONED,

            AT  C_POS_FIFAC '|' NO-GAP.

    ENDLOOP.

*   Escribimos líne final

    ULINE AT /C_POS_NFACT(C_ANCHO_FAC).

* Si no se han seleccionado

  ELSE.

*   Mostramos mensaje de información con

*   'No existen facturas para el cliente:'xxx 'para la sociedad:' xxx

    MESSAGE I000(38) WITH TEXT-015 PE_NCLIE TEXT-016 P_BUKRS.

  ENDIF.

ENDFORM.                    " MOSTRAR_FACTURAS_CLIENTE

*&---------------------------------------------------------------------*

*&      Form  MOSTRAR_DETALLE_FACTURA

*&---------------------------------------------------------------------*

*      Muestra el detalle de una factura, mostrando el cliente        *

*   al que pertenece y calculando el IVA correspondiente a la misma    *

*   su poniendo un tipo de IVA fijo del 16 %.                          *

*----------------------------------------------------------------------*

*      --> PE_NFACT   Código de factura                                *

*----------------------------------------------------------------------*

FORM MOSTRAR_DETALLE_FACTURA USING VALUE(PE_NFACT) LIKE ZFACTUXX-NFACT.

************************************************************************

* Definición de variables locales                                      *

************************************************************************

DATA: D_NOMBRE_CLIENTE(70) TYPE C,         " Nombre completo

      D_IMPOR_IVA   LIKE ZFACTUXX-IMPNT,   " Importe del IVA

      D_IMPOR_TOTAL LIKE ZFACTUXX-IMPNT.   " Importe total

************************************************************************

* Proceso                                                              *

************************************************************************

* Obtenemos los datos de la factura para ello leemos de la tabla interna

  READ TABLE I_ZFACTUXX WITH KEY NFACT = PE_NFACT.

* Comprobamos que existe

  CHECK ( SY-SUBRC = 0 ).

* Obtenemos los datos necesarios del cliente

  SELECT SINGLE NOMBR APEL1 APEL2

  FROM ZCLIENXX INTO (ZCLIENXX-NOMBR, ZCLIENXX-APEL1, ZCLIENXX-APEL2)

  WHERE BUKRS = P_BUKRS

  AND   NCLIE = I_ZFACTUXX-NCLIE.

* Comprobamos que existe.

  CHECK ( SY-SUBRC = 0 ).

* Escribimos los datos en la forma adecuada

  ULINE AT /C_POS_D_INICI(C_POS_D_FINAL).

* Nº de factura.

  WRITE: AT /C_POS_D_INICI '|',

         AT  C_POS_D_NFACT TEXT-022, I_ZFACTUXX-NFACT,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Formamos el nombre completo del cliente del cliente

  CONCATENATE ZCLIENXX-NOMBR ZCLIENXX-APEL1 ZCLIENXX-APEL2

              INTO D_NOMBRE_CLIENTE SEPARATED BY SPACE.

* Nombre del cliente

  WRITE: AT /C_POS_D_INICI '|',TEXT-023, D_NOMBRE_CLIENTE,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Fecha de factura

  WRITE: AT /C_POS_D_INICI '|', TEXT-024, I_ZFACTUXX-FECHA,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Mes de la factura.

  WRITE: AT /C_POS_D_INICI '|', TEXT-025, I_ZFACTUXX-MESFA,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Moneda de la factura.

  WRITE: AT /C_POS_D_INICI '|', TEXT-026, I_ZFACTUXX-MONED,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Importe neto:

  WRITE: AT /C_POS_D_INICI '|',

         AT C_POS_D_IMPOR   TEXT-027,

                            I_ZFACTUXX-IMPNT CURRENCY I_ZFACTUXX-MONED,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Calculamos el importe de IVA. (16 %).

  D_IMPOR_IVA = ( I_ZFACTUXX-IMPNT * 16 ) / 100.

  WRITE: AT /C_POS_D_INICI '|',

         AT  C_POS_D_IMPOR  TEXT-028,

                            D_IMPOR_IVA CURRENCY I_ZFACTUXX-MONED,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Calculamos el importe total

  D_IMPOR_TOTAL = I_ZFACTUXX-IMPNT + D_IMPOR_IVA.

  WRITE: AT /C_POS_D_INICI '|',

         AT  C_POS_D_IMPOR  TEXT-029,

                            D_IMPOR_TOTAL CURRENCY I_ZFACTUXX-MONED,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Mostramos línea final

  ULINE AT /C_POS_D_INICI(C_POS_D_FINAL).

ENDFORM.                    " MOSTRAR_DETALLE_FACTURA

1.3.6 Ejemplo de listado interactivo II.

Vamos a ver otro ejemplo de listado interactivo, es este caso va a consistir en mostrar un listado con los clientes existentes y poder marcar y desmarcar algunos de ellos para su posterior tratamiento.

Para ello vamos a realizar una copia del programa anterior  ‘ZREPO3XX’ a la que llamaremos ‘ZREPO4XX ‘y sobre este realizaremos las siguientes modificaciones:

En la tabla interna I_ZCLIENXX, será necesario añadir un nuevo campo (CHAR 1) para poder ser usado como campo para marcar y desmarcar. Para ello definiremos la tabla del siguiente modo:

* Tabla interna de datos a listar                             

DATA: BEGIN OF I_ZCLIENXX OCCURS 0.                           

DATA  MARCA(1) TYPE C.               " Campo para marcar     

               INCLUDE STRUCTURE ZCLIENXX.                     

DATA: END OF I_ZCLIENXX.

Al incluir un nuevo campo en la tabla tendremos que tenerlo en cuenta en la instrucción  SELECT  (de acceso a la tabla ZCLIENXX) ya que ahora la instrucción habrá que realizar el siguiente cambio.

FROM ZCLIENXX INTO TABLE I_ZCLIENXX.

Por

FROM ZCLIENXX INTO CORRESPONDING FIELDS OF TABLE I_ZCLIENXX.

Si no hiciéramos esto, estaríamos recuperando la información pero  el valor se volcaría de forma errónea sobre los campos.

También modificaremos el evento TOP-OF-PAGE para ajustar la cabecera al nuevo campo. (Deberemos cambiar también el valor de las constantes para llevar el listado dos posiciones a la derecha para abrir hueco al campo Marca).

Será necesario además escribir el campo en el listado en forma de casilla de selección y de entrada para que pueda ser marcado y desmarcado. Para ello modificamos el evento END-OF-SELECTION al que añadiremos:

*   Escribimos el campo MARCA en forma de casilla de selección  

*   y con los atributos de campo de entrada.                    

    WRITE: AT /C_POS_INICI '|' NO-GAP,                          

              I_ZCLIENXX-MARCA AS CHECKBOX INPUT ON.

Con la inclusión de estos cambios el listado tendrá el siguiente aspecto:

El siguiente paso será dar la posibilidad de  mostrar una ventana con  las facturas con los clientes seleccionados, mediante la opción de un botón en la barra de pulsadores.

Para crear opciones de usuario, bien como botones  el la barra de pulsadores, bien como opciones de menú en la barra de menús, es necesario definir un STATUS.

Existen  cuatro tipos de STATUS dependiendo de su utilización:

· Dynpro

· Ventana de diálogo

· Lista

· Lista en Ventana de dialogo

Los dos primeros se utilizan en pantallas (Programación de dialogo que se verá en capítulos posteriores).

Los dos siguientes se utilizan en la programación de listados interactivos.  El tipo lista es el que se utiliza normalmente en los listados a pantalla completa, quedando la lista en ventana de diálogo para los listados en ventanas. No obstante podemos utilizar un tipo Lista en una ventana y viceversa.

Vamos a crear un STATUS tipo lista al que llamaremos ‘S00’ para ello escribimos la instrucción para activar el STATUS, en nuestro caso,  la situaremos al comienzo del evento END-OF-SELECTION

SET PF-STATUS ‘S00’.

Esta instrucción provocará que se llame al STATUS S00 en lugar de al STATUS por defecto que se llamaba hasta ahora.

Usamos la utilidad del editor de navegar al objeto haciendo Doble-Click sobre él , nos saldrá una ventana indicando que el Status no existe, pero nos da la posibilidad de crearlo, nos aparecerá la siguiente ventana donde deberemos indicar los atributos del STATUS, pondremos ‘Listado clientes’ en texto breve y seleccionaremos el tipo ‘Lista’.

Pulsamos continuar, llegaremos a la siguiente pantalla de para la definición de STATUS:

Por defecto, aparecen todas las opciones existentes en los listados (En los ejemplos anteriores teníamos ya estas opciones sin crear nuestro propio STATUS), podemos eliminar estas opciones, borrando el código de comando correspondiente de la barra de símbolos. (En nuestro caso no vamos a eliminar ninguna).

Creamos nuestro botón facturas, para ello, escribimos el código de comando que vamos a asociar al botón en el primer hueco disponible de la barra de pulsadores, pondremos ‘FACT’ al pulsar ENTER nos saldrá una ventana para elegir la tecla de función asociada al comando, seleccionaremos ‘F5’, a continuación nos saldrá la siguiente ventana:

Donde pondremos el texto que mostrará el botón, ‘Visualizar Facturas’ en nuestro caso. Tras aceptar  tendremos nuestro primer botón.

Aprovechamos para incluir dos nuevos botones, un botón que nos servirá para marcar a todos los clientes del listado y otro que permita desmarcarlos. Estos botones van a tener la particularidad de que van a tener asociado un  Icono.

Vamos a crear el botón ‘MARCAR’ al que le asociaremos el ICONO ([image: image49.png]


). Para ello, nos situamos en la primera tecla de función de libre definición, que este libre (las veremos haciendo un avance de página)  en nuestro caso (‘F6’), seleccionamos la opción del menú ‘Tratar(Insertar(Función con Icono (Ctrl. + F10), nos aparecerá una ventana donde pondremos el código de comando asociado ‘MARC’ en nuestro caso, nos llevará a la siguiente ventana:

ICONO:
Es en nombre del icono, que se mostrará. (En nuestro seleccionaremos el icono ‘ICON_SELECT_ALL’ ([image: image51.png]


)).

QuickInfo/ Texto menú:
Es el texto de información que se mostrará en la ventana de ayuda que sale al situarse sobre un botón. (En nuestro caso pondremos ‘Marcar todos’.)

Texto Icono:
Corresponde al texto que tendrá el botón además del icono. (En nuestro caso lo dejamos en blanco).

Tras pulsar aceptar aparecerá la opción en la línea correspondiente, (‘F6’ en nuestro caso.

Repetimos el proceso para el botón ‘DESMARCAR’ asociándole a la tecla de función ‘F7’, el comando ‘DESM’, el  ICONO ‘ICON_DESELECT_ALL’ ([image: image52.png]


) y el texto de ayuda ‘Desmarcar todos’.

Una vez definidos ambas comandos, escribiremos, MARC y DESM en la barra de pulsadores. Junto a ellos añadiremos el comando predefinido ‘PICK’ correspondiente a la selección de líneas (Este es el que hasta ahora se activa al hacer Click y Doble-Click) de esta manera queda más claro en que listados está disponible la opción de selección.

A continuación grabaremos y activaremos el STATUS.

Será necesario definirse un  nuevo STATUS  con las siguientes propiedades:

S01:   Será de tipo lista en ventana de diálogo, será el que activemos cuando llamemos a la ventana donde mostrar las facturas de los clientes.


Tendrá las opciones que se crean por defecto más la opción ‘PICK’ que añadiremos a la barra de pulsadores.

Para el listado de detalle de factura, no será necesario definirse ningún STATUS ya que utilizaremos el STATUS estándar, activándolo con ’SET PF-STATUS SPACE’.

De vuelta  al código del programa ya tendremos un nuevo STATUS con las opciones disponibles. Ahora falta realizar el tratamiento de dichas opciones. Este tratamiento se realiza en el evento AT USER-COMMAND, en el que hay que controlar:

· Visualizar facturas:

Para visualizar las facturas de los clientes seleccionados, será necesario, hacer dos cosas.

  En primer lugar, leer las marcas del listado, es decir, ver que clientes están marcados en pantalla, actualizaremos el campo marca en la tabla interna I_ZCLIENTES. Esta tarea se realiza dentro del procedimiento ‘LEER_MARCAS_LISTADO’.

  Por otra parte será necesario mostrar las facturas de aquellos clientes seleccionados,  es decir, de aquellos clientes que tengan el campo marca activado. Esto se hace dentro del procedimiento ‘MOSTRAR_FACTURAS_CLIENTES’ en el que se activa el STATUS ‘S01’ correspondiente al listado de facturas y se activa el listado en forma  de ventana utilizando la instrucción WINDOW STARTING AT… ENDING AT… En la llamada al STATUS, eliminamos la opción de selección para impedir que se muestre el detalle de la factura.  Posteriormente se llama al procedimiento, ya existente, ‘MOSTRAR_FACTURAS_CLIENTE’  para mostrar las facturas de cada cliente seleccionado.

· Marcar  todos los clientes.

Para marcar y visualizar en el listado las marcas, es necesario:

Por una parte modificar el campo MARCA  de todos los clientes de la tabla interna I_ZCLIENXX para ponerles a ‘X’, utilizamos para ello el procedimiento ‘MARCAR_DESMARCAR’ al que le pasamos como parámetro el valor ‘X’.

Una vez modificado este campo será necesario ‘refrescar’ el listado, para ello, tendremos que escribir de nuevo el listado con los clientes, permaneciendo en el mismo nivel de lista, esto lo controlamos descontando uno a la variable SY-LSIND. 

· Desmarcar todos los clientes.

El procedimiento para desmarcar los clientes es similar, a diferencia de que actualizamos el campo marca con el valor SPACE.

Cambiamos el modo de controlar los niveles de lista, utilizaremos la variable SY-PFKEY que nos indica el STATUS actual, en función de este, mostraremos una cabecera u otra un listado u otro, en los eventos AT LINE-SELECTION, AT TOP-PAGE DURING LINE-SELECTION.

El listado completo del programa será algo así:

************************************************************************

* PROGRAMA: ZREPO4XX.                                                  *

* DESCRIPCION: Muestra un listado de los clientes existentes en el     *

*             que se detalla en nº de cliente junto con el nombre y    *

*             apellidos . Así mismo se podrán visualizar las facturas  *

*             de un cliente/s y también el detalle de las mismas.      *

* AUTOR : Nombre y Apellidos        FECHA: 13/08/2001                  *

*----------------------------------------------------------------------*

* CONTROL DE MODIFICACIONES                                            *

*  FECHA.           AUTOR.           DESCRIPCION MODIFICACION.         *

************************************************************************

REPORT ZREPO4XX NO STANDARD PAGE HEADING LINE-SIZE 120 LINE-COUNT 80.

************************************************************************

* Tablas del diccionario de datos                                      *

************************************************************************

TABLES: ZCLIENXX,                   " Maestro de clientes.

        ZFACTUXX.                   " Facturas clientes

************************************************************************

* Definición de constantes                                             *

************************************************************************

CONSTANTS:

*   Constantes posiciones clientes

    C_POS_INICI(3)   TYPE N VALUE   1,  " Posición de inicio

    C_POS_NCLIE(3)   TYPE N VALUE   3,  " Posición nº cliente

    C_POS_NOMBR(3)   TYPE N VALUE  14,  " Posición nombre

    C_POS_APEL1(3)   TYPE N VALUE  35,  " Posición apellido1

    C_POS_APEL2(3)   TYPE N VALUE  61,  " Posición apellido2

    C_POS_FNACI(3)   TYPE N VALUE  87,  " Posición F. Nacimiento

    C_POS_FINAL(3)   TYPE N VALUE  98,  " Posición final

    C_ANCHO_TOTAL(3) TYPE N VALUE  98,  " Ancho total del informe,

    C_POS_TITUL(3)   TYPE N VALUE   1,  " Posición título,

    C_POS_USUAR(3)   TYPE N VALUE  37,  " Posición usuario

    C_POS_FECHA(3)   TYPE N VALUE  60,  " Posición fecha

    C_POS_PAGNO(3)   TYPE N VALUE  85,  " Posición nº de página

*   Constantes posiciones facturas.

    C_POS_NFACT(3)   TYPE N VALUE   1,  " Posición Nº de factura

    C_POS_FEFAC(3)   TYPE N VALUE  12,  " Posición Fecha

    C_POS_MESFA(3)   TYPE N VALUE  23,  " Posición Mes

    C_POS_IMPNT(3)   TYPE N VALUE  34,  " Posición Importe

    C_POS_MONED(3)   TYPE N VALUE  51,  " Poscición moneda

    C_POS_FIFAC(3)   TYPE N VALUE  57,  " Posición final factura.

    C_ANCHO_FAC(3)   TYPE N VALUE  57,  " Ancho detalle factura.

*   Constantes de posiciones de detalle de factura

    C_POS_D_INICI(3) TYPE N VALUE  1,  " Posición inicial

    C_POS_D_FINAL(3) TYPE N VALUE 80,  " Posición final

    C_POS_D_NFACT(3) TYPE N VALUE 20,  " Posicíon nfactura

    C_POS_D_IMPOR(3) TYPE N VALUE 20.  " Posición importes

************************************************************************

* Tablas internas                                                      *

************************************************************************

* Tabla interna de datos a listar de clientes

DATA: BEGIN OF I_ZCLIENXX OCCURS 0.

DATA   MARCA(1) TYPE C.               " Campo para marcar

       INCLUDE STRUCTURE ZCLIENXX.

DATA: END OF I_ZCLIENXX.

* Tabla interna con datos de facturas

DATA: BEGIN OF I_ZFACTUXX OCCURS 0.

        INCLUDE STRUCTURE ZFACTUXX.

DATA: END OF I_ZFACTUXX.

************************************************************************

* Pantalla de selección                                                *

************************************************************************

* Parámetros

SELECTION-SCREEN BEGIN OF BLOCK BLOQ1 WITH FRAME TITLE TEXT-001.

PARAMETERS P_BUKRS LIKE ZCLIENXX-BUKRS OBLIGATORY

                                       DEFAULT '0001'. " Sociedad

SELECTION-SCREEN END OF BLOCK BLOQ1.

* Rangos de selección

SELECTION-SCREEN BEGIN OF BLOCK BLOQ2 WITH FRAME TITLE TEXT-002.

SELECT-OPTIONS:

* Añadimos el match-code para el código de cliente

  S_NCLIE FOR ZCLIENXX-NCLIE MATCHCODE OBJECT ZCXX,  " Nº de cliente

  S_NOMBR FOR ZCLIENXX-NOMBR,        " Nombre cliente

  S_FNACI FOR ZCLIENXX-FNACI.        " Fecha de nacimiento

SELECTION-SCREEN END OF BLOCK BLOQ2.

************************************************************************

* Comienzo de selección                                                *

************************************************************************

START-OF-SELECTION.

*  Inicializamos la tabla interna.

   REFRESH I_ZCLIENXX.

   CLEAR I_ZCLIENXX.

*  Seleccionamos los datos de los clientes que cumplen criterios

   SELECT *

   FROM ZCLIENXX INTO CORRESPONDING FIELDS OF TABLE I_ZCLIENXX

   WHERE BUKRS = P_BUKRS

   AND   NCLIE IN S_NCLIE

   AND   NOMBR IN S_NOMBR

   AND   FNACI IN S_FNACI.

*  Comprobamos si no hay datos seleccionados en cuyo caso,

*  escribiremos un texto que lo indique.

   IF NOT ( SY-SUBRC = 0 ).

     WRITE / TEXT-003.  " No hay datos para la selección indicada

   ENDIF.

************************************************************************

* Final de la selección                                                *

************************************************************************

END-OF-SELECTION.

* Activamos el status de listado de clientes

  SET PF-STATUS 'S00'.

* Llamamos al procedimiento que muestra el listado de clientes

  PERFORM MOSTRAR_CLIENTES.

************************************************************************

* Cabecera de página                                                   *

************************************************************************

TOP-OF-PAGE.

* Llamamos al proce. que escribe la cabecera del listado de clientes

  PERFORM MOSTRAR_CABECERA_CLIENTES.

************************************************************************

* Cabecera de página  durante la selección                             *

************************************************************************

TOP-OF-PAGE DURING LINE-SELECTION.

* Comprobamos el STATUS activo en función del cual mostraremos la

* cabecera correspondiente

  CASE SY-PFKEY.

*   Si status S00, activa cabecera clientes

    WHEN  'S00'.

*     Llamamos al proce. que escribe cabecera de clientes

      PERFORM MOSTRAR_CABECERA_CLIENTES.

*   Si status S01, actvivamos cabecera facturas

    WHEN 'S01'.

*      Escribimos una línea horizontal

       ULINE AT /C_POS_NFACT(C_ANCHO_FAC).

*      Activamos color de cabecera

       FORMAT COLOR COL_HEADING.

*      Escribimos el detalle de la cabecera de factura.

       WRITE:  AT /C_POS_NFACT '|' NO-GAP, TEXT-017,

               AT  C_POS_FEFAC '|' NO-GAP, TEXT-018,

               AT  C_POS_MESFA '|' NO-GAP, TEXT-019,

               AT  C_POS_IMPNT '|' NO-GAP, TEXT-020,

               AT  C_POS_MONED '|' NO-GAP, TEXT-021,

               AT  C_POS_FIFAC '|' NO-GAP.

*      Escribimos una línea horizontal

       ULINE AT /C_POS_NFACT(C_ANCHO_FAC).

  ENDCASE.

************************************************************************

* Evento de selección de línea                                         *

************************************************************************

AT LINE-SELECTION.

*  En función del STATUS existente

   CASE SY-PFKEY.

*   Si Listado de clientes 'S00'.

    WHEN  'S00'.

*     Comprobamos si es una línea válida,

*     es decir el campo de código de cliente está informado

      IF ( I_ZCLIENXX-NCLIE IS INITIAL ).

        MESSAGE E000(38) WITH TEXT-013.

*     Si se ha seleccionado un línea válida.

      ELSE.

*       Activamos el STATUS del listado de facturas

        SET PF-STATUS 'S01'.

*       Llamamos al procedimiento para mostrar las facturas del cliente.

        PERFORM MOSTRAR_FACTURAS_CLIENTE USING I_ZCLIENXX-NCLIE.

      ENDIF.

*     Inicializamos el contenido del campo.

      CLEAR I_ZCLIENXX-NCLIE.

*   Si listado de facturas 'S01'.

    WHEN 'S01'.

*     Comprobamos si es una línea válida, nfact informado

      IF ( I_ZFACTUXX-NFACT IS INITIAL ).

        MESSAGE E000(38) WITH TEXT-013.

*     Si se ha seleccionado una línea válida

      ELSE.

*       Activamos el STATUS por defecto.

        SET PF-STATUS SPACE.

*       Mostramos el detalle de factura.

        PERFORM MOSTRAR_DETALLE_FACTURA USING I_ZFACTUXX-NFACT.

*       Inicializamos el contenido del campo.

        CLEAR I_ZFACTUXX-NFACT.

      ENDIF.

   ENDCASE.

************************************************************************

* Evento de comandos de usuario                                        *

************************************************************************

AT USER-COMMAND.

* En función del comando seleccionado

  CASE SY-UCOMM.

*  Si opción facturas

   WHEN 'FACT'.

*    Leemos los clientes seleccionados

     PERFORM LEER_MARCAS_LISTADO.

*    Mostramos las facturas de los clientes

     PERFORM MOSTRAR_FACTURAS_CLIENTES.

*  Si opción de marcar todos los clientes

   WHEN 'MARC'.

*    Marcamos a todos los cliente desmarcados

     PERFORM MARCAR_DESMARCAR USING 'X'.

*    Mantenemos el mismo nivel de lista actual

     SY-LSIND = SY-LSIND - 1.

*    Reescribimos el listado en  los nuevos valores actualizados

     PERFORM MOSTRAR_CLIENTES.

*  Si opción de desmarcar todos los clientes.

   WHEN 'DESM'.

*    Desmarcamos a todos los clientes marcados

     PERFORM MARCAR_DESMARCAR USING SPACE.

*    Mantenemos el mismo nivel de lista actual

     SY-LSIND = SY-LSIND - 1.

*    Reescribimos el listado en  los nuevos valores actualizados

     PERFORM MOSTRAR_CLIENTES.

  ENDCASE.

************************************************************************

* Control de la pantalla de selección.                                 *

************************************************************************

* Valores posibles para el rango S_Nombre valor desde.

AT SELECTION-SCREEN ON VALUE-REQUEST FOR S_NOMBR-LOW.

* Llamada a la rutina que muestra los valores existentes

  PERFORM VALORES_POSIBLES_NOMBRE USING S_NOMBR-LOW.

* Valores posibles para el rango S_Nombre valor hasta

AT SELECTION-SCREEN ON VALUE-REQUEST FOR S_NOMBR-HIGH.

* Llamada a la rutina que muestra los valores existentes

  PERFORM VALORES_POSIBLES_NOMBRE USING S_NOMBR-HIGH.

************************************************************************

* Rutinas adicionales.                                                 *

************************************************************************

*&---------------------------------------------------------------------*

*&      Form  VALORES_POSIBLES_NOMBRE

*&---------------------------------------------------------------------*

*       Muestra una ventana con los distintos nombres existentes en    *

* la tabla maestro de clientes ZCLIENXX. Dando la posibilidad de       *

* seleccionar uno de ellos. Devuelve en el parámetro PS_NOMBR  el      *

* nombre seleccionado o vacio si no se realiza selección.              *

*----------------------------------------------------------------------*

*      -->PS_NOMBR    Nombre seleccionado                              *

*----------------------------------------------------------------------*

FORM VALORES_POSIBLES_NOMBRE USING  PS_NOMBR  LIKE ZCLIENXX-NOMBR.

************************************************************************

* Variables locales                                                    *

************************************************************************

* Tabla interna para almacenar los campos y atributos a visualizar.

DATA : BEGIN OF I_FIELDS OCCURS 1.

         INCLUDE STRUCTURE HELP_VALUE.

DATA:  END OF I_FIELDS.

* Tabla interna para almacenar el valor de campos a mostrar.

DATA : BEGIN OF I_VALUES OCCURS 0,

         VALOR(20) TYPE C,

       END OF I_VALUES.

************************************************************************

* Proceso                                                              *

************************************************************************

* Insertamos en la tabla I_FIELDS los atributos del campo a mostrar

  CLEAR I_FIELDS.

  I_FIELDS-TABNAME = 'ZCLIENXX'.      " Nombre de la tabla

  I_FIELDS-FIELDNAME = 'NOMBR'.       " Nombre del campo

  I_FIELDS-SELECTFLAG = 'X'.          " Valor seleccionable

  APPEND I_FIELDS.

* Seleccionamos los distintos nombres existentes en la tabla ZCLIENXX

* y los almacenamos en la tabla de valores

  SELECT DISTINCT NOMBR INTO ZCLIENXX-NOMBR

  FROM ZCLIENXX

  WHERE BUKRS = P_BUKRS.

    I_VALUES-VALOR = ZCLIENXX-NOMBR.

    APPEND I_VALUES.

  ENDSELECT.

* Llamamos a la función predefinida que muestra los datos

  CALL FUNCTION 'HELP_VALUES_GET_WITH_TABLE'

*      EXPORTING

*           CUCOL                         = 0

*           CUROW                         = 0

*           DISPLAY                       = ' '

*           FIELDNAME                     = ' '

*           TABNAME                       = ' '

*           NO_MARKING_OF_CHECKVALUE      = ' '

*           TITLE_IN_VALUES_LIST          = ' '

*           TITEL                         = ' '

*           SHOW_ALL_VALUES_AT_FIRST_TIME = ' '

*           NO_CONVERSION                 = ' '

       IMPORTING

            SELECT_VALUE                  = PS_NOMBR

       TABLES

            FIELDS                        = I_FIELDS

            VALUETAB                      = I_VALUES

       EXCEPTIONS

            FIELD_NOT_IN_DDIC             = 1

            MORE_THEN_ONE_SELECTFIELD     = 2

            NO_SELECTFIELD                = 3

            OTHERS                        = 4.

ENDFORM.                    " VALORES_POSIBLES_NOMBRE

*&---------------------------------------------------------------------*

*&      Form  MOSTRAR_FACTURAS_CLIENTE

*&---------------------------------------------------------------------*

*      Selecciona las facturas correspondientes al cliente recibido    *

*  como parámetro y las muestra en pantalla.                           *

*  Si no se seleccionan facturas, se muestra el mensaje correspondiente

*----------------------------------------------------------------------*

*      -->PE_CLIENX   -- Nº de cliente.                                *

*----------------------------------------------------------------------*

FORM MOSTRAR_FACTURAS_CLIENTE USING VALUE(PE_NCLIE) LIKE ZCLIENXX-NCLIE.

************************************************************************

* Proceso                                                              *

************************************************************************

* Seleccionamos las facturas del cliente para la sociedad indicada

  SELECT *

  FROM ZFACTUXX INTO TABLE I_ZFACTUXX

  WHERE BUKRS = P_BUKRS

  AND   NCLIE = PE_NCLIE.

* Si se han seleccionado facturas

  IF ( SY-SUBRC = 0 ).

*   Activamos color de detalle

    FORMAT COLOR COL_NORMAL.

*   Para cada una de las facturas seleccionadas

    LOOP AT I_ZFACTUXX.

*    Escribimos los datos de las facturas.

     WRITE: AT /C_POS_NFACT '|' NO-GAP, I_ZFACTUXX-NFACT.

*    Almacenamos el valor del nº de factura.

     HIDE I_ZFACTUXX-NFACT.

     WRITE: AT  C_POS_FEFAC '|' NO-GAP, I_ZFACTUXX-FECHA,

            AT  C_POS_MESFA '|' NO-GAP, I_ZFACTUXX-MESFA,

            AT  S_IMPNT     '|' NO-GAP, I_ZFACTUXX-IMPNT 

                               CURRENCY I_ZFACTUXX-MONED,

            AT  C_POS_MONED '|' NO-GAP, I_ZFACTUXX-MONED,

            AT  C_POS_FIFAC '|' NO-GAP.

    ENDLOOP.

*   Escribimos línea final

    ULINE AT /C_POS_NFACT(C_ANCHO_FAC).

* Si no se han seleccionado

  ELSE.

*   Mostramos mensaje de Status con

*   'No existen facturas para el cliente:'xxx 'para la sociedad:' xxx

    MESSAGE S000(38) WITH TEXT-015 PE_NCLIE TEXT-016 P_BUKRS.

  ENDIF.

ENDFORM.                    " MOSTRAR_FACTURAS_CLIENTE

*&---------------------------------------------------------------------*

*&      Form  MOSTRAR_DETALLE_FACTURA

*&---------------------------------------------------------------------*

*      Muestra el detalle de una factura, mostrando el cliente        *

*   al que pertenece y calculando el IVA correspondiente a la misma    *

*   su poniendo un tipo de IVA fijo del 16 %.                          *

*----------------------------------------------------------------------*

*      --> PE_NFACT   Código de factura                                *

*----------------------------------------------------------------------*

FORM MOSTRAR_DETALLE_FACTURA USING VALUE(PE_NFACT) LIKE ZFACTUXX-NFACT.

************************************************************************

* Definición de variables locales                                      *

************************************************************************

DATA: D_NOMBRE_CLIENTE(70) TYPE C,         " Nombre completo

      D_IMPOR_IVA   LIKE ZFACTUXX-IMPNT,   " Importe del IVA

      D_IMPOR_TOTAL LIKE ZFACTUXX-IMPNT.   " Importe total

************************************************************************

* Proceso                                                              *

************************************************************************

* Obtenemos los datos de la factura para ello leemos de la tabla interna

  READ TABLE I_ZFACTUXX WITH KEY NFACT = PE_NFACT.

* Comprobamos que existe

  CHECK ( SY-SUBRC = 0 ).

* Obtenemos los datos necesarios del cliente

  SELECT SINGLE NOMBR APEL1 APEL2

  FROM ZCLIENXX INTO (ZCLIENXX-NOMBR, ZCLIENXX-APEL1, ZCLIENXX-APEL2)

  WHERE BUKRS = P_BUKRS

  AND   NCLIE = I_ZFACTUXX-NCLIE.

* Comprobamos que existe.

  CHECK ( SY-SUBRC = 0 ).

* Escribimos los datos en la forma adecuada

  ULINE AT /C_POS_D_INICI(C_POS_D_FINAL).

* Nº de factura.

  WRITE: AT /C_POS_D_INICI '|',

         AT  C_POS_D_NFACT TEXT-022, I_ZFACTUXX-NFACT,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Formamos el nombre completo del cliente del cliente

  CONCATENATE ZCLIENXX-NOMBR ZCLIENXX-APEL1 ZCLIENXX-APEL2

              INTO D_NOMBRE_CLIENTE SEPARATED BY SPACE.

* Nombre del cliente

  WRITE: AT /C_POS_D_INICI '|',TEXT-023, D_NOMBRE_CLIENTE,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Fecha de factura

  WRITE: AT /C_POS_D_INICI '|', TEXT-024, I_ZFACTUXX-FECHA,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Mes de la factura.

  WRITE: AT /C_POS_D_INICI '|', TEXT-025, I_ZFACTUXX-MESFA,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Moneda de la factura.

  WRITE: AT /C_POS_D_INICI '|', TEXT-026, I_ZFACTUXX-MONED,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Importe neto:

  WRITE: AT /C_POS_D_INICI '|',

         AT C_POS_D_IMPOR   TEXT-027,

                            I_ZFACTUXX-IMPNT CURRENCY I_ZFACTUXX-MONED,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Calculamos el importe de IVA. (16 %).

  D_IMPOR_IVA = ( I_ZFACTUXX-IMPNT * 16 ) / 100.

  WRITE: AT /C_POS_D_INICI '|',

         AT  C_POS_D_IMPOR  TEXT-028,

                            D_IMPOR_IVA CURRENCY I_ZFACTUXX-MONED,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Calculamos el importe total

  D_IMPOR_TOTAL = I_ZFACTUXX-IMPNT + D_IMPOR_IVA.

  WRITE: AT /C_POS_D_INICI '|',

         AT  C_POS_D_IMPOR  TEXT-029,

                            D_IMPOR_TOTAL CURRENCY I_ZFACTUXX-MONED,

         AT  C_POS_D_FINAL '|' NO-GAP.

* Mostramos línea final

  ULINE AT /C_POS_D_INICI(C_POS_D_FINAL).

ENDFORM.                    " MOSTRAR_DETALLE_FACTURA

*&---------------------------------------------------------------------*

*&      Form  LEER_MARCAS_LISTADO

*&---------------------------------------------------------------------*

*      Actualiza el campo MARCA de la tabla interna I_ZCLIENXX con el  *

*  valor actual en el informe.                                         *

*----------------------------------------------------------------------*

FORM LEER_MARCAS_LISTADO.

************************************************************************

* Definición de variables locales

*

************************************************************************

DATA: L_LINEA LIKE SY-INDEX,        " Nº de línea de pantalla

      L_MARCA(1) TYPE C,            " Marca de pantalla

      L_NCLIE LIKE ZCLIENXX-NCLIE.  " Cliente de pantalla

************************************************************************

* Proceso                                                              *

************************************************************************

*  Inicializamos el contador de líneas

   L_LINEA = 0.

*  Leemos las líneas todas las líneas de pantalla

   DO.

*    Incrementamos el contador de líneas

     L_LINEA = L_LINEA + 1.

*    Leemos la línea correspondiente

     READ LINE L_LINEA FIELD VALUE I_ZCLIENXX-MARCA INTO L_MARCA

                                   I_ZCLIENXX-NCLIE INTO L_NCLIE.

*    Si no hay error en la lectura

     IF ( SY-SUBRC = 0 ).

*      Comprobamos si es una línea de detalle

       CHECK NOT ( L_NCLIE IS INITIAL ).

*      Leemos la entrada de la tabla correspondiente

       READ TABLE I_ZCLIENXX WITH KEY NCLIE = L_NCLIE BINARY SEARCH.

*      Si no hay error

       CHECK ( SY-SUBRC = 0 ).

*      Comprobamos si el campo marca es distinto al de la tabla

       IF ( I_ZCLIENXX-MARCA <> L_MARCA ).

*        Modificamos la entrada con el valor de pantalla

         I_ZCLIENXX-MARCA = L_MARCA.

         MODIFY I_ZCLIENXX INDEX SY-TABIX.

       ENDIF.

*    Si hay error en la lectura significará final de lista, terminamos

     ELSE.

       EXIT.

     ENDIF.

   ENDDO.

ENDFORM.                    " LEER_MARCAS_LISTADO

*&---------------------------------------------------------------------*

*&      Form  MOSTRAR_FACTURAS_CLIENTES

*&---------------------------------------------------------------------*

*       Muestra las facturas de los clientes seleccionados, es  decir *

*  de los que tengan el campo MARCA de la tabla interna I_ZCLIENXX     *

*  un valor 'X'. Para ello llama al procedimiento de mostrar           *

*  facturas de un cliente.

*----------------------------------------------------------------------*

FORM MOSTRAR_FACTURAS_CLIENTES.

************************************************************************

* Proceso                                                              *

************************************************************************

* Activamos el STATUS para la ventana, eliminando la opción

* de seleccionar para impedir navegar al detalle de factura.

  SET PF-STATUS 'S01' EXCLUDING 'PICK'.

* Abrimos una ventana para mostrar las facturas

  WINDOW STARTING AT 10 5 ENDING AT 68 20.

* Para cada uno de los clientes seleccionados

  LOOP AT I_ZCLIENXX

  WHERE MARCA = 'X'.

*  Con cada uno de los clientes llamamos al procedimiento,

*  que muestra las facturas de un cliente

   PERFORM MOSTRAR_FACTURAS_CLIENTE USING I_ZCLIENXX-NCLIE.

  ENDLOOP.

ENDFORM.                    " MOSTRAR_FACTURAS_CLIENTES

*&---------------------------------------------------------------------*

*&      Form  MARCAR_DESMARCAR

*&---------------------------------------------------------------------*

*       Actualizar el campo marca de todos los registros de la tabla   *

*  interna I_ZCLIENXX con el valor indicado en el parámetro.           *

*----------------------------------------------------------------------*

*      -->PE_MARCA     Valor para marcar ('X' / SPACE)                *

*----------------------------------------------------------------------*

FORM MARCAR_DESMARCAR USING  VALUE(PE_MARCA) TYPE C.

************************************************************************

* Proceso                                                              *

************************************************************************

* Actualizamos cada uno de los registros de la tabla interna que

* sean distinto al valor al que actualizar.

  LOOP AT I_ZCLIENXX

  WHERE MARCA <> PE_MARCA.

*   Actualizamos el valor

    I_ZCLIENXX-MARCA = PE_MARCA.

*   Modificamos el registro.

    MODIFY I_ZCLIENXX.

  ENDLOOP.

ENDFORM.                    " MARCAR_DESMARCAR

*&---------------------------------------------------------------------*

*&      Form  MOSTRAR_CLIENTES

*&---------------------------------------------------------------------*

*       Lista los clientes seleccionados por pantalla, es decir, los *

* clientes seleccionados en la tabla interna I_ZCLIENXX                *

*----------------------------------------------------------------------*

FORM MOSTRAR_CLIENTES.

************************************************************************

* Proceso                                                              *

************************************************************************

* Vamos a escribir el listado con los clientes seleccionados en la

* tabla interna I_ZCLIENXX

  LOOP AT I_ZCLIENXX.

*   Escribimos el campo MARCA en forma de casilla de selección

*   y con los atributos de campo de entrada.

    WRITE: AT /C_POS_INICI '|' NO-GAP,

              I_ZCLIENXX-MARCA AS CHECKBOX INPUT ON.

*   Escribimos el nº de cliente en color clave, el resto en otro color

    FORMAT COLOR COL_KEY.

*   Escribimos en modo Click.

    WRITE:  AT C_POS_NCLIE '|' NO-GAP, I_ZCLIENXX-NCLIE HOTSPOT.

*   Almacenamos el valor del campo

    HIDE I_ZCLIENXX-NCLIE.

    FORMAT COLOR COL_NORMAL.

    WRITE:  AT  C_POS_NOMBR '|' NO-GAP, I_ZCLIENXX-NOMBR,

            AT  C_POS_APEL1 '|' NO-GAP, I_ZCLIENXX-APEL1,

            AT  C_POS_APEL2 '|' NO-GAP, I_ZCLIENXX-APEL2,

            AT  C_POS_FNACI '|' NO-GAP, I_ZCLIENXX-FNACI,

            AT  C_POS_FINAL '|' NO-GAP.

  ENDLOOP.

* Si se han seleccionado datos, escribimos la línea final.

  IF ( SY-SUBRC = 0 ).

    ULINE AT /C_POS_INICI(C_ANCHO_TOTAL).

  ENDIF.

* Inicializamos el valor de la cabecera de la tabla

  CLEAR I_ZCLIENXX.

ENDFORM.                    " MOSTRAR_CLIENTES

*&---------------------------------------------------------------------*

*&      Form  MOSTRAR_CABECERA_CLIENTES

*&---------------------------------------------------------------------*

*      Muestra la cabecera del listado de los clientes existentes      *

*----------------------------------------------------------------------*

FORM MOSTRAR_CABECERA_CLIENTES.

************************************************************************

* Proceso                                                              *

************************************************************************

* Hacemos que la línea permanezca fija en el scroll horizontal

  NEW-LINE NO-SCROLLING.

* Activamos el color de cabecera

  FORMAT COLOR COL_HEADING INTENSIFIED OFF.

* Escribimos información de cabecera resaltando los valores

* Título y sociedad

  WRITE: AT C_POS_TITUL TEXT-009.

  WRITE P_BUKRS INTENSIFIED ON.

* Usuario de creación del informe

  WRITE AT C_POS_USUAR TEXT-010.

  WRITE SY-UNAME INTENSIFIED ON.

* Fecha de creación

  WRITE AT C_POS_FECHA TEXT-011.

  WRITE SY-DATUM INTENSIFIED ON USING EDIT MASK '__/__/____'.

* Nº de página.

  WRITE AT C_POS_PAGNO TEXT-012.

  WRITE SY-PAGNO INTENSIFIED ON.

* Escribimos una línea horizontal

  ULINE AT /C_POS_INICI(C_ANCHO_TOTAL).

* Activamos el color de cabeceras intensificado

  FORMAT COLOR COL_HEADING INTENSIFIED ON.

* Escribimos un espacio en blanco para el campo de marca

  WRITE AT /C_POS_INICI '|'.

* Escribimos la cabecera de las columnas separadas por el carácter '|'.

* Hacemos que hasta la columna de nº.cli. sea fija en el scroll horiz.

  WRITE: AT C_POS_NCLIE '|' NO-GAP, TEXT-004.

* Hasta la columna hacemos fijo

  SET LEFT SCROLL-BOUNDARY.

* Escribimos el detalle de cabecera  de cliente

  WRITE: AT  C_POS_NOMBR '|' NO-GAP, TEXT-005,

         AT  C_POS_APEL1 '|' NO-GAP, TEXT-006,

         AT  C_POS_APEL2 '|' NO-GAP, TEXT-007,

         AT  C_POS_FNACI '|' NO-GAP, TEXT-008,

         AT  C_POS_FINAL '|' NO-GAP.

* Escribimos una línea horizontal

  ULINE AT /C_POS_INICI(C_ANCHO_TOTAL).

ENDFORM.                    " MOSTRAR_CABECERA_CLIENTES

