1. FORMULARIOS

1.1 ¿Qué es un formulario?

Las empresas necesitan imprimir habitualmente informes, facturas, pedidos, etc. con un formato predefinido. Estos documentos pueden contener textos que se mantienen constantes entre documentos, pero también puede haber datos variables. En muchas ocasiones la creación de estos documentos se produce en masa creando múltiples copias de un mismo tipo de documento pero con propiedades distintas. De esta forma, la creación de los documentos como facturas, pedidos etc. se puede realizar de forma automática. SAPscript resuelve todos estos problemas con una plataforma común.

La creación de los documentos implica a varios elementos. Desde el punto de vista del programador SAPscript se puede dividir en formularios y programas de control de formularios. En realidad están implicados también el diccionario de SAP y el composer (un elemento que crea el formulario a partir de los elementos y de los requisitos del programa de control) Los formularios especifican la apariencia del texto en el documento (forman la plantilla del documento) y el programa de control especifica que valores contendrán los campos del formulario. De esta forma, para cambiar la apariencia de un documento sólo es necesario cambiar el formulario. Por el contrario, si se desea cambiar el contenido del documento es necesario cambiar los elementos de texto así como el programa de control del formulario. 

El programa de control del formulario controla la salida del documento hacia la impresora, fax, pantalla, número de copias... Este programa selecciona los datos que se van a mostrar en el formulario del diccionario de ABAP o de las entradas del usuario, así como el formulario controlando los textos que se deben imprimir, su secuencia y frecuencia.

La apariencia final del documento depende de la interacción del programa de control con el formulario. El programa de control inicializa y finaliza el proceso de impresión, transfiriendo los comandos de SAPscript al composer. El composer formatea el documento a partir de la información del diseño del formulario especificado en el programa de control. Si el documento contiene variables, el composer reemplaza el contenido de estas en su posición (fecha, hora...)

Ejemplo de una factura:

1.2 Editores de formularios

1.2.1 Editor Gráfico

Si entramos al editor de texto, seleccionando primero una ventana y pulsando luego en Elementos de texto veremos la siguiente pantalla:

En el campo de Tipo de Párrafo podemos escoger entre los formato de párrafo que hemos definido en dicha ventana al igual que en el campo de Formato de Carácter podemos escoger entre los formatos de carácter que nos hemos definido.

Para introducir comandos, símbolos, elemento de textos, etc. dentro del formulario en el menú Tratar ( Comando ( Insertar Comando nos permite insertar dicha información adicional.

1.2.2 Editor Alfanumérico

Para pasar al editor de texto alfanumérico, desde el menú Pasar seleccionar la opción de Cambiar editor. 

A diferencia del editor gráfico en el tipo de párrafo a parte de tener nuestro formato de párrafo definido anteriormente también podemos informar si la línea en cuestión es un comando, comentario, etc.

1.3 Estructura de un formulario

Para acceder al editor de formularios se sigue la ruta Herramientas -> Imprimir formularios->SAPscript-> Formulario, o directamente con la transacción SE71.

Accederemos a continuación a la pantalla Form Painter. Podemos crear un nuevo formulario, editar uno ya existente o modificarlo.

Nota: En el caso de no mostrar esta pantalla significará que estamos usando Form Painter gráfico. Es igual que el modo alfanumérico pero más visual. No obstante aquí seguiremos el modo alfanumérico accediendo por el menú->Opciones->Form Painter y desmarcando el flag de modo gráfico. 

Un formulario especifica la disposición de los siguientes elementos en las páginas de un documento:

· Cabecera 
· Páginas 
· Ventanas 
· Ventana página 
· Formatos de párrafo 
· Formatos caracteres 
· Elementos de texto 
1.3.1 Cabecera

La cabecera de un formulario consiste en atributos globales del formulario. Estos pueden ser datos de gestión (Nombre del formulario, descripción, clase de desarrollo...) y parametrizaciones básicas (formato de página, fuente por defecto, párrafo por defecto...). Cuando se crea un formulario, se accede directamente a la cabecera, la cual sirve para definir los datos generales del formulario. 

· Datos de gestión

Está ventana esta conformada por dos subdivisiones, info gestión y atributos idioma.

· Info gestión: 

Formulario: El nombre de un formulario no deberá tener más de 16 posiciones y sólo deberán utilizarse letras mayúsculas y cifras, comenzando siempre por una letra.

Significado: Descripción breve del formulario

· Atributos idioma:

      Se define el idioma principal y si el formulario podrá ser traducido a otros idiomas, o solo podrá mostrarse en el idioma original.

· Parametrizaciones básicas

En la ventana de parametrizaciones básicas se distinguen dos divisiones, una conformada por la configuración de la página y la otra con los valores propuestos para el formato del texto

Configurar página:

Formato de página: Indica el tamaño del papel.

Formato hoja: Indica la orientación del papel.

Líneas/pulg: Indica la cantidad de líneas de impresión por pulgada.

Carácter/pulg: Indica la cantidad de caracteres de impresión por pulgada.

Página inicial: Primera página que se va ha imprimir de nuestro formulario.
Valores propuesta p. formatear texto: Se especifican los valores tomados por defecto.
Párrafo por defecto: Si se indica '*' para el párrafo en el editor SAPscript, el sistema tomará el párrafo default para editar el texto.
Tabulaciones: Distancia entre las tabulaciones en un formulario.

Familia fonts, altura font...: Definen el formato de carácter por defecto.

1.3.2 Páginas

Representan las distintas páginas del documento. Estas suelen tener distinto aspecto unas de otras, la primera página de un fax contiene información distinta a la que se mostrara en las páginas siguientes. Para cada formulario se ha de definir al menos una página. En este apartado daremos una descripción a la página, indicaremos cual es la página siguiente y cómo se actualiza el contador de páginas, así como los atributos de impresión como son el modo de impresión (Por defecto, SYMPLEX, DUPLEX o TRIPLEX) y el nombre del recurso que será la bandeja de la impresora de donde tomará el papel (TRY01, TRY02, TRY03).

1.3.3 Ventanas

Las ventanas representan áreas que se posicionarán sobre las páginas. En ellas pondremos el párrafo por defecto, el nombre de la ventana, su significado y el tipo de ventana.

Hay 4 tipos de ventanas:

MAIN: Es la ventana principal en la que se escribirá el texto variable, como podría ser el cuerpo de una carta. Esta ventana puede extenderse a más de una página.

VAR: Es una ventana cuyo contenido puede variar. Estas ventanas deben definirse en cada página.

CONST: Define una ventana cuyo contenido no cambia.

GRAPH: Define una ventana con una imagen gráfica.

1.3.4 Ventana página

En este apartado, se especifica la posición y tamaño de las ventanas en cada página. Para añadir página, se realiza por medio del botón derecho del ratón en la opción crear elemento.

1.3.5 Formatos de párrafos
Los formatos de párrafos definen las características de estos, esta pantalla esta dividida en dos partes. En la zona superior se muestra una lista con todos los formatos de párrafo para el formulario, observándose las características generales de cada párrafo.

 Los atributos de cada párrafo están divididos en 4 partes: 

Para la creación de un nuevo párrafo se añade en la casilla de párrafo en la zona de atributos se añade el nombre que le identificara con uno o dos caracteres

· Datos estándar


Se define las características propias del párrafo, como márgenes, sangría y alineación.

Párrafo: Define el nombre con el cual se identificara con posterioridad las distintas características del párrafo creado.

Significado: Breve descripción identificativa del párrafo.

Margen izquierdo y derecho: Distancia con la ventana que contenga el párrafo.

Sangría 1ª línea: Sangría. 

Esp. anterior / posterior: Espacio de comienzo con respecto al párrafo anterior / posterior.

Alineación: Alineación del párrafo.

Interlineado: Espacio entre cada línea.

· Datos de Fuentes

Contiene las características de la fuente así como su formato.

Familia: fuente que se usará en el párrafo.

Altura: Tamaño de la fuente a utilizar

Negrita, cursiva y subrayado: Como indica el titulo permite la opción de que el párrafo este formateado con alguna de estas características si se marca la opción “On” quedarán activadas, si se selecciona la opción “Conser” se usara el valor que tenía el texto anterior.  

· Datos de tabuladores

Aquí se definen las distintas posiciones de tabulación que necesitemos para cada párrafo.

Le deberemos indicar una posición, ya sea en centímetros (CM), caracteres (CH), milímetros (MM), puntos (PT) y un tipo de alineación: izquierda (LEFT), derecha (RIGHT), centrado (CENTER), al signo (SIGN) o a la coma decimal (DECIMAL)

· Datos de esquema

Es posible definir numeración y marcas automáticas de forma que podamos estructurar texto en capítulos, subcapítulos y secciones.

El signo fijo será el carácter que se antepondrá siempre al párrafo, si el signo es ‘_’ está se convertirá en espacio en blanco.

1.3.6 Formatos de caracteres

Dentro de cualquier párrafo es posible cambiar el tipo de letra para uno o más caracteres. 

· Atributos estándar

Formato caráct.: Nombre con un máximo de dos letras con el cual se identificara el tipo de formato de carácter creado.

Signific.: Breve descripción que defina el formato de caracteres.

Cód. Barras: Se introduce el código numérico de barras para la posterior visualización del mismo.

Lín. Unida, oculto, superíndice, subíndice: Se especifican las distintas opciones que tendrá el formato de caracteres. “Conser” conservara los parámetros especificados en el texto anterior.

· Atributos font

Contiene los datos de la fuente que se usara en el formato de carácter.

Las características en los atributos Font son los mismos que para los formatos de párrafo.

1.3.7 Elementos de texto

Los elementos de texto son componentes individuales de una ventana. Puede haber más de un elemento de texto por ventana y se distinguen dos tipos. Los elementos de texto pueden contener variables e instrucciones de control SAPscript. Para acceder a ellos nos situamos en la pantalla de Ventana Pagina, seleccionamos la ventana donde queremos insertar los elementos de texto y pulsamos el botón de Elementos de texto (F9).

El programa de impresión accede a los elementos de texto por nombre, los formatea e imprime en la ventana correspondiente.

1.4 SAPScript

1.4.1 Cajas, líneas y sombreados

Se pueden dibujar cajas y líneas en SAPScript mediante los siguientes comandos:

BOX: Dibuja una caja o una línea.

POSITION: Especifica el punto inicial de una caja o línea

SIZE: Especifica la anchura o altura de una caja.

Ejemplo:

Creamos una nueva ventana de tipo CONST de idénticas proporciones que la ventana MAIN que dibuje una caja que rodee las dos columnas de la ventana MAIN del ejemplo.

/:  BOX XPOS '0.5' CM WIDTH 8 CM HEIGHT 15 CM FRAME 10 TW.

/:  BOX XPOS '8.5' CM WIDTH 8 CM HEIGHT 15 CM FRAME 10 TW.

1.4.2 Comandos de control

SAPScript dispone de una serie de comandos que permiten obtener un control total sobre el texto. Estos comandos son introducidos en el editor de textos igual que una línea normal excepto que el deberemos seleccionar el párrafo /:

NEW-PAGE: Provoca el salto automático de página

PROTECT – ENDPROTECT: Se asegura que el texto introducido entre estos dos comandos aparezca siempre en una misma página.

NEW-WINDOW: Llama automáticamente a la siguiente ventana MAIN definida en una misma página.

DEFINE: Permite crear una constante con un valor dado.

SET DATE MASK: Define el formato de los campos fecha.

SET TIME MASK: Define el formato de los campos hora.

SET COUNTRY: Define el formato de ciertos campos como el punto para los millares adaptados a un país específico.

SET SIGN: Indica la posición del signo.

RESET: Inicializa el contador de un párrafo con numeración.

INCLUDE: Inserta el contenido de otro texto en el texto actual.

STYLE: Cambia el contenido del estilo actual del texto.

ADDRESS – ENDADDRESS: Formatea una dirección de acuerdo con las convenciones postales del país definido por el parámetro COUNTRY.

TOP – ENDTOP: Especifica líneas de texto que siempre aparecerán en la parte superior de la ventana MAIN.

BOTTOM – ENDBOTTOM: Especifica líneas de texto que aparecerán en la parte inferior de la ventana MAIN.

IF – ENDIF: Permite especificar que líneas debería imprimirse cuando se cumplan ciertas condiciones.

CASE: Cubre el caso de múltiples comandos IF anidados.

PERFORM: Permite llamar una rutina de un programa ABAP

PRINT-CONTROL: Llama directamente ciertas funciones de la  impresora.

BOX, POSITION, LINE: Dibuja líneas y cajas.

HEX – ENDHEX: Envía a la impresora órdenes en el lenguaje que utiliza la impresora.

SUMMING: Acumula un valor total para un símbolo de programa.

1.4.3 Símbolos SAPScript 

La información variable se introduce en los formularios SAPScript mediante “símbolos” o variables que SAP reconoce porque van rodeadas de ‘&’. Ej.: &symbol&.

Símbolos del sistema: variable como fecha, hora…

Símbolos de programa: variables almacenas en aplicaciones SAP como campos del diccionario de datos o variables globales de los programas.

Símbolos estándar definidos en la tabla TTDTG: El valor de estos símbolos es dependiente del lenguaje y puede contener hasta 60 caracteres. SAP mantiene esta tabla con valores estándar.

Símbolos de texto: Aquellos que no corresponden a los tipos de símbolos definidos anteriormente. Son definidos por el usuario en el editor de texto, eligiendo Incluir -> Símbolos ->Texto. O bien definiéndolos con el comando INCLUDE.

1.4.4 Símbolos del sistema
	&DATE& 
	Fecha del sistema. Se imprimirá de acuerdo al tipo de SET DATE MASK definida anteriormente.

	&DAY&    
	Día del sistema.

	&MONTH&
	Mes

	&YEAR&
	Año

	&NAME_OF_DAY&
	Nombre del día de la semana del recogido en el SET DATE MASK.

	&NAME_OF_MONTH&
	Nombre del mes

	&TIME&
	Hora según el formato especificado en el SET TIME MASK.

	&HOURS&
	Hora

	&MINUTES&
	Minuto

	&SECONDS&
	Segundo

	&PAGE&
	Número de la página actual

	&NEXTPAGE&
	Número de la página siguiente a la actual

	&DEVICE&       
	Tipo del dispositivo de salida (PRINTER, SCREEN, TELEX, ABAP).

	&SPACE&         
	Espacio.

	&ULINE&                       
	Línea de subrayado

	&VLINE&                     
	Línea vertical


1.4.5 Campos generales de SAPScript

&SAPSCRIPT-SUBRC& Recibe un valore después de ejecutar un comando INCLUDE. (0 si lo encuentra, 4 si no).

&SAPSCRIPT-DRIVER& Nombre del dispositivo de salida (POST, HPL2, PRES).

&SAPSCRIPT-FORMPAGES& Nº total de páginas del form.

&SAPSCRIPT-JOBPAGES& Nº total de páginas de todos los formularios contenido en la petición de impresión actual.

&SAPSCRIPT_CONTER_X& (x = 0…9) Representan 10 variables de contador que se puede utilizar en el texto y formularios para cualquier propósito.

1.4.6 Opciones de formato de los símbolos

Desplazamiento 


&symbol+offset&

Longitud de salida 


&symbol(length)&

Omisión del signo


&symbol(S)&

Mostrar el signo por la izquierda

&symbol(<)&

Mostrar el signo por la derecha

&symbol(>)&

Omitir los ceros iniciales


&symbol(Z)&

Comprimir los espacios


&symbol(C)&

Número de decimales


&symbol(.2)&

Omitir el indicador de miles

&symbol(T)&

Especificar exponente 


&symbol(E2)&

Alinear a la derecha


&symbol(8R)&

Rellenar de caracteres


&symbol(Ff)&

Suprimir valores iniciales

&symbol(I)&

Ignorar rutinas de conversión

&symbol(K)&

Cambiar valor de un contador 

&SAPSCRIPT_COUNTER_X(+)&

Textos precedentes


&’pre-text’symbol’post-text’&

1.4.7 Formularios en varios idiomas

Una vez tenemos creado el formulario en un idioma padre, podemos crearlo en otros idiomas. Para ello en la pantalla de mantenimiento de formularios, seleccionamos el formulario que acabamos de crear, seleccionamos el nuevo lenguaje del formulario y seleccionamos crear. Veremos que trabajamos con una copia del formulario anterior en el que lo único que deberemos hacer será traducir los textos y adaptar el formulario, si fuera necesario, a las peculiaridades del nuevo idioma.

Para llamar al formulario en distintos idiomas se realiza mediante la función de abrir formulario, indicando el idioma requerido. Si el formulario no existiera en ese idioma, se abrirá el formulario en el idioma padre.

CALL FUNCTION 'OPEN_FORM'

     EXPORTING

          FORM     = 'Z_PRUEBA'

          LANGUAGE = P_IDIOMA

          OPTIONS  = ITCPO

          DEVICE   = 'PRINTER'

          DIALOG   = 'X'

     EXCEPTIONS

          OTHERS   = 1.

1.4.8 Inclusión de gráficos

Para incluir gráficos en un formulario, primero este debe estar grabado en SAP en formato TIFF, BMP o en forma de elemento de texto. Luego se incluye en la ventana deseada a través del menú Incluir -> Función gráfica y seleccionando el gráfico. Esto generará una línea de comando como la siguiente:

BITMAP IDES_LOGO OBJECT GRAPHICS ID BMAP TYPE BMON DPI 300

1.5 Programa de impresión del formulario

Una vez finalizado el diseño del formulario es necesario crear un programa que gestione la impresión del mismo. Para ello hay una serie de funciones Standard de SAP que gestionan todos los parámetros de salida. De todas ellas, las más relevantes son:

OPEN_FORM

WRITE_FORM

CLOSE_FORM

OPEN_FORM: Este modulo de función abre un formulario para su impresión. Esta función ha de ejecutarse antes que cualquier otra función que actúe sobre el formulario (WRITE_FORM, START_FORM, CONTROL_FORM...). Cada vez que se utiliza la función OPEN_FORM es necesario cerrar el formulario (CLOSE_FORM) para que este se imprima. Dentro de un mismo programa puede haber varios pares de llamadas a las funciones OPEN_FORM y CLOSE_FORM.

CLOSE_FORM: Cierra un formulario abierto previamente con la función OPEN_FORM.

WRITE_FORM: El sistema muestra un elemento de texto determinado del formulario. El elemento de texto se especifica en el parámetro exportado ELEMENT. En el parámetro WINDOW se puede especificar el nombre de la ventana de salida. 

Estructura ITCPO representa los parámetros de control del formato de salida. Esta estructura se puede utilizar en los módulos de función PRINT_TEXT y OPEN_FORM en el parámetro OPTIONS. 

	TDPAGESLCT 
	SAPscript: seleccionar página de impresión 

	TDPREVIEW 
	SAPscript: habilitar vista previa 

	TDNOPREV 
	SAPscript: deshabilitar vista previa

	TDNOPRINT 
	SAPscript: deshabilitar impresión desde vista previa 

	TDTITLE 
	SAPscript: Título de la pantalla de selección 

	TDPROGRAM 
	SAPscript: nombre del programa de símbolos de sustitución 

	TDTEST 
	SAPscript: visualización previa

	TDIEXIT 
	SAPscript: volver inmediatamente después de la impresión 

	TDGETOTF 
	SAPscript: valor de retorno te la tabla OTF, no hay impresión 

	TDSCRNPOS 
	SAPscript: posición del OTF en la pantalla

	TDDEST 
	Spool: nombre del dispositivo de salida 

	TDPRINTE 
	Spool: nombre del tipo de dispositivo 

	TDCOPIES 
	Spool: numero de copias 

	TDNEWID 
	Spool: nueva petición 

	TDIMMED 
	Spool: petición de impresión inmediata 

	TDDELETE 
	Spool: borrar petición después de la impresión 

	TDLIFETIME 
	Spool: tiempo de retención de la petición 

	TDDATASET 
	Spool: identificación de la petición 

	TDSUFFIX1 
	Spool: primer sufijo de la petición 

	TDSUFFIX2 
	Spool: segundo sufijo de la petición 

	TDAUTORITY 
	Spool: autorización para la petición 

	TDARMOD 
	Spool: modo de archivo

	TDCOVER 
	Spool: imprimir portada

	TDCOVTITLE 
	Spool: portada: título 

	TDRECEIVER 
	Spool: portada: nombre del destinatario

	TDDIVISION 
	Spool: portada: nombre de la división

	TDSCHEDULE 
	 SAPcomm: tipo del tiempo de envío estimado 

	 TDSENDDATE 
	 SAPcomm: fecha de envío solicitada 

	 TDSENDTIME 
	 SAPcomm: hora de envío solicitada 

	 TDTELELAND 
	 SAPcomm: código del país destinatario 

	 TDTELENUM 
	 SAPcomm: número de marcación


Ejemplo de formulario

Formulario 

--------------------------------------------------------------------------------

Formulario           ZPRUEBA

--------------------------------------------------------------------------------

Significado          Formulario de prueba

Atributos std.

  Página inicial      PRICIPAL

  Párrafo defecto     DF

  Tabulaciones        1,00 CM

  Formato página      DINA4

  Formato hoja        Formato vertical

  Líneas/pulg.          6,00

  Carácter/pulg.       10,00

Atributos font

  Familia fonts       COURIER

  Altura font         12,0 Punto

  Negrita             no

  Cursiva             no

  Subrayado           no

--------------------------------------------------------------------------------

Carácteres    Atributos

--------------------------------------------------------------------------------

  DF          Normal

              Atributos std.

                 Marca             no

              Atributos font

                 Familia fonts     HELVE

                 Altura font       10,0 Punto

  NG          Negrita

              Atributos std.

                 Marca             no

              Atributos font

                 Familia fonts     HELVE

                 Altura font       10,0 Punto

                 Negrita           sí

--------------------------------------------------------------------------------

Párrafos      Atributos

--------------------------------------------------------------------------------

  AD          Dirección propia

              Atributos std.

                 Interlineado      0.50 LN

                 Alineación        Alin.derecha

              Atributos font

                 Familia fonts     HELVE

                 Altura font       8,0 Punto

  DF          Párrafo por defecto

              Atributos std.

                 Interlineado      1.00 LN

                 Alineación        alin.izq.

              Atributos font

                 Familia fonts     HELVE

                 Altura font       10,0 Punto

  NG          Negrita

              Atributos std.

                 Interlineado      1.00 LN

                 Alineación        alin.izq.

              Atributos font

                 Familia fonts     HELVE

                 Altura font       10,0 Punto

                 Negrita           sí

--------------------------------------------------------------------------------

Ventanas      Atributos

--------------------------------------------------------------------------------

  ADDRESS     Dirección

                 Tipo ventana      VAR

                 Párr.p.defecto    DF

  CONTROL     Ventana de control

                 Tipo ventana      VAR

                 Párr.p.defecto    DF

  D_FISCAL    Datos fiscales

                 Tipo ventana      VAR

                 Párr.p.defecto    DF

  FACTURA     Datos Factura

                 Tipo ventana      VAR

                 Párr.p.defecto    DF

  LOGO        Ventana de logo

                 Tipo ventana      CONST

                 Párr.p.defecto    DF

  MAIN        Ventana princ.

                 Tipo ventana      MAIN

                 Párr.p.defecto    DF

  MARCO       marco de la ventana main

                 Tipo ventana      CONST

                 Párr.p.defecto    DF

  MY_ADD      Dirección propia

                 Tipo ventana      CONST

                 Párr.p.defecto    AD

  PAGO        Forma de pago

                 Tipo ventana      VAR

                 Párr.p.defecto    DF

--------------------------------------------------------------------------------

Páginas       Atributos

--------------------------------------------------------------------------------

  PRICIPAL    Primera página

              Atributos std.

                 Pág.subsiguiente  SEGUNDA

              Atributos impresión

                 Modo impresión    S

              Contad.pág.

                 Modo              INC

                 Tipo numeración   cifras árabes

              Ventana página

                 MAIN                 Margen izqdo.        1.25  CM

                                      Margen superior      8.89  CM

                                      Ancho ventana       18.27  CM

                                      Altura ventana      12.47  CM

                 ADDRESS              Margen izqdo.       10.49  CM

                                      Margen superior      3.95  CM

                                      Ancho ventana        9.14  CM

                                      Altura ventana       1.98  CM

                 CONTROL              Margen izqdo.       16.50  CM

                                      Margen superior     22.00  CM

                                      Ancho ventana        3.25  CM

                                      Altura ventana       3.00  CM

                 D_FISCAL             Margen izqdo.        1.25  CM

                                      Margen superior      3.95  CM

                                      Ancho ventana        8.89  CM

                                      Altura ventana       1.98  CM

                 FACTURA              Margen izqdo.        1.25  CM

                                      Margen superior      6.91  CM

                                      Ancho ventana       18.40  CM

                                      Altura ventana       1.48  CM

                 LOGO                 Margen izqdo.        1.25  CM

                                      Margen superior      1.25  CM

                                      Ancho ventana        1.25  CM

                                      Altura ventana       1.25  CM

                 MARCO                Margen izqdo.        1.20  CM

                                      Margen superior      8.84  CM

                                      Ancho ventana       18.27  CM

                                      Altura ventana      12.47  CM

                 MY_ADD               Margen izqdo.       10.50  CM

                                      Margen superior      1.25  CM

                                      Ancho ventana        9.25  CM

                                      Altura ventana       2.75  CM

                 PAGO                 Margen izqdo.        1.25  CM

                                      Margen superior     22.00  CM

                                      Ancho ventana       14.70  CM

                                      Altura ventana       3.00  CM

  SEGUNDA     Página segunda y siguientes

              Atributos std.

                 Pág.subsiguiente  SEGUNDA

              Contad.pág.

                 Modo              INC

                 Tipo numeración   cifras árabes

              Ventana página

                 MAIN                 Margen izqdo.        1.25  CM

                                      Margen superior     11.50  CM

                                      Ancho ventana       18.27  CM

                                      Altura ventana      10.00  CM

--------------------------------------------------------------------------------

Elementos texto para ventanas:

--------------------------------------------------------------------------------

ADDRESS

*  <ng>DIRECCIÓN POSTAL</>

/: ADDRESS

*  &CALLE&

*  &CIUDAD&

*  &PAIS&

/: ENDADDRESS

*

CONTROL

*  <ng>CONTROL</>

*

D_FISCAL

*  <ng>DATOS FISCALES</>

*  &C_NIF&

*  &N_RESERVA&

FACTURA

NG NºFACTURA,,COD CLIENTE,,NOMBRE,, ,, ,,FECHA PEDIDO,,FECHA FACTURA

DF &CABECERA-NUMERO_FACTURA&,,&CABECERA-IDCLIENTE&,,&CABECERA-DESCRIPCION&

   ,,&cabecera-fecha_pedido&,,&Cabecera-fecha_factura&

LOGO

*

/: BITMAP IDES_LOGO OBJECT GRAPHICS ID BMAP TYPE BMON DPI 300

*

MAIN

Elemento ITEM_HEADER

NG POSICION,,CONCEPTO,, ,, ,, ,, ,, ,, ,, ,, CANTIDAD,, ,,PRECIO

Elemento DATOS

DF &i_posicion-posicion&,, ,,&i_posicion-concepto(50)&,,

   &i_posicion-cantidad&,, ,,  &I_POSICION-PRECIO(5)&

Elemento PUBLICIDAD

/: PROTECT

NG ************************************************************************

NG &C_MENSAJE&

NG ************************************************************************

/: ENDPROTECT

Elemento TOTAL

DF

DF

DF ,, ,, ,, ,, ,, ,, ,, ,, ,, ,, SUBTOTAL,, &SUBTOTAL&

DF ,, ,, ,, ,, ,, ,, ,, ,, ,, ,, IVA,, ,, ,, &IVA&

NG  ,, ,, ,, ,, ,, ,, ,, ,, ,, ,,TOTAL,, &TOTAL&

MARCO

/: BOX XPOS 0 CM WIDTH '16.57' CM HEIGHT '12.47' CM FRAME 10 TW.

MY_ADD

AD IDES Holding AG

AD Neurottstrasse 16

AD Waldorf, 69190

AD Germany

PAGO

*  FORMA DE PAGO

Programa de control

REPORT zform .

************************************************************************

* VARIABLES

************************************************************************

DATA BEGIN OF itcpo.

        INCLUDE STRUCTURE itcpo.         "SAPscript Salida interfase

DATA END OF itcpo.

DATA: c_nif(10) TYPE c VALUE '12345678-Z',

      c_mensaje(74) TYPE c,

      n_reserva(5) TYPE n VALUE '12345',

      total(10),

      subtotal(10),

      iva(10).

c_mensaje =

'¡Nueva línea de productos !'.

DATA: BEGIN OF cabecera,

        numero_factura(10) TYPE c VALUE '0025698094',

        idcliente(10) TYPE c VALUE      '5556981254',

        descripcion(50) TYPE c VALUE 'Mario Lopez Alvarez',

        fecha_pedido LIKE sy-datum VALUE '20030210',

        fecha_factura LIKE sy-datum VALUE '20030214',

      END OF cabecera.

DATA: BEGIN OF i_posicion OCCURS 0,

        posicion(3) TYPE c,

        concepto(50) TYPE c,

        cantidad TYPE i,

        precio(5) TYPE c,

      END OF i_posicion.

* Variables para comunicarnos con el formulario

DATA: calle(40),

      ciudad(20),

      pais(20).

* Configuración de la impresora

PERFORM configurar_impresora.

* Abrimos el formulario

CALL FUNCTION 'OPEN_FORM'

     EXPORTING

          form     = 'ZPRUEBA'

*          LANGUAGE = P_IDIOMA

          options  = itcpo

          device   = 'PRINTER'

*          DIALOG   = 'X'

          dialog = space            " Sin diálogo

     EXCEPTIONS

          OTHERS   = 1.

IF sy-subrc NE 0.

  WRITE /'error al abrir formulario'.

  STOP.

ENDIF.

* Carga el contenido de las variables

PERFORM llenar_variables.

* Muestra el cuerpo del formulario

PERFORM mostrar_cuerpo_form.

CALL FUNCTION 'WRITE_FORM'

     EXPORTING

          window = 'MARCO'

     EXCEPTIONS

          OTHERS = 1.

* Imprimimos la ventana dirección

calle = 'C/ Modesto Lafuente 23'.

ciudad  = '28003, Madrid'.

pais = 'SPAIN'.

CALL FUNCTION 'WRITE_FORM'

     EXPORTING

          window = 'ADDRESS'

     EXCEPTIONS

          OTHERS = 1.

IF sy-subrc NE 0.

  WRITE /'error al escribir en ventana DIRECCION'.

  STOP.

ENDIF.

CALL FUNCTION 'CLOSE_FORM'.

*&---------------------------------------------------------------------*

*&      Form  CONFIGURAR_IMPRESORA

*&---------------------------------------------------------------------*

*       Configura los atributos de la impresora

*----------------------------------------------------------------------*

*  -->  p1

*  <--  p2

*----------------------------------------------------------------------*

FORM configurar_impresora.

  itcpo-tdpageslct  = space.             "Todas las páginas

  itcpo-tdnewid     = 'X'.               "Crea nuevo spool

  itcpo-tdcopies    = 1.                                    "1 copia

  itcpo-tddest      = 'LP01'.            "Nombre de la impresora

  itcpo-tdpreview = 'X'.                 " Visualización previa

  itcpo-tdcover     = space.             "No portada

  itcpo-tdimmed     = 'X'.               "Imprime inmediatamente

  itcpo-tddelete    = 'X'.               "Borra después de imprimir

  itcpo-tdcovtitle  = 'Ejemplo Formularios'.

  itcpo-tdtitle     = 'Ejemplo Formularios'.

ENDFORM.                    " CONFIGURAR_IMPRESORA

*&---------------------------------------------------------------------*

*&      Form  LLENAR_VARIABLES

*&---------------------------------------------------------------------*

*       Da valores a las variables que se mostrarán en el formulario

*----------------------------------------------------------------------*

*  -->  p1

*  <--  p2

*----------------------------------------------------------------------*

FORM llenar_variables.

* Llena la tabla interna de posiciones

  i_posicion-posicion = '1'.

  i_posicion-concepto = 'Papel fotocopiadora (2500 hojas)'.

  i_posicion-cantidad = 5.

  i_posicion-precio = ' 8.50'.

  APPEND i_posicion.

  i_posicion-posicion = '2'.

  i_posicion-concepto = 'Boligrafo tinta azul'.

  i_posicion-cantidad = 100.

  i_posicion-precio = '50.00'.

  APPEND i_posicion.

  i_posicion-posicion = '3'.

  i_posicion-concepto = 'Portaminas 0.5'.

  i_posicion-cantidad = 50.

  i_posicion-precio = '99.35'.

  APPEND i_posicion.

  i_posicion-posicion = '4'.

  i_posicion-concepto = 'Caja grapas (500)'.

  i_posicion-cantidad = 25.

  i_posicion-precio = '12.50'.

  APPEND i_posicion.

  i_posicion-posicion = '5'.

  i_posicion-concepto = 'Hojas transparencias'.

  i_posicion-cantidad = 15.

  i_posicion-precio = '90.50'.

  APPEND i_posicion.

*  TOTALES

  total = '302.59 Euros'.

  iva = ' 41.74'.

  subtotal = '250.85 Euros'.

ENDFORM.                    " LLENAR_POSICIONES

*&---------------------------------------------------------------------*

*&      Form  mostrar_cuerpo_form

*&---------------------------------------------------------------------*

*       text

*----------------------------------------------------------------------*

*  -->  p1        text

*  <--  p2        text

*----------------------------------------------------------------------*

FORM mostrar_cuerpo_form .

* Imprimos la ventana principal

  CALL FUNCTION 'WRITE_FORM'

       EXPORTING

            window  = 'MAIN'

            element = 'ITEM_HEADER'

       EXCEPTIONS

            OTHERS  = 1.

* Muestra las posiciones

  LOOP AT i_posicion.

    CALL FUNCTION 'WRITE_FORM'

         EXPORTING

              window  = 'MAIN'

              element = 'DATOS'

         EXCEPTIONS

              OTHERS  = 1.

  ENDLOOP.

* Muestra los totales

  CALL FUNCTION 'WRITE_FORM'

       EXPORTING

            window  = 'MAIN'

            element = 'TOTAL'

       EXCEPTIONS

            OTHERS  = 1.

* Imprime la zona de publicidad

  CALL FUNCTION 'WRITE_FORM'

       EXPORTING

            window  = 'MAIN'

            element = 'PUBLICIDAD'

       EXCEPTIONS

            OTHERS  = 1.

ENDFORM.                    " mostrar_cuerpo_form

